

# Deklaratív programozás

---

Hanák Péter

hanak@inf.bme.hu

Irányítástechnika és Informatika Tanszék

Szeredi Péter

szeredi@cs.bme.hu

Számítástudományi és Információelméleti Tanszék

# KÖVETELMÉNYEK, TUDNIVALÓK

---

## Deklaratív programozás: tudnivalók

---

### ● Honlap, levelezési lista

- Honlap: <<http://dp.iit.bme.hu>>

- Levlista: <<http://www.iit.bme.hu/mailman/listinfo/dp-l>>.

A listatagoknak szóló levelet a <[dp-l@www.iit.bme.hu](mailto:dp-l@www.iit.bme.hu)> címre kell küldeni.

Csak a feliratkozottak levele jut el moderátori jóváhagyás nélkül a listatagokhoz.

### ● Jegyzet

- Szeredi Péter, Benkő Tamás: Deklaratív programozás. Bevezetés a logikai programozásba (1000 Ft)

- Elektronikus változata elérhető a honlapról (ps, pdf)

- A nyomtatott változat **KORLÁTOZOTT SZÁMBAN** megvásárolható a SZIT tanszék V2 épületbeli titkárságán a V2.104 szobában, Bazsó Lászlónénál, 10:30-12:00 (hétfő-péntek) és 13:30-15:30 (hétfő-csütörtök).

- Kellő számú további igény esetén megszervezzük az újranyomtatást.

## Deklaratív programozás: tudnivalók (folyt.)

---

### Fordító- és értelmezőprogramok

- SICStus Prolog — 3.12 verzió (licenz az ETS-en keresztül kérhető)
- Erlang (szabad szoftver)
- Mindkettő telepítve van a `kempelen.inf.bme.hu`-n
- Mindkettő letölthető a honlapról (linux, Win95/98/NT)
- Webes gyakorló felület az ETS-ben (ld. honlap)
- Kézikönyvek HTML-, ill. PDF-változatban
- Más programok: swiProlog, gnuProlog
- emacs-szövegszerkesztő Erlang-, ill. Prolog-módban (linux, Win95/98/NT)

## Deklaratív programozás: félévközi követelmények

---

### Nagy házi feladat (NHF)

- Programozás mindkét nyelven (Prolog, Erlang)
- Mindenkinek önállóan kell kódolnia (programoznia)!
- Hatékony (időlimit!), jól dokumentált („kommentezett”) programok
- A két programhoz közös, 5–10 oldalas fejlesztői dokumentáció (TXT,  $\text{T}_{\text{E}}\text{X}/\text{L}_{\text{A}}\text{T}_{\text{E}}\text{X}$ , HTML, PDF, PS; de nem DOC vagy RTF)
- Kiadás a 6. héten, a honlapon, letölthető keretprogrammal
- Beadás a 12. héten; elektronikus úton (ld. honlap)
- A beadáskor és a pontozáskor külön-külön tesztorozatot használunk (nehézségben hasonlókat, de nem azonosakat)
- A minden tesztesetet hibátlanul megoldó programok *létraversenyen* vesznek részt (hatékonyság, gyorsaság plusz pontokért)

## Deklaratív programozás: félévközi követelmények (folyt.)

---

### Nagy házi feladat (folyt.)

- Kötelező
- A beadási határidőig többször is beadható, csak az utolsót értékeljük
- Pontozása mindkét nyelvből:
  - helyes és időkorláton belüli futás esetén a 10 tesztet mindegyikére 0,5-0,5 pont, összesen max. 5 pont, feltéve, hogy legalább 4 tesztet sikeres
  - a dokumentációra, a kód olvashatóságára, kommentezettségére max. 2,5 pont
  - tehát nyelvenként összesen max. 7,5 pont szerezhető
- A NHF súlya az osztályzatban: 15% (a 100 pontból 15)

## Deklaratív programozás: félévközi követelmények (folyt.)

---

### Kis házi feladatok (KHF)

- 3 feladat Prologból is, Erlang-ból is
- Beadás elektronikus úton (ld. honlap)
- Kötelező legalább a KHF-ek 50%-ának a beadása
- Minden feladat jó megoldásáért 1-1 jutalompont jár

### Gyakorló feladatok

- Nem kötelező, de a sikeres ZH-hoz, vizsgához *elengedhetetlen!*
- Gyakorlás az ETS rendszerben (lásd honlap)

### Konzultációk

- Rendszeres – számítógép melletti – konzultációs lehetőség

## Deklaratív programozás: félévközi követelmények (folyt.)

---

### Nagyzárthelyi, pótzárthelyi (NZH, PZH, PPZH)

- A zárthelyi kötelező, semmilyen jegyzet, segédlet nem használható!
- 40%-os szabály (nyelvenként a maximális részpontszám 40%-a kell az eredményességhez).  
Kivétel: a korábban aláírást szerzett hallgató zárthelyin szerzett pontszámát az alsó ponthatártól függetlenül beszámítjuk a félévvégi osztályzatba. A korábbi félévekben szerzett pontokat nem számítjuk be!
- Az NZH az órarendben előírt héten, a PZH az utolsó oktatási hetekben lesz
- A PPZH-ra indokolt esetben a pótlási időszakban egyetlen alkalommal adunk lehetőséget
- Az NZH anyaga az addig előadott tananyag.
- A PZH, ill. a PPZH anyaga azonos az NZH anyagával
- A zárthelyi súlya az osztályzatban: 15% (a 100 pontból 15)



## Deklaratív programozás: vizsga

---

### Vizsga

- Vizsgára az a hallgató bocsátható, aki aláírást szerzett a jelen félévben vagy a jelen félévet megelőző négy félévben
- A vizsga szóbeli, felkészülés írásban
- Prolog, Erlang: több kisebb feladat (programírás, -elemzés) kétszer 35 pontért
- A vizsgán szerezhető max. 70 ponthoz adjuk hozzá a félévközi munkával szerzett pontokat: ZH: max. 15 pont, NHF: max. 15 pont, továbbá a pluszpontokat (KHF, létraverseny)
- A vizsgán semmilyen jegyzet, segédlet nem használható, de lehet segítséget kérni
- Ellenőrizzük a nagy házi feladat és a zárthelyi „hitelességét”
- 40%-os szabály (nyelvenként a max. részpontszám 40%-a kell az eredményességhez)
- Korábbi vizsgakérdések a honlapon találhatóak

# BEVEZETÉS A LOGIKAI PROGRAMOZÁSBA

## A logikai programozás alapgondolata

---

- Logikai programozás (LP):
  - Programozás a matematikai logika segítségével
    - egy logikai program nem más mint **logikai állítások halmaza**
    - egy logikai **program futása** nem más mint **következtetési folyamat**
  - De: a logikai következtetés óriási keresési tér bejárását jelenti
    - szorítsuk meg a logika nyelvét
    - válasszunk egyszerű, ember által is követhető következtetési algoritmusokat
  - Az LP máig legelterjedtebb megvalósítása a **Prolog = Programozás logikában (Programming in logic)**
    - az elsőrendű logika egy erősen megszorított résznyelve az ún. **definit-** vagy **Horn-klózik** nyelve,
    - végrehajtási mechanizmusa: **mintaillesztés**es eljáráshíváson alapuló **visszalépés**es keresés.

## Az előadás LP részének áttekintése

---

- **1. blokk:** A Prolog nyelv alapjai
  - Logikai háttér
  - Szintaxis
  - Végrehajtási mechanizmus
- **2. blokk:** Prolog programozási módszerek
  - A legfontosabb beépített eljárások
  - Fejlettebb nyelvi és rendszerelemek
- Kitekintés: Új irányzatok a logikai programozásban

## A Prolog/LP rövid történeti áttekintése

---

1960-as évek	Első tételbizonyító programok
1970-72	A logikai programozás elméleti alapjai (R A Kowalski)
1972	Az első Prolog interpreter (A Colmerauer)
1975	A második Prolog interpreter (Szeredi P)
1977	Az első Prolog fordítóprogram (D H D Warren)
1977–79	Számos kísérleti Prolog alkalmazás Magyarországon
1981	A japán 5. generációs projekt a logikai programozást választja
1982	A magyar MProlog az egyik első kereskedelmi forgalomba kerülő Prolog megvalósítás
1983	Egy új fordítási modell és absztrakt Prolog gép (WAM) megjelenése (D H D Warren)
1986	Prolog szabványosítás kezdete
1987–89	Új logikai programozási nyelvek (CLP, Gödel stb.)
1990–...	Prolog megjelenése párhuzamos számítógépeken Nagyhatékonyságú Prolog fordítóprogramok .....

## Információk a logikai programozásról

---

- A legfontosabb Prolog megvalósítások:

- SWI Prolog: <http://www.swi-prolog.org/>
- SICStus Prolog: <http://www.sics.se/sicstus>
- GNU Prolog: <http://pauillac.inria.fr/~diaz/gnu-prolog/>

- Hálózati információforrások:

- The WWW Virtual Library: Logic Programming:  
<http://www.afm.sbu.ac.uk/logic-prog>
- CMU Prolog Repository:  
(a <http://www.cs.cmu.edu/afs/cs/project/ai-repository/ai/lang/prolog/> címen belül)
  - Főlap: [0.html](#)
  - Prolog FAQ: [faq/prolog.faq](#)
  - Prolog Resource Guide: [faq/prg\\_1.faq](#), [faq/prg\\_2.faq](#)

## Magyar nyelvű Prolog irodalom

---

**Farkas Zsuzsa, Futó Iván, Langer Tamás, Szeredi Péter:**

Az MProlog programozási nyelv.

Műszaki Könyvkiadó, 1989

*jó bevezetés, sajnos az MProlog beépített eljárásai nem szabványosak.*

**Márkus Zsuzsa:** Prologban programozni könnyű.

Novotrade, 1988

*mint fent*

**Futó Iván (szerk.):** Mesterséges intelligencia. (9.2 fejezet, Szeredi Péter)

Aula Kiadó, 1999

*csak egy rövid fejezet a Prologról*

**Peter Flach:** Logikai Programozás. Az intelligens következtetés példákon keresztül.

Panem — John Wiley & Sons, 2001

*jó áttekintés, inkább elméleti érdeklődésű olvasók számára*

## English Textbooks on Prolog

---

- Logic, Programming and Prolog, 2nd Ed., by Ulf Nilsson and Jan Maluszynski, Previously published by John Wiley & Sons Ltd. (1995)  
Downloadable as a pdf file from <http://www.ida.liu.se/~ulfni/lpp>
- Prolog Programming for Artificial Intelligence, 3rd Ed., Ivan Bratko, Longman, Paperback - March 2000
- The Art of PROLOG: Advanced Programming Techniques, Leon Sterling, Ehud Shapiro, The MIT Press, Paperback - April 1994
- Programming in PROLOG: Using the ISO Standard, C.S. Mellish, W.F. Clocksin, Springer-Verlag Berlin, Paperback - July 2003



# PROLOG: EGY KIS GYAKORLATI BEMUTATÁS

## Példák

---

- adatbázis jellegű: családi kapcsolatok
- aritmetika: faktoriális
- adatstruktúrák: bináris fák

## A családi kapcsolatok példája

---

### • Adatok

Adottak gyerek–szülő kapcsolatra vonatkozó állítások, pl.

gyerek	szülő
Imre	István
Imre	Gizella
István	Géza
István	Sarolta
Gizella	Civakodó Henrik
Gizella	Burgundi Gizella

### • Feladatok:

- Definiálандó az unoka–nagyözülő kapcsolat, pl. keressük egy adott személy nagyözüleit.
- Definiálандó az leszármazott–ős kapcsolat, pl. keressük egy adott személy őseit.

## A nagyszülő feladat — C nyelvű megoldás

---

```
/* Az adatbázis */
struct gysz {
    char *gyerek, *szulo;
} szulok[] = {
    "Imre",    "István",
    "Imre",    "Gizella",
    "István",  "Géza",
    "István",  "Sarolt",
    "Gizella", "Civakodó Henrik",
    "Gizella", "Burgundi Gizella",
    NULL,     NULL
};
```

```
/* unoka nagyszüleinek kiiratása */
void nagyszuloi(char *unoka)
{
    struct gysz *mgysz = szulok;
    for (; mgysz->gyerek; ++mgysz)
        if(!strcmp(unoka, mgysz->gyerek))
        { struct gysz *mszn = szulok;
          for (; mszn->gyerek; ++mszn)
              if (!strcmp(mgysz->szulo,
                          mszn->gyerek))
                  puts(mszn->szulo);
        }
}
```

## A nagyszülő feladat — Prolog megoldás

---

```
% szuloje(Gy, Sz):Gy szülője Sz.
szuloje('Imre', 'István').
szuloje('Imre', 'Gizella').
szuloje('István', 'Géza').
szuloje('István', 'Sarolt').
szuloje('Gizella',
        'Civakodó Henrik').
szuloje('Gizella',
        'Burgundi Gizella').

% Gyerek nagyszülője Nagyszulo.
nagyszuloje(Gyerek, Nagyszulo) :-
    szuloje(Gyerek, Szulo),
    szuloje(Szulo, Nagyszulo).
```

```
% Kik Imre nagyszülei?
| ?- nagyszuloje('Imre', NSz).
NSz = 'Géza' ? ;
NSz = 'Sarolt' ? ;
NSz = 'Civakodó Henrik' ? ;
NSz = 'Burgundi Gizella' ? ;
no

% Kik Géza unokái?
| ?- nagyszuloje(U, 'Géza').
U = 'Imre' ? ;
no
```

## A Prolog és az adatbáziskezelés

---

- Miben különbözik a Prolog egy adatbáziskezelőtől
- Mivel több?
  - rekurzió
  - összetett adatszerkezetek
- De: a Prolog egy programozási nyelv
  - pl. nem optimalizálja a részkérdések sorrendjét

## Az őse – rekurzív – kapcsolat

---

- Egy egyszerű megoldás

- A szülő ős.
- A szülő őse is ős.

```
% ose(Lesz, Os): A Lesz leszármazott őse Os
ose(Lesz, Os) :- szuloje(Lesz, Os).
ose(Lesz, Os) :- szuloje(Lesz, Sz),
                  ose(Sz, Os).
```

- Egy alternatív megoldás, diszjunkció használatával

- Tekintsük sorra a szülőket.
  - Az adott szülő ős.
  - Az adott szülő őse is ős.

```
ose1(Lesz, Os) :-
    szuloje(Lesz, Sz),
    (   Os = Sz
    ;   ose1(Sz, Os)
    ).
```

## A Prolog végrehajtási mechanizmusa dióhéjban

---

- A Prolog eljárásos szemléletben
  - Egy eljárás: azon klózek összesége, amelyek fejének neve és argumentumszáma azonos.
  - Egy klóz:  $\text{Fej} :- \text{Törzs}$ , ahol Törzs egy célsorozat
  - Egy célsorozat:  $C_1, \dots, C_n$ , célok (eljáráshívások) sorozata,  $n \geq 0$
- Végrehajtás: adott egy program és egy futtatandó célsorozat
  - Redukciós lépés:
    - a célsorozat *első* tagjához keresünk egy vele *egyesíthető* klózfejet,
    - az egyesítéshez szükséges *változó-behelyettesítéseket* elvégezzük,
    - az első célt helyettesítjük az adott klóz törzsével
  - Egyesítés: két Prolog kifejezés azonos alakra hozása változók behelyettesítésével, a lehető legáltalánosabb módon
  - Keresés:
    - a redukciós lépésben a klózeket a felírás sorrendjében (felülről lefele) nézzük végig,
    - ha egy cél több klózfejjel is egyesíthető, akkor a Prolog *minden* lehetséges redukciós lépést megpróbál (meghiúsulás, visszalépés esetén)



## Aritmetika Prologban – faktoriális

---

```
% fakt(N, F): F = N!.  
fakt(0, 1).  
fakt(N, F) :-  
    N > 0,  
    N1 is N-1,  
    fakt(N1, F1),  
    F is F1*N.
```

## Néhány beépített predikátum

- Kifejezések egyesítése:  $x = y$ : az  $x$  és  $y$  **szimbolikus** kifejezések változók behelyettesítésével azonos alakra hozhatók (és el is végzi a behelyettesítéseket).
- Kifejezések nem-egyesíthetősége:  $x \neq y$ : az  $x$  és  $y$  kifejezések nem egyesíthetőek.
- Aritmetikai predikátumok
  - $x \text{ is } Kif$ :  $A \text{ Kif}$  **aritmetikai** kifejezést kiértékeli és **értékét** egyesíti  $x$ -szel.
  - $Kif1 < Kif2$ ,  $Kif1 = < Kif2$ ,  $Kif1 > Kif2$ ,  $Kif1 \geq Kif2$ ,  $Kif1 =:= Kif2$ ,  $Kif1 \neq Kif2$ :  
 $A \text{ Kif1}$  és  $Kif2$  aritmetikai kifejezések értéke a megadott relációban van egymással  
 $(=:=)$  jelentése: aritmetikai egyenlőség,  $(\neq)$  jelentése aritmetikai nem-egyenlőség).
  - Ha  $Kif$ ,  $Kif1$ ,  $Kif2$  valamelyike nem **tömör** (változómentes) aritmetikai kifejezés  $\Rightarrow$  hiba.
  - Legfontosabb aritmetikai operátorok:  $+$ ,  $-$ ,  $*$ ,  $/$ ,  $\text{rem}$ ,  $//$  (egész-osztás)
- Kiíró predikátumok
  - $\text{write}(X)$ : Az  $x$  Prolog kifejezést kiírja.
  - $\text{nl}$ : Kiír egy újsort.
- Egyéb predikátumok
  - $\text{true}$ ,  $\text{fail}$ : Mindig sikerül ill. mindig megghiúsul.
  - $\text{trace}$ ,  $\text{notrace}$ :  $A$  (teljes) nyomkövetést be- ill. kikapcsolja.

## Programfejlesztési beépített predikátumok

---

- `consult(File)` vagy `[File]`: A `File` állományban levő programot beolvassa és értelmezendő alakban eltárolja. (`File = user`  $\Rightarrow$  terminálról olvas.)
- `listing` vagy `listing(Predikátum)`: Az értelmezendő alakban eltárolt összes ill. adott nevű predikátumokat kilistázza.
- `compile(File)`: A `File` állományban levő programot beolvassa, lefordítja.
- A lefordított alak gyorsabb, de nem listázható, **kicsit** kevésbé pontosan nyomkövethető.
- `halt`: A Prolog rendszer befejezi működését.

```
> sicstus
SICStus 3.12.7 (x86-linux-glibc2.3): Fri Oct 6 00:10:34 CEST 2006
| ?- consult(szulok).
% consulted /home/user/szulok.pl in module user, 0 msec 376 bytes
yes
| ?- nagyszuloje('Imre', 'Istvan').
no
| ?- listing(nagyszuloje).
(...)
yes
| ?- halt.
>
```

## Adatstruktúrák Prologban — példa

---

- A bináris fa adatstruktúra
  - vagy egy csomópont (`node`), amelynek két részfája van mutat (`left`, `right`)
  - vagy egy levél (`leaf`), amely egy egészt tartalmaz
- Binárisfa-struktúrák különböző nyelveken

```
% Struktúra deklarációk C-ben
enum treetype Node, Leaf;
struct tree {
    enum treetype type;
    union {
        struct { struct tree *left;
                  struct tree *right;
                } node;
        struct { int value;
                } leaf;
    } u;
};
```

```
% Adattípus-leírás Prologban
% (ún. Mercury jelölés):

% :- type tree --->
%         node(tree, tree)
%         | leaf(int).
```

## Bináris fák összegzése

---

- Egy bináris fa levélösszegének kiszámítása:
  - csomópont esetén a két részfa levélösszegének összege
  - levél esetén a levélben tárolt egész

```
% C nyelvű (deklaratív) függvény
int tree_sum(struct tree *tree)
{
    switch(tree->type) {
        case Leaf:
            return tree->u.leaf.value;
        case Node:
            return
                tree_sum(tree->u.node.left) +
                tree_sum(tree->u.node.right);
    }
}
```

```
% Prolog eljárás (predikátum)
tree_sum(leaf(Value), Value).
tree_sum(node(Left,Right), S) :-
    tree_sum(Left, S1),
    tree_sum(Right, S2),
    S is S1+S2.
```

## Bináris fák összegzése

---

### ● Prolog példafutás

```
% sicstus -f
SICStus 3.10.0 (x86-linux-glibc2.1): Tue Dec 17 15:12:52 CET 2002
Licensed to BUTE DP course
| ?- consult(tree).
% consulting /home/szeredi/peldak/tree.pl...
% consulted /home/szeredi/peldak/tree.pl in module user, 0 msec 704 bytes
yes
| ?- tree_sum(node(leaf(5),
                    node(leaf(3), leaf(2))), Sum).

Sum = 10 ? ;
no
| ?- tree_sum(Tree, 10).
Tree = leaf(10) ? ;
! Instantiation error in argument 2 of is/2
! goal: 10 is _73+_74
| ?- halt.
%
```

### ● A hiba oka: a beépített aritmetika egyirányú: a 10 is S1+S2 hívás hibát jelez!

## Peano aritmetika — összeadás

---

- A természetes számok halmazán az összeadást definiálhatjuk a Peano axiómákkal ha a számokat az  $s(X)$  „rákövetkező” függvény segítségével ábrázoljuk:

$1 = s(0)$ ,  $2 = s(s(0))$ ,  $3 = s(s(s(0)))$ , ... (Peano ábrázolás).

```
% plus(X, Y, Z): X és Y összege Z (X, Y, Z Peano ábrázolású).
plus(0, X, X).                % 0+X = X.
plus(s(X), Y, s(Z)) :-
    plus(X, Y, Z).            % s(X)+Y = s(X+Y).
```

- A `plus` predikátum több irányban is használható:

```
| ?- plus(s(0), s(s(0)), Z).      Z = s(s(s(0))) ? ; no      % 1+2 = 3

| ?- plus(s(0), Y, s(s(s(0))))).  Y = s(s(0)) ? ; no        % 3-1 = 2

| ?- plus(X, Y, s(s(0))).         X = 0, Y = s(s(0)) ? ;    % 2 = 0+2
                                X = s(0), Y = s(0) ? ;      % 2 = 1+1
                                X = s(s(0)), Y = 0 ? ;      % 2 = 2+0
                                no

| ?-
```

## Adott összegű fák építése

---

### ● Adott összegű fát építő eljárás Peano aritmetikával:

```
tree_sum(leaf(Value), Value).
tree_sum(node(Left, Right), S) :-
    plus(S1, S2, S),
    S1 \= 0, S2 \= 0,          % X \= Y beépített eljárás, jelentése:
                                % X és Y nem egyesíthető
                                % A 0-t kizárjuk, mert különben  $\infty$  sok megoldás van.
    tree_sum(Left, S1),
    tree_sum(Right, S2).
```

### ● Az eljárás futása:

```
| ?- tree_sum(Tree, s(s(s(0)))).
Tree = leaf(s(s(s(0)))) ? ;          % 3
Tree = node(leaf(s(0)),leaf(s(s(0)))) ? ;          % (1+2)
Tree = node(leaf(s(0)),node(leaf(s(0)),leaf(s(0)))) ? ;          % (1+(1+1))
Tree = node(leaf(s(s(0))),leaf(s(0))) ? ;          % (2+1)
Tree = node(node(leaf(s(0)),leaf(s(0))),leaf(s(0))) ? ;          % ((1+1)+1)
no
```



## A Prolog adatfogalma, a Prolog kifejezés

---

- konstans (*atomic*)
  - számkonstans (*number*) — egész vagy lebegőpontos, pl. 1, -2.3, 3.0e10
  - névkonstans (*atom*), pl. 'István', szuloje, +, -, <, sum\_tree
- összetett- vagy struktúra-kifejezés (*compound*)
  - ún. kanonikus alak:  $\langle \text{struktúranév} \rangle (\langle \text{arg}_1 \rangle, \dots)$ 
    - a  $\langle \text{struktúranév} \rangle$  egy névkonstans, az  $\langle \text{arg}_i \rangle$  argumentumok tetszőleges Prolog kifejezések
    - példák: leaf(1), person(william,smith,2003,1,22), <(X,Y), is(X, +(Y,1))
  - szintaktikus „édesítőszerek”, pl. operátorok:  $X \text{ is } Y+1 \equiv \text{is}(X, +(Y,1))$
- változó (*var*)
  - pl. X, Szulo, X2, \_valt, \_, \_123
  - a változó alaphelyzetben behelyettesítetlen, értékkel nem bír, az egyesítés (mintaillesztés) művelete során egy tetszőleges Prolog kifejezést vehet fel értékül (akár egy másik változót)

# A PROLOG NYELV KÖZELÍTŐ SZINTAXISA

## Predikátumok, klózek

---

### ● Példa:

```
% két klózból álló predikátum definíciója, funktora: tree_sum/2
tree_sum(leaf(Val), Val).                %                1. klóz, tényállítás
tree_sum(node(Left,Right), S) :-         %                fej   \
    tree_sum(Left, S1),                  % cél           \   |
    tree_sum(Right, S2),                 % cél           | törzs | 2. klóz, szabály
    S is S1+S2.                         % cél           /   /
```

### ● Szintaxis:

```
⟨ Prolog program ⟩ ::= ⟨ predikátum ⟩ ...
⟨ predikátum ⟩      ::= ⟨ klóz ⟩ ...           { azonos funktorú }
⟨ klóz ⟩            ::= ⟨ tényállítás ⟩.⊥ |
                        ⟨ szabály ⟩.⊥          { klóz funktora = fej funktora }

⟨ tényállítás ⟩    ::= ⟨ fej ⟩
⟨ szabály ⟩        ::= ⟨ fej ⟩ :- ⟨ törzs ⟩
⟨ törzs ⟩          ::= ⟨ cél ⟩, ...
⟨ cél ⟩            ::= ⟨ kifejezés ⟩
⟨ fej ⟩            ::= ⟨ kifejezés ⟩
```

## Prolog programok formázása

---

- Programok javasolt formázása:

- Az egy predikátumhoz tartozó klózok legyenek egymás mellett a programban, közéjük ne tegyünk üres sort. A predikátumokat válasszuk el üres sorokkal.
- A klózfejet írjuk sor elejére, minden célt lehetőleg külön sorba, néhány szóközzel beljebb kezdve

## Prolog kifejezések

---

### ● Példa — egy klózfej mint kifejezés:

```
%      tree_sum(node(Left,Right), S)      % összetett kif., funktora tree_sum/2
%
%      -----
%      |           |           |
% struktúranév      |           argumentum, változó
%                  \- argumentum, összetett kif.
```

### ● Szintaxis:

⟨ kifejezés ⟩	::=	⟨ változó ⟩	{Nincs funktora}
		⟨ konstans ⟩	{Funktora: ⟨ konstans ⟩/0}
		⟨ összetett kifejezés ⟩	{Funktora: ⟨ struktúranév ⟩/⟨ arg.szám ⟩}
		⟨ egyéb kifejezés ⟩	{Operátoros, lista, zárójeles, ld. később}
⟨ konstans ⟩	::=	⟨ névkonstans ⟩	
		⟨ számkonstans ⟩	
⟨ számkonstans ⟩	::=	⟨ egész szám ⟩	
		⟨ lebegőpontos szám ⟩	
⟨ összetett kifejezés ⟩	::=	⟨ struktúranév ⟩ ( ⟨ argumentum ⟩, ... )	
⟨ struktúranév ⟩	::=	⟨ névkonstans ⟩	
⟨ argumentum ⟩	::=	⟨ kifejezés ⟩	

## Lexikai elemek

---

### ● Példák:

```
% változó:      Fakt FAKT _fakt X2 _2 _
% névkonstans:  fakt ≡ 'fakt' 'István' [] ; ', ' += ** \= ≡ '\\\='
% számkonstans: 0 -123 10.0 -12.1e8
% nem névkonstans: !=, Istvan
% nem számkonstans: 1e8 1.e2
```

### ● Szintaxis:

⟨ változó ⟩	::=	⟨ nagybetű ⟩⟨ alfanumerikus jel ⟩...  _⟨ alfanumerikus jel ⟩...
⟨ névkonstans ⟩	::=	'⟨ idézett karakter kar ⟩...'   ⟨ kisbetű ⟩⟨ alfanumerikus jel ⟩...  ⟨ tapadó jel ⟩...  !   ;   [ ]   { }
⟨ egész szám ⟩	::=	{előjeles vagy előjeltelen számjegysorozat}
⟨ lebegőpontos szám ⟩	::=	{belsejében tizedespontot tartalmazó számjegysorozat esetleges exponenssel}
⟨ idézett karakter ⟩	::=	{tetszőleges nem ' és nem \ karakter}   \⟨ escape szekvencia ⟩
⟨ alfanumerikus jel ⟩	::=	⟨ kisbetű ⟩   ⟨ nagybetű ⟩   ⟨ számjegy ⟩   _
⟨ tapadó jel ⟩	::=	+   -   *   /   \   \$   ^   <   >   =   '   ~   :   .   ?   @   #   &

LISTA, MINT SZINTAKTIKUS „ÉDESÍTŐSZER”

---

## A Prolog lista-fogalma

---

### ● A Prolog lista

- Az üres lista a `[]` névkonstans. A nem-üres lista `'.'(Fej, Farok)` struktúra ahol
  - Fej a lista feje (első eleme), míg
  - Farok a lista farka, azaz a fennmaradó elemekből álló lista.
- A listák írhatók egyszerűsített alakban („szintaktikus édesítés”).
- Megvalósításuk optimalizált, időben és helyben is hatékonyabb, mint a „közönséges” struktúráké.

### ● Példa

```
számlista(.(E,L)) :-
    number(E), számlista(L).
számlista([]).



| ?- listing(számlista).
számlista([A|B]) :-
    number(A),
    számlista(B).
számlista([]).

| ?- számlista([1,2]).      % [1,2] == .(1,.(2,[])) == [1|[2|[]]]
    yes
| ?- számlista([1,a,f(2)]).
    no
```



## Listák írásmódjai

- Egy  $N$  elemű lista lehetséges írásmódjai:
  - alapstruktúra-alak:  $\cdot(\text{Elem}_1, \cdot(\text{Elem}_2, \dots, \cdot(\text{Elem}_N, []) \dots))$
  - ekvivalens lista-alak:  $[\text{Elem}_1, \text{Elem}_2, \dots, \text{Elem}_N]$
  - kevésbe kényelmes ekvivalens alak:  $[\text{Elem}_1 | [\text{Elem}_2 | \dots | [\text{Elem}_N | [] ] \dots]]$
- A listák fastruktúra alakja és megvalósítása



## Listák jelölése — szintaktikus édesítőszer

---

- az alapvető édesítés:  $[Fej|Farok] \equiv .(Fej, Farok)$
- $N$ -szeri alkalmazás kevesebb zárójellel:  
 $[Elem_1, Elem_2, \dots, Elem_N|Farok] \equiv [Elem_1|[Elem_2|\dots|[Elem_N|Farok]\dots]]$
- Ha a farok  $[]$ :  $[Elem_1, Elem_2, \dots, Elem_N] \equiv [Elem_1, Elem_2, \dots, Elem_N|[]]$

?- $[1,2] = [X Y]$ .	$\Rightarrow X = 1, Y = [2] ?$
?- $[1,2] = [X,Y]$ .	$\Rightarrow X = 1, Y = 2 ?$
?- $[1,2,3] = [X Y]$ .	$\Rightarrow X = 1, Y = [2,3] ?$
?- $[1,2,3] = [X,Y]$ .	$\Rightarrow \text{no}$
?- $[1,2,3,4] = [X,Y Z]$ .	$\Rightarrow X = 1, Y = 2, Z = [3,4] ?$
?- $L = [1 _], L = [_ ,2 _]$ .	$\Rightarrow L = [1,2 _A] ?$ % nyílt végű
?- $L = .(1, [2,3 []])$ .	$\Rightarrow L = [1,2,3] ?$
?- $L = [1,2 . (3, [])]$ .	$\Rightarrow L = [1,2,3] ?$
?- $[X [3-Y/X Y]] = .(A, [A-B,6])$ .	$\Rightarrow A=3, B=[6]/3, X=3, Y=[6] ?$

## Tömör és minta-kifejezések, lista-minták, nyílt végű listák

- (Ismétlés:) Tömör (ground) kifejezés: változót nem tartalmazó kifejezés
- Minta: egy általában nem nem tömör kifejezés, mindazon kifejezéseket „képviseli”, amelyek belőle változó-behelyettesítéssel előállnak.
- Lista-minta: listát (is) képviselő minta.
- Nyílt végű lista: olyan lista-minta, amely bármilyen hosszú listát is képvisel.
- Zárt végű lista: olyan lista(-minta), amely egyféle hosszú listát képvisel.

Zárt végű	Milyen listákat képvisel	Nyílt végű	Milyen listákat képvisel
$[X]$	egyelemű	$X$	tetszőleges
$[X, Y]$	kételemű	$[X   Y]$	nem üres (legalább 1 elemű)
$[X, X]$	két egyforma elemből álló	$[X, Y   Z]$	legalább 2 elemű
$[X, 1, Y]$	3 elemből áll, 2. eleme 1	$[a, b   Z]$	legalább 2 elemű, elemei: $a, b, \dots$

## A logikai változó

---

- A logikai változó fogalma:
  - kifejezésként, kifejezésben egyaránt előfordulhat, vö. a változókat a (lista) mintákban.
  - két változó azonossá tehető (azaz egyesíthető): pl. két azonos változó egy kifejezésben.
  - a változó „teljes jogú” állampolgár a (rész)kifejezések világában
- Erlang-ban is van mintaillesztés, de a minta csak szétszedésre használható, összerakásra nem; a mintabeli változók mindig (tömör) értéket kapnak.
- (Egyes újabb funkcionális nyelvek, pl. az Oz nyelv, támogatják a logikai változókat.)
- Példa: Az alábbi célsorozat egy két **azonos** elemből álló listát épít fel az L változóban. Az elemek értéke **azonos** lesz a célsorozatbeli x változóval:

```
első_eleme([E|_], E).
második_eleme([_,E|_], E).
```

```
| ?- első_eleme(L, X), második_eleme(L, X). => L = [X,X|_A] ? ; no
```

- Ha az egyesített változók bármelyike értéket kap, a többi is erre az értékre helyettesítődik:

```
| ?- első_eleme(L, X), második_eleme(L, X), X = alma.
    => X = alma, L = [alma,alma|_A] ? ; no
| ?- első_eleme(L, X), második_eleme(L, X), második_eleme(L, bor)
    => X = bor, L = [bor,bor|_A] ? ; no
```

## Listák összefűzése: az append/3 eljárás

- `append(L1, L2, L3)`: Az L3 lista az L1 és L2 listák elemeinek egymás után fűzésével áll elő (jelöljük:  $L3 = L1 \oplus L2$ ) — két megoldás:

```
append0([], L2, L) :- L = L2.
append0([X|L1], L2, L) :-
    append0(L1, L2, L3), L = [X|L3].
```

```
> append0([1,2,3],[4],A)
(2) > append0([2,3],[4],B), A=[1|B]
(2) > append0([3],[4],C), B=[2|C], A=[1|B]
(2) > append0([], [4],D),C=[3|D],B=[2|C],A=[1|B]
(1) > D=[4], C=[3|D], B=[2|C], A=[1|B]
BIP > C=[3,4], B=[2|C], A=[1|B]
BIP > B=[2,3,4], A=[1|B]
BIP > A=[1,2,3,4]
BIP > []
L = [1,2,3,4] ?
```

```
append([], L, L).
append([X|L1], L2, [X|L3]) :-
    append(L1, L2, L3).
```

```
> append([1,2,3],[4],A), write(A)
(2) > append([2,3],[4],B), write([1|B])
(2) > append([3],[4],C), write([1,2|C])
(2) > append([], [4],D), write([1,2,3|D])
(1) > write([1,2,3,4])
[1,2,3,4]
BIP > []
L = [1,2,3,4] ?
```

- Az `append0/append(L1, ...)` komplexitása: futási ideje arányos L1 hosszával.
- Miért jobb az `append/3` mint az `append0/3`?

- `append/3` **jobbrekurzív**, ciklussal ekvivalens (nem fogyaszt vermet)
- `append([1,...,1000],[0],[2,...])` azonnal, `append0(...)` 1000 lépésben hiúsul meg
- `append/3` használható szétszedésre is (lásd később), míg `append0/3` nem.

## Lista építése *előlről* — nyílt végű listákkal

- Az append eljárás már az első redukciónál felépíti az eredmény fejét!  
(az eredményparaméter egy lista-minta lesz, a farok még ismeretlen, vö. logikai változó)

```
append([], L, L).
append([X|L1], L2, [X|L3]) :-      append(L1, L2, L3).
| ?- append([1,2,3], [4], Ered) ==> Ered = [1|A], append([2,3], [4], A)
```

- Haladó nyomkövetési lehetőségek ennek demonstrálására
  - library(debugger\_examples) — példák a nyomkövető programozására, új parancsokra
  - új parancs: 'N <név>' — fókuszált argumentum elnevezése
  - szabványos parancs: '^ <argszám>' — adott argumentumra fókuszálás
  - új parancs: 'P [<név>]' — adott nevű (ill összes) kifejezés kiírása

```
| ?- use_module(library(debugger_examples)).
| ?- trace, append([1,2,3], [4,5,6], A).
    1      1 Call: append([1,2,3], [4,5,6], _543) ? ^ 3
    1      1 Call: ^3 _543 ? N Ered
    1      1 Call: ^3 _543 ? P                      => Ered = _543
    2      2 Call: append([2,3], [4,5,6], _2700) ? P => Ered = [1|_2700]
    3      3 Call: append([3], [4,5,6], _3625) ? P   => Ered = [1,2|_3625]
    4      4 Call: append([], [4,5,6], _4550) ? P    => Ered = [1,2,3|_4550]
    4      4 Exit: append([], [4,5,6], [4,5,6]) ? P => Ered = [1,2,3,4,5,6]
    3      3 Exit: append([3], [4,5,6], [3,4,5,6]) ?
    2      2 Exit: append([2,3], [4,5,6], [2,3,4,5,6]) ?
    1      1 Exit: append([1,2,3], [4,5,6], [1,2,3,4,5,6]) ?
=> A = [1,2,3,4,5,6] ? ; no
```

## Listák megfordítása

---

- Naív (négyzetes lépésszámú) megoldás

```
% nrev(L, R): Az R lista az L megfordítása.  
nrev([], []).  
nrev([X|L], R) :-  
    nrev(L, RL),  
    append(RL, [X], R).
```

- Lineáris lépésszámú megoldás

```
% reverse(R, L): Az R lista az L megfordítása.  
reverse(R, L) :- revapp(L, [], R).
```

```
% revapp(L1, L2, R): L1 megfordítását L2 elé fűzve kapjuk R-t.  
revapp([], R, R).  
revapp([X|L1], L2, R) :-  
    revapp(L1, [X|L2], R).
```

- A `lists` könyvtár tartalmazza az `append/3` és `reverse/2` eljárások definícióját.

- A könyvtár betöltése:

```
:- use_module(library(lists)).
```

## append és revapp — listák gyűjtési iránya

---

### ● Prolog megvalósítás

```
append([], L, L).
append([X|L1], L2, [X/L3]) :-
    append(L1, L2, L3).
```

```
revapp([], L, L).
revapp([X|L1], L2, L3) :-
    revapp(L1, [X/L2], L3).
```

### ● C++ megvalósítás

```
struct link { link *next;
              char elem;
              link(char e): elem(e) {}
            };
typedef link *list;
```

```
list append(list list1, list list2)
{ list list3, *lp = &list3;
  for (list p=list1; p; p=p->next)
  { list newl = new link(p->elem);
    *lp = newl; lp = &newl->next;
  }
  *lp = list2;
  return list3;
}
```

```
list revapp(list list1, list list2)
{ list l = list2;
  for (list p=list1; p; p=p->next)
  { list newl = new link(p->elem);
    newl->next = l; l = newl;
  }


  return l;
}
```



## Listák szétbontása az append/3 segítségével

```
% append(L1, L2, L3):
% Az L3 lista az L1 és L2
% listák elemeinek egymás
% után fűzésével áll elő.
append([], L, L).
append([X|L1], L2, [X|L3]) :-
    append(L1, L2, L3).

| ?- append(A, B, [1,2,3,4]).
A = [], B = [1,2,3,4] ? ;
A = [1], B = [2,3,4] ? ;
A = [1,2], B = [3,4] ? ;
A = [1,2,3], B = [4] ? ;
A = [1,2,3,4], B = [] ? ;
no
```



## Variációk appendre 1. — Három lista összefűzése

---

- Az `append/3` keresési tere **véges**, ha első és harmadik argumentuma közül legalább az egyik zárt végű lista.

- `append(L1, L2, L3, L123) : L1  $\oplus$  L2  $\oplus$  L3 = L123`

```
append(L1, L2, L3, L123) :-
    append(L1, L2, L12), append(L12, L3, L123).
```

- Nem hatékony, pl.: `append([1, ..., 100], [1, 2, 3], [1], L)` 103 helyett 203 lépés!

- Szétszedésre nem alkalmas — végtelen választási pontot hoz létre

- Szétszedésre is alkalmas, hatékony változat

```
% L1  $\oplus$  L2  $\oplus$  L3 = L123, ahol vagy L1 és L2, vagy L123 adott (zárt végű).
append(L1, L2, L3, L123) :-
    append(L1, L23, L123), append(L2, L3, L23).
```

- Az első `append/3` hívás nyílt végű listát állít elő:

```
| ?- append([1,2], L23, L).       $\Rightarrow$       L = [1,2|L23] ?
```

- Az `L3` argumentum behelyettesítettsége (nyílt vagy zárt végű lista-e) nem számít.

## Mintakeresés append/3-mal

---

### ● Párban előforduló elemek

```
% párban(Lista, Elem): A Lista számlistának Elem olyan  
% eleme, amelyet egy ugyanilyen elem követ.  
párban(L, E) :-  
    append(_, [E,E|_], L).  
  
| ?- párban([1,8,8,3,4,4], E).  
    E = 8 ? ; E = 4 ? ; no
```

### ● Dadogó részek

```
% dadogó(L, D): D olyan nem üres részlistája L-nek,  
% amelyet egy vele megegyező részlista követ.  
dadogó(L, D) :-  
    append(_, Farok, L),  
    D = [_|_],  
    append(D, Vég, Farok),  
    append(D, _, Vég).  
  
| ?- dadogó([2,2,1,2,2,1], D).  
    D = [2] ? ; D = [2,2,1] ? ; D = [2] ? ; no
```

## Keresés listában

- member(E, L): E az L lista eleme

```
member(Elem, [Elem|_]).
member(Elem, [_|Farok]) :-
    member(Elem, Farok).
```

```
member(Elem, [Fej|Farok]) :-
    (   Elem = Fej
    ;   member(Elem, Farok)
    ).
```

- A member/2 felhasználási lehetőségei

- Eldöntendő (igen-nem) kérdés:

```
| ?- member(2, [1,2,3]).      ⇒   yes
```

- Lista elemeinek felsorolása:

```
| ?- member(X, [1,2,3]).      ⇒   X = 1 ? ; X = 2 ? ; X = 3 ? ; no
| ?- member(X, [1,2,1]).      ⇒   X = 1 ? ; X = 2 ? ; X = 1 ? ; no
```

- Listák közös elemeinek felsorolása – mindkét fenti hívásmintát használja:

```
| ?- member(X, [1,2,3]),
    member(X, [5,4,3,2,3]).    ⇒   X = 2 ? ; X = 3 ? ; X = 3 ? ; no
```

- Egy értéket egy (nyílt végű) lista elemévé tesz, végtelen választás!

```
| ?- member(1, L).           ⇒   L = [1|_A] ? ; L = [_A,1|_B] ? ;
                                L = [_A,_B,1|_C] ? ; ...
```

- A member/2 keresési tere **véges**, ha második argumentuma zárt végű lista.

## member/2 általánosítása: select/3

---

- `select(Elem, Lista, Marad)`: Elemet a Listából elhagyva marad Marad.

```
select(Elem, [Elem|Marad], Marad).      % Elhagyjuk a fejet, marad a farok.
select(Elem, [X|Farok], [X|Marad0]) :- % Marad a fej,
    select(Elem, Farok, Marad0).      % a farokból hagyunk el elemet.
```

- Felhasználási lehetőségek:

```
| ?- select(1, [2,1,3], L).              % Adott elem elhagyása
    L = [2,3] ? ; no
| ?- select(X, [1,2,3], L).              % Akármelyik elem elhagyása
    L=[2,3], X=1 ? ; L=[1,3], X=2 ? ; L=[1,2], X=3 ? ; no
| ?- select(3, L, [1,2]).                % Adott elem beszúrása!
    L = [3,1,2] ? ; L = [1,3,2] ? ; L = [1,2,3] ? ; no
| ?- select(3, [2|L], [1,2,7,3,2,1,8,9,4]).
                                           % Beszúrható-e 3 az [1,...]-ba
    no                                   % úgy, hogy [2,...]-t kapjunk?
| ?- select(1, [X,2,X,3], L).
    L = [2,1,3], X = 1 ? ; L = [1,2,3], X = 1 ? ; no
```

- A `lists` könyvtár tartalmazza a `member/2` és `select/3` eljárások definícióját is.
- A `select/3` keresési tere **véges**, ha 2. és 3. argumentuma közül legalább az egyik zárt végű.

## Listák permutációja

---

- `permutation(Lista, Perm)`: Lista permutációja a Perm lista.  
(Az alábbi definíció a `library(lists)` könyvtárból származik:)

```
permutation([], []).
permutation(Lista, [Elso|Perm]) :-
    select(Elso, Lista, Maradek),
    permutation(Maradek, Perm).
```

- Felhasználási példák:

```
| ?- permutation([1,2], L).
    L = [1,2] ? ; L = [2,1] ? ; no

| ?- permutation([a,b,c], L).
    L = [a,b,c] ? ; L = [a,c,b] ? ; L = [b,a,c] ? ;
    L = [b,c,a] ? ; L = [c,a,b] ? ; L = [c,b,a] ? ;
    no

| ?- permutation(L, [1,2]).
    L = [1,2] ? ;
    végtelen keresési tér
```

- Ha `permutation/2`-ben az első argumentum ismeretlen, akkor a `select` hívás keresési tere végtelen!

OPERÁTOROK, MINT SZINTAKTIKUS „ÉDESÍTŐSZER”

---

## Operátor-kifejezések

---

### ● Példa:

% S is  $-S1+S2$  ekvivalens az  $is(S, +(-(S1),S2))$  kifejezéssel

### ● Operátoros kifejezések

$\langle \text{összetett kifejezés} \rangle ::=$

$\langle \text{struktúranév} \rangle ( \langle \text{argumentum} \rangle, \dots )$	{eddig csak ez volt}
$\langle \text{argumentum} \rangle \langle \text{operátornév} \rangle \langle \text{argumentum} \rangle$	{infix kifejezés}
$\langle \text{operátornév} \rangle \langle \text{argumentum} \rangle$	{prefix kifejezés}
$\langle \text{argumentum} \rangle \langle \text{operátornév} \rangle$	{posztfix kifejezés}
$\langle \text{operátornév} \rangle ::= \langle \text{struktúranév} \rangle$	{ha operátorként lett definiálva}

### ● Operátor-kezelő beépített predikátumok:

- $op(\text{Prioritás}, \text{Fajta}, \text{OpNév})$  vagy  $op(\text{Prioritás}, \text{Fajta}, [\text{OpNév}_1, \text{OpNév}_2, \dots])$ :
  - Prioritás: 0–1200 közötti egész
  - Fajta: az yfx, xfy, xfx, fy, fx, yf, xf névkonstansok egyike
  - OpNév: tetszőleges névkonstans
  - pozitív prioritás esetén definiálja az operátor(oka)t, 0 prioritás esetén megszünteti azokat.
- $current\_op(\text{Prioritás}, \text{Fajta}, \text{OpNév})$ : felsorolja a definiált operátorokat.



## Szabványos, beépített operátorok

---

### Szabványos operátorok

```

1200 xfx :- -->
1200 fx  :- ?-
1100 xfy ;
1050 xfy ->
1000 xfy ', '
900  fy \+
700  xfx < = \= =..
      := =< == \==
      =\= > >= is
      @< @=< @> @>=
500  yfx + - /\ \/
400  yfx * / // rem
      mod << >>
200  xfx **
200  xfy ^
200  fy  - \

```

### Egyéb beépített operátorok SICStus Prologban

```

1150 fx dynamic multifile
      block meta_predicate
900  fy spy nosp
550 xfy :
500 yfx #
500 fx +

```

## Operátorok jellemzői

- Egy operátort jellemez a fajtája és prioritása
- A fajta meghatározza az operátor-osztályt (írásmódot) és az asszociativitást:

Fajta			Osztály	Értelmezés
bal-asszoc.	jobb-asszoc.	nem-asszoc.		
yfx	xfy	xfx	infix	$X \ f \ Y \equiv f(X, Y)$
	fy	fx	prefix	$f \ X \equiv f(X)$
yf		xf	posztfix	$X \ f \equiv f(X)$

- Több-operátoros kifejezésben a zárójelezést a prioritás és az asszociativitás határozza meg, pl.
  - $a/b+c*d \equiv (a/b)+(c*d)$  mert  $/$  és  $*$  prioritása 400, ami **kisebb** mint a  $+$  prioritása (500) (kisebb prioritás = **erősebb** kötés).
  - $a+b+c \equiv (a+b)+c$  mert a  $+$  operátor fajtája yfx, azaz bal-asszociatív — balra köt, balról jobbra zárójelez (a fajtanévben az y betű mutatja az asszociativitás irányát)
  - $a^b^c \equiv a^(b^c)$  mert a  $^$  operátor fajtája xfy, azaz jobb-asszociatív (jobbra köt, jobbról balra zárójelez)
  - $a=b=c$  szintaktikusan hibás, mert az  $=$  operátor fajtája xfx, azaz nem-asszociatív

## Operátorok: zárójelezés

---

- Induljunk ki egy teljesen zárójelezett, több operátort tartalmazó kifejezésből!
- Egy részkifejezés prioritása a (legkülső) operátorának a prioritása.
- Egy  $op$  prioritású operátor  $ap$  prioritású argumentumát körülvevő zárójelpár elhagyható ha:
  - $ap < op$  pl.  $a+(b*c) \equiv a+b*c$  ( $ap = 400, op = 500$ )
  - $ap = op$ , jobb-asszociatív operátor jobboldali argumentuma esetén, pl.  $a^(b^c) \equiv a^b^c$  ( $ap = 200, op = 200$ )
  - $ap = op$ , bal-asszociatív operátor baloldali argumentuma esetén, pl.  $(1+2)+3 \equiv 1+2+3$ . Kivétel: ha a baloldali argumentum operátora jobb-asszociatív, azaz az előző feltétel alkalmazható.
- Példa a kivétel esetére:
  - `:- op(500, xfy, +^).`  
`| ?- :- write((1 +^ 2) + 3), nl.   ⇒ (1+^2)+3`  
`| ?- :- write(1 +^ (2 + 3)), nl.   ⇒ 1+^2+3`
  - tehát: konfliktus esetén az első operátor asszociativitása „győz”.

## Operátorok — kiegészítő megjegyzések

---

- Azonos nevű, azonos osztályba tartozó operátorok egyidejűleg nem megengedettek.

- Egy program szövegében direktívákkal definiálhatunk operátorokat, pl.

```
:- op(500, xfx, --).           :- op(450, fx, @).
tree_sum(@V, V).              (...)
```

- A „vessző” kettős szerepe

- struktúra-kifejezés argumentumait választja el
- 1000 prioritású xfy operátorként működik pl.:  $(p \text{ :- } a, b, c) = \text{:-}(p, ', '(a, ', '(b, c)))$
- a „pucér” vessző (,) nem névkonstans, de operátorként aposztrofok nélkül is írható.
- struktúra-argumentumban 999-nél nagyobb prioritású kifejezést zárójelezni kell:

```
| ?- write_canonical((a,b,c)).  => ', '(a, ', '(b,c))
| ?- write_canonical(a,b,c).    => ! procedure write_canonical/3 does not exist
```

- Az egyértelmű elemezhetőség érdekében a Prolog szabvány kiköti, hogy

- operandusként előforduló operátort zárójelbe kell tenni, pl.  $\text{Comp} = (>)$
- nem létezhet azonos nevű infix és posztfix operátor.

- Sok Prolog rendszerben nem kötelező betartani ezeket a megszorításokat.

## Operátorok felhasználása

---

### ● Mire jók az operátorok?

- aritmetikai eljárások kényelmes írására, pl.  $X \text{ is } (Y+3) \bmod 4$
- aritmetikai kifejezések szimbolikus feldolgozására (pl. szimbolikus deriválás)
- klózok leírására ( $:-$  és  $,$  is operátor)
- klózok átadhatók meta-eljárásoknak, pl. `asserta( (p(X):-q(X),r(X)) )`
- eljárásfejek, eljáráshívások olvashatóbbá tételére:

`:- op(800, xfx, [nagyszülője, szülője]).`

`Gy nagyszülője N :- Gy szülője Sz, Sz szülője N.`

- adatstruktúrák olvashatóbbá tételére, pl.

`:- op(100, xfx, [.]).`

`sav(kén, h.2-s-o.4).`

### ● Miért rosszak az operátorok?

- egyetlen globális erőforrás, ez nagyobb projektben gondot okozhat.

## Aritmetika Prologban

---

- Az operátorok teszik lehetővé azt is, hogy a matematikában ill. más programozási nyelvekben megszokott módon írassunk le aritmetikai kifejezéseket.
- Az `is` beépített predikátum egy aritmetikai kifejezést vár a jobboldalán (2. argumentumában), azt kiértékeli, és az eredményt egyesíti a baloldali argumentummal
- Az `==` beépített predikátum mindkét oldalán aritmetikai kifejezést vár, azokat kiértékeli, és csak akkor sikerül, ha az értékek megegyeznek.

- Példák:

```
| ?- X = 1+2, write(X), write(' '), write_canonical(X), Y is X.
⇒                1+2                +(1,2)    ⇒ X = 1+2, Y = 3 ? ; no
| ?- X = 4, Y is X/2, Y == 2.    ⇒ X = 4, Y = 2.0 ? ; no
| ?- X = 4, Y is X/2, Y = 2.    ⇒ no
```

- **Fontos:** az aritmetikai operátorokkal (+,-,...) képzett kifejezések **összetett Prolog kifejezést** jelentenek. Csak az aritmetikai beépített predikátumok értékelik ki ezeket!
- A Prolog kifejezések alapvetően szimbolikusak, az aritmetikai kiértékelés a „kivétel”.

## Klasszikus szimbolikus kifejezés-feldolgozás: deriválás

---

- Írjunk olyan Prolog predikátumot, amely számokból és az  $x$  névkonstansból a  $+$ ,  $-$ ,  $*$  műveletekkel képzett kifejezések deriválását elvégzi!

```
% deriv(Kif, D): Kif-nek az x szerinti deriváltja D.
deriv(x, 1).
deriv(C, 0) :-                               number(C).
deriv(U+V, DU+DV) :-                         deriv(U, DU), deriv(V, DV).
deriv(U-V, DU-DV) :-                         deriv(U, DU), deriv(V, DV).
deriv(U*V, DU*V + U*DV) :-                  deriv(U, DU), deriv(V, DV).
```

```
| ?- deriv(x*x+x, D).
    => D = 1*x+x*1+1 ? ; no
```

```
| ?- deriv((x+1)*(x+1), D).
    => D = (1+0)*(x+1)+(x+1)*(1+0) ? ; no
```

```
| ?- deriv(I, 1*x+x*1+1).
    => I = x*x+x ? ; no
```

```
| ?- deriv(I, 0).
    => no
```

## Operátoros példa: polinom behelyettesítési értéke

---

- Formula: számokból és az 'x' névkonstansból '+' és '\*' operátorokkal felépülő kifejezés.
- A feladat: Egy formula értékének kiszámolása egy adott x érték esetén.

```
% erteke(Kif, X, E): A Kif formula értéke E, az x=X behelyettesítéssel.
```

```
erteke(x, X, E) :-
```

```
    E = X.
```

```
erteke(Kif, _, E) :-
```

```
    number(Kif), E = Kif.
```

```
erteke(K1+K2, X, E) :-
```

```
    erteke(K1, X, E1),
```

```
    erteke(K2, X, E2),
```

```
    E is E1+E2.
```

```
erteke(K1*K2, X, E) :-
```

```
    erteke(K1, X, E1),
```

```
    erteke(K2, X, E2),
```

```
    E is E1*E2.
```

```
| ?- erteke((x+1)*x+x+2*(x+x+3), 2, E).
```

```
E = 22 ? ;
```

```
no
```



# TÍPUSOK PROLOGBAN



## Típusok leírása Prologban

---

- Típusleírás: (tömör) Prolog kifejezések egy halmazának megadása

- Alaptípusok leírása: `int`, `float`, `number`, `atom`, `any`

- Új típusok felépítése:

$\{ \text{str}(T_1, \dots, T_n) \}$  jelentése  $\{ \text{str}(e_1, \dots, e_n) \mid e_1 \in T_1, \dots, e_n \in T_n \}, n \geq 0$

Példa:  $\{\text{személy}(\text{atom}, \text{atom}, \text{int})\}$  az olyan `személy/3` funktorú struktúrák halmaza, amelyben az első két argumentum `atom`, a harmadik egész.

- Típusok, mint halmazok úniója képezhető a `\|` operátorral.

$\{\text{személy}(\text{atom}, \text{atom}, \text{int})\} \| \{\text{atom-atom}\} \| \text{atom}$

- Egy típusleírás elnevezhető (kommentben): `:- type tnév == tleírás.`

`:- type t1 == {atom-atom} \| atom.,`

`:- type ember == {ember-atom} \| {semmi}.`

- Megkülönböztetett únió: csupa különböző funktorú összetett típus úniója. Ha  $S_1, \dots, S_n$  mind különböző funktorú, alkalmazható az egyszerűsített (Mercury) jelölés:

`:- type T == { S1 } \| ... \| { Sn }.  $\Rightarrow$  :- type T ---> S1 ; ... ; Sn. Példák:`

`:- type ember ---> ember-atom; semmi.`

`:- type fa ---> leaf(int) ; node(fa,fa).`

## Típusok leírása Prologban — folytatás

---

### ● Paraméteres típusok — példák

```
:- type pair(T1, T2) ---> T1 - T2.      % egy '-' nevű kétarg.-ú struktúra,
                                         % első arg. T1, a második T2 típusú.
:- type tree(T) ---> leaf(T)             % T típusú elemekből álló
    ; node(tree(T), tree(T)).           % bináris fa
:- type assoc_tree(KeyT, ValueT)         % KeyT és ValueT típusú
    == tree(pair(KeyT, ValueT)).        % párokból álló fa
:- type szótár == assoc_tree(szó, szó).
:- type szó == atom.
```

### ● Típusdeklarációk szintaxisa

```
⟨ típusdeklaráció ⟩ ::= ⟨ típuselnevezés ⟩ | ⟨ típuskonstrukció ⟩
⟨ típuselnevezés ⟩  ::= :- type ⟨ típusazonosító ⟩ == ⟨ típusleírás ⟩ .
⟨ típuskonstrukció ⟩ ::= :- type ⟨ típusazonosító ⟩ ---> ⟨ megkülönb. únió ⟩ .
⟨ megkülönb. únió ⟩ ::= ⟨ konstruktor ⟩ ; ...
⟨ konstruktor ⟩    ::= ⟨ névkonstans ⟩ | ⟨ struktúranév ⟩ (⟨ típusleírás ⟩, ...)
⟨ típusleírás ⟩    ::= ⟨ típusazonosító ⟩ | ⟨ típusváltozó ⟩ | { ⟨ konstruktor ⟩ } |
                    ⟨ típusleírás ⟩ \ / ⟨ típusleírás ⟩
⟨ típusazonosító ⟩ ::= ⟨ típusnév ⟩ | ⟨ típusnév ⟩ (⟨ típusváltozó ⟩, ...)
⟨ típusnév ⟩      ::= ⟨ névkonstans ⟩
⟨ típusváltozó ⟩  ::= ⟨ változó ⟩
```

## Predikátumtípus-deklarációk

---

- Predikátumtípus-deklaráció

`:- pred <eljárásnév>(<típusazonosító>, ...)`

- Példa:

`:- pred tree_sum(tree(int), int).`

- Predikátummód-deklaráció (Nem kötelező, több is megadható.)

`:- mode <eljárásnév>(<módazonosító>, ...) ahol <módazonosító> ::= in | out | inout.`

(Mercury-ban az inout módazonosító nem megengedett.)

- Példák:

```
:- mode tree_sum(in, in).    % ellenőrzés
:- mode tree_sum(in, out).  % fa-összeg előállítása
:- mode tree_sum(out,in).   % adott összegű fa építése
```

- Vegyes típus- és móddeklaráció

`:- pred <eljárásnév>(<típusazonosító>::<módazonosító>, ...)`

- Példa:

`:- pred between(int::in, int::in, int::out).`

## Móddeklaráció: a SICStus kézikönyv által használt alak

---

- A SICStus kézikönyv egy másik jelölést használ a bemenő/kimenő argumentumok jelzésére, pl.

`tree_sum(+T, ?Sum) .`

- Mód-jelölő karakterek:

- + bemenő argumentum (behelyettesített)
- - kimenő argumentum (behelyettesítetlen)
- : eljárás-paraméter (meta-eljárásokban)
- ? tetszőleges

# AZ EGYESÍTÉSI ALGORITMUS

## A Prolog alapvető adatkezelő művelete: az egyesítés

---

- Egyesítés (*unification*): két Prolog kifejezés (pl. egy eljáráshívás és egy klózfej) azonos alakra hozása, változók esetleges behelyettesítésével.
- Példák
  - Bemenő paraméterátadás — a fej változóit helyettesíti be:  
hívás: `nagyszuloje('Imre', Nsz)`,  
fej: `nagyszuloje(Gy, N)`,  
behelyettesítés: `Gy = 'Imre', N = Nsz`
  - Kimenő paraméterátadás — a hívás változóit helyettesíti be:  
hívás: `szuloje('Imre', Sz)`,  
fej: `szuloje('Imre', 'István')`,  
behelyettesítés: `Sz = 'István'`
  - Bemenő/kimenő paraméterátadás — a fej és a hívás változóit is behelyettesíti:  
hívás: `sum_tree(leaf(5), Sum)`  
fej: `sum_tree(leaf(V), V)`  
behelyettesítés: `V = 5, Sum = 5`

## Egyesítés: változók behelyettesítése

### ● A behelyettesítés fogalma

- A behelyettesítés egy olyan függvény, amely bizonyos változókhoz kifejezéseket rendel.
  - Példa:  $\sigma = \{X \leftarrow a, Y \leftarrow s(b, B), Z \leftarrow C\}$ . Itt  $Dom(\sigma) = \{X, Y, Z\}$
  - A  $\sigma$  behelyettesítés  $x$ -hez  $a$ -t,  $Y$ -hoz  $s(b, B)$ -t  $Z$ -hez  $C$ -t rendel. Jelölés:  $X\sigma = a$  stb.
- A behelyettesítés-függvény természetes módon kiterjeszthető az összes kifejezésre:
  - $K\sigma$ :  $\sigma$  alkalmazása  $K$  kifejezésre:  $\sigma$  behelyettesítéseit *egyidejűleg* elvégezzük  $K$ -ban.
  - Példa:  $f(g(Z, h), A, Y)\sigma = f(g(C, h), A, s(b, B))$
- A  $\sigma$  és  $\theta$  behelyettesítések kompozíciója ( $\sigma \otimes \theta$ ) — egymás utáni alkalmazásuk
  - A  $\sigma \otimes \theta$  behelyettesítés az  $x \in Dom(\sigma)$  változókhoz az  $(x\sigma)\theta$  kifejezést, a többi  $y \in Dom(\theta) \setminus Dom(\sigma)$  változóhoz  $y\theta$ -t rendel ( $Dom(\sigma \otimes \theta) = Dom(\sigma) \cup Dom(\theta)$ ):
 
$$\sigma \otimes \theta = \{x \leftarrow (x\sigma)\theta \mid x \in Dom(\sigma)\} \cup \{y \leftarrow y\theta \mid y \in Dom(\theta) \setminus Dom(\sigma)\}$$
  - Pl.  $\theta = \{X \leftarrow b, B \leftarrow d\}$  esetén  $\sigma \otimes \theta = \{X \leftarrow a, Y \leftarrow s(b, d), Z \leftarrow C, B \leftarrow d\}$
- Egy  $G$  kifejezés **általánosabb** mint egy  $S$ , ha létezik olyan  $\rho$  behelyettesítés, hogy  $S = G\rho$ 
  - Példa:  $G = f(A, Y)$  általánosabb mint  $S = f(1, s(Z))$ , mert  $\rho = \{A \leftarrow 1, Y \leftarrow s(Z)\}$  esetén  $S = G\rho$ .



## Egyesítés: legáltalánosabb egyesítő

---

- $A$  és  $B$  kifejezések egyesíthetők ha létezik egy olyan  $\sigma$  behelyettesítés, hogy  $A\sigma = B\sigma$ . Ezt az  $A\sigma = B\sigma$  kifejezést  $A$  és  $B$  egyesített alakjának nevezzük.
- Két kifejezésnek általában több egyesített alakja lehet.
  - Példa:  $A = f(X, Y)$  és  $B = f(s(U), U)$  egyesített alakja pl.
    - $K_1 = f(s(a), a)$  a  $\sigma_1 = \{X \leftarrow s(a), Y \leftarrow a, U \leftarrow a\}$  behelyettesítéssel
    - $K_2 = f(s(U), U)$  a  $\sigma_2 = \{X \leftarrow s(U), Y \leftarrow U\}$  behelyettesítéssel
    - $K_3 = f(s(Y), Y)$  a  $\sigma_3 = \{X \leftarrow s(Y), U \leftarrow Y\}$  behelyettesítéssel
- $A$  és  $B$  legáltalánosabb egyesített alakja egy olyan  $C$  kifejezés, amely  $A$  és  $B$  minden egyesített alakjánál általánosabb
  - A fenti példában  $K_2$  és  $K_3$  legáltalánosabb egyesített alakok
- **Tétel:** A legáltalánosabb egyesített alak, változó-átnevezéstől eltekintve egyértelmű.
- $A$  és  $B$  legáltalánosabb egyesítője egy olyan  $\sigma = mgu(A, B)$  behelyettesítés, amelyre  $A\sigma$  és  $B\sigma$  a két kifejezés legáltalánosabb egyesített alakja.
  - A fenti példában  $\sigma_2$  és  $\sigma_3$  legáltalánosabb egyesítő.
- **Tétel:** A legáltalánosabb egyesítő, változó-átnevezéstől eltekintve egyértelmű.

## Az egyesítési algoritmus

---

- Az egyesítési algoritmus
  - bemenete: két Prolog kifejezés:  $A$  és  $B$
  - feladata: a két kifejezés egyesíthetőségének eldöntése
  - eredménye: sikeresség esetén a legáltalánosabb egyesítő ( $mgu(A, B)$ ) előállítása.
- Az egyesítési algoritmus,  $\sigma = mgu(A, B)$  előállítása
  1. Ha  $A$  és  $B$  azonos változók vagy konstansok, akkor  $\sigma = \{\}$  (üres behelyettesítés).
  2. Egyébként, ha  $A$  változó, akkor  $\sigma = \{A \leftarrow B\}$ .
  3. Egyébként, ha  $B$  változó, akkor  $\sigma = \{B \leftarrow A\}$ .
  4. Egyébként, ha  $A$  és  $B$  azonos nevű és argumentumszámú összetett kifejezések és argumentum-listáik  $A_1, \dots, A_N$  ill.  $B_1, \dots, B_N$ , és
    - a.  $A_1$  és  $B_1$  legáltalánosabb egyesítője  $\sigma_1$ ,
    - b.  $A_2\sigma_1$  és  $B_2\sigma_1$  legáltalánosabb egyesítője  $\sigma_2$ ,
    - c.  $A_3\sigma_1\sigma_2$  és  $B_3\sigma_1\sigma_2$  legáltalánosabb egyesítője  $\sigma_3$ ,
    - d. ...akkor  $\sigma = \sigma_1 \otimes \sigma_2 \otimes \sigma_3 \otimes \dots$
  5. Minden más esetben a  $A$  és  $B$  nem egyesíthető.

## Egyesítési példák

---

- $A = \text{sum\_tree}(\text{leaf}(V), V), B = \text{sum\_tree}(\text{leaf}(5), S)$ 
  - (4.)  $A$  és  $B$  neve és argumentumszáma megegyezik
    - (a.)  $\text{mgu}(\text{leaf}(V), \text{leaf}(5))$  (4., majd 2. szerint)  $= \{V \leftarrow 5\} = \sigma_1$
    - (b.)  $\text{mgu}(V\sigma_1, S) = \text{mgu}(5, S)$  (3. szerint)  $= \{S \leftarrow 5\} = \sigma_2$
  - tehát  $\text{mgu}(A, B) = \sigma_1 \otimes \sigma_2 = \{V \leftarrow 5, S \leftarrow 5\}$
- $A = \text{node}(\text{leaf}(X), T), B = \text{node}(T, \text{leaf}(3))$ 
  - (4.)  $A$  és  $B$  neve és argumentumszáma megegyezik
    - (a.)  $\text{mgu}(\text{leaf}(X), T)$  (3. szerint)  $= \{T \leftarrow \text{leaf}(X)\} = \sigma_1$
    - (b.)  $\text{mgu}(T\sigma_1, \text{leaf}(3)) = \text{mgu}(\text{leaf}(X), \text{leaf}(3))$  (4., majd 2. szerint)  $= \{X \leftarrow 3\} = \sigma_2$
  - tehát  $\text{mgu}(A, B) = \sigma_1 \otimes \sigma_2 = \{T \leftarrow \text{leaf}(3), X \leftarrow 3\}$

## Egyesítési példák a gyakorlatban

---

- Az egyesítéssel kapcsolatos beépített eljárások:

- $X = Y$  egyesíti a két argumentumát, megghiúsul, ha ez nem lehetséges.
- $X \backslash= Y$  sikerül, ha két argumentuma nem egyesíthető, egyébként megghiúsul.

- Példák:

```
| ?- 3+(4+5) = Left+Right.
      Left = 3, Right = 4+5 ?
| ?- node(leaf(X), T) = node(T, leaf(3)).
      T = leaf(3), X = 3 ?
| ?- X*Y = 1+2*3.                % mert 1+2*3 ≡ 1+(2*3)
      no
| ?- X*Y = (1+2)*3.
      X = 1+2, Y = 3 ?
| ?- f(X, 3/Y-X, Y) = f(U, B-a, 3).
      B = 3/3, U = a, X = a, Y = 3 ?
| ?- f(f(X), U+2*2) = f(U, f(3)+Z).
      U = f(3), X = 3, Z = 2*2 ?
```

## Az egyesítés kiegészítése: előfordulás-ellenőrzés (*occurs check*)

---

● Kérdés:  $X$  és  $s(X)$  egyesíthető-e?

- A matematikai válasz: *nem*, egy változó nem egyesíthető egy olyan struktúrával, amelyben előfordul (ez az előfordulás-ellenőrzés).
- Az ellenőrzés költséges, ezért alaphelyzetben nem alkalmazzák, így ciklikus kifejezések keletkezhetnek.
- Szabványos eljárásként rendelkezésre áll: `unify_with_occurs_check/2`
- Kiterjesztés (pl. SICStus): az előfordulás-ellenőrzés elhagyása miatt keletkező ciklikus kifejezések tisztességes kezelése.

● Példák:

```
| ?- X = s(1,X).
      X = s(1,s(1,s(1,s(1,s(...)))))) ?
| ?- unify_with_occurs_check(X, s(1,X)).
      no
| ?- X = s(X), Y = s(s(Y)), X = Y.
      X = s(s(s(s(s(...))))), Y = s(s(s(s(s(...)))))) ?
```

# A PROLOG VÉGREHAJTÁSI MECHANIZMUSA



## A Prolog végrehajtás eljárásos modelljei

---

- Az azonos funktorú klózek alkotnak egy eljárást
- Egy eljárás meghívása a hívás és klózfej mintaillesztésével (egyesítésével) történik
- A végrehajtás lépéseinek modellezése:
  - Eljárás-redukciós modell
    - Az alaplépés: egy hívás-sorozat (azaz célsorozat) redukálása egy klóz segítségével (ez a már ismert redukciós lépés).
    - Visszalépés: visszatérünk egy korábbi célsorozathoz, és újabb klózzal próbálkozunk.
    - A modell előnyei: pontosan definiálható, a keresési tér szemléltethető
  - Eljárás-doboz modell
    - Az alapgondolat: egymásba skatulyázott eljárás-dobozok kapuin lépünk be és ki.
    - Egy eljárás-doboz kapui: hívás (belépés), sikeres kilépés, sikertelen kilépés.
    - Visszalépés: új megoldást kérünk egy már lefutott eljárástól (újra kapu).
    - A modell előnyei: közel van a hagyományos rekurzív eljárásmodellhez, a Prolog beépített nyomkövetője is ezen alapul.

## A eljárás-redukciós végrehajtási modell

---

- A redukciós végrehajtási modell alapgondolata
  - A végrehajtás egy állapota: egy célsorozat
  - A végrehajtás kétféle lépésből áll:
    - redukciós lépés: egy célsorozat + klóz  $\rightarrow$  új célsorozat
    - zsákutca esetén visszalépés: visszatérés a legutolsó választási ponthoz
  - Választási pont:
    - létrehozása: olyan redukciós lépés amely nem a legutolsó klózzal illesztett
    - aktiválása: visszalépéskor visszatérünk a választási pont célsorozatához és a **további** klózek között keresünk illeszthetőt  
(Emiatt a választási pontban a célsorozat mellett az illesztett klóz sorszámát is tárolni kell.)
    - az ún. indexelés segít a választási pontok számának csökkentésében
- A redukciós modell keresési fával szemléltethető
  - A végrehajtás során a fa csomópontjait járjuk be mélységi kereséssel
  - A fa gyökerétől egy adott pontig terjedő szakaszon kell a választási pontokat megjegyezni — ez a választási verem (choice point stack)



## A redukciós modell alapeleme: redukciós lépés

---

- Redukciós lépés: egy célsorozat redukálása egy újabb célsorozattá
  - egy programklóz segítségével (az első cél felhasználói eljárást hív):
    - A klózt **lemásoljuk**, minden változót szisztematikusan új változóra cserélve.
    - A célsorozatot szétbontjuk az első hívásra és a maradékra.
    - Az első hívást **egyesítjük** a klózfejjel
    - A szükséges behelyettesítéseket elvégezzük a klóz **törzsén** és a **célsorozat** maradékán is
    - Az új célsorozat: a klóztörzs és utána a maradék célsorozat
    - Ha a hívás és a klózfej nem egyesíthető, akkor a redukciós lépés megghiúsul.
  - egy beépített eljárás segítségével (az első cél beépített eljárást hív):
    - A célsorozatot szétbontjuk az első hívásra és a maradékra.
    - A beépített eljárás hívást végrehajtjuk.
    - Ez lehet sikeres (változó-behelyettesítésekkel), vagy lehet sikertelen.
    - Siker esetén a behelyettesítéseket elvégezzük a célsorozat maradékán.
    - Az új célsorozat: az (első hívás elhagyása után fennmaradó) maradék célsorozat.
    - Ha a beépített eljárás hívása sikertelen, akkor a redukciós lépés megghiúsul.

## A Prolog végrehajtási algoritmus

---

1. *(Kezdeti beállítások:)* A verem üres,  $CS := \text{célsorozat}$
2. *(Beépített eljárások:)* Ha  $CS$  első hívása beépített akkor hajtsuk végre,
  - a. Ha sikertelen  $\Rightarrow$  6. lépés.
  - b. Ha sikeres,  $CS :=$  a redukciós lépés eredménye  $\Rightarrow$  5. lépés.
3. *(Klózszámláló kezdőértékezése:)*  $I = 1$ .
4. *(Redukciós lépés:)* Tekintsük  $CS$  első hívására vonatkoztatható klózok listáját. Ez indexelés nélkül a predikátum összes klóza lesz, indexelés esetén ennek egy megszűrt részsorozata. Tegyük fel, hogy ez a lista  $N$  elemű.
  - a. Ha  $I > N \Rightarrow$  6. lépés.
  - b. Redukciós lépés a lista  $I$ -edik klóza és a  $CS$  célsorozat között.
  - c. Ha sikertelen, akkor  $I := I+1 \Rightarrow$  4. lépés.
  - d. Ha  $I < N$  (nem utolsó), akkor vermeljük  $\langle CS, I \rangle$ -t.
  - e.  $CS :=$  a redukciós lépés eredménye
5. *(Siker:)* Ha  $CS$  üres, akkor sikeres vég, egyébként  $\Rightarrow$  2. lépés.
6. *(Sikertelenség:)* Ha a verem üres, akkor sikertelen vég.
7. *(Visszalépés:)* Ha a verem nem üres, akkor leemeljük a veremből  $\langle CS, I \rangle$ -t,  $I := I+1$ , és  $\Rightarrow$  4. lépés.

## Indexelés (előzetes)

---

- Mi az indexelés?
  - egy hívásra vonatkoztatható (potenciálisan illeszthető) klózik gyors kiválasztása,
  - egy eljárás klózikainak **fordítási idejű** csoportosításával.
- A legtöbb Prolog rendszer, így a SICStus Prolog is, az első fej-argumentum alapján indexel (first argument indexing).
- Az indexelés alapja az első fejargumentum külső funktora:
  - C szám vagy névkonstans esetén C/0;
  - R nevű és N argumentumú struktúra esetén R/N;
  - változó esetén nem értelmezett (minden funktorhoz besoroltatik).
- Az indexelés megvalósítása:
  - Fordítási időben minden funktorhoz elkészítjük az alkalmazható klózik listáját
  - Futáskor lényegében konstans idő alatt elő tudjuk vennie a megfelelő klóziklistát
  - *Fontos:* ha egyelemű a részhalmaz, nem hozunk létre választási pontot!
- Például `szuloje('István', X)` kételemű klóziklistára szűkít, de `szuloje(X, 'István')` mind a 6 klózik megtartja (mert a SICStus Prolog csak az első argumentum szerint indexel)

## Redukciós modell — előnyök és hátrányok

---

### ● Előnyök

- (viszonylag) egyszerű és (viszonylag) precíz definíció
- a keresési tér megjeleníthető, grafikusán szemléltethető

### ● Hátrányok

- az eljárásokból való kilépést elfedi, pl.

p :- q, r.	G0: p ?	
q :- s, t.	G1: q, r ?	
s.	G2: s, t, r ?	
t.	G3: t, r ?	
r.	G4: r ?	$\Leftarrow q\text{-ből való kilépés}$
	G5: [] ?	

- nem jól illeszkedik a Prolog megvalósítások tényleges végrehajtási mechanizmusához
- nem alkalmazható „igazi” Prolog programok nyomkövetésére (hosszú célsorozatok)

### ● Ezért van létjogosultsága egy másik modellnek:

- eljárás-doboz (procedure box) modell
- (szokás még 4-kapus doboz ill. Byrd doboz modellnek is nevezni)
- a Prolog rendszerek nyomkövető szolgáltatása erre a modellre épül

## Az eljárás-doboz modell

---

- A Prolog eljárás-végrehajtás két fázisa
  - előre menő végrehajtás: egymásba skatulyázott eljárás-belépések és - kilépések
  - visszafelé menő végrehajtás: újabb megoldás kérése egy már lefutott eljárástól

- Egy egyszerű példa

$q(2) . \quad q(4) . \quad q(7) .$


$p(X) :- q(X), X > 3.$

- Belépünk a  $p/1$  eljárásba (Hívási kapu, Call port)
- Belépünk a  $q/1$  eljárásba (Call)
- A  $q/1$  eljárás sikeresen lefut a  $q(2)$  eredménnyel (Kilépési kapu, Exit port)
- A  $> /2$  eljárásba belépünk a  $2 > 3$  hívással (Call)
- A  $> /2$  eljárás sikertelenül fut le (Meghiúsulási kapu, Fail port)
- (visszafelé menő futás): visszatérünk (a már lefutott)  $q/1$ -be, újabb megoldást kérve (Újra kapu, Redo Port)
- A  $q/1$  eljárás sikeresen lefut a  $q(4)$  eredménnyel (Exit)
- A  $4 > 3$  eljáráshívással a  $> /2$ -be belépünk majd sikeresen kilépünk (Call, Exit)
- A  $p/1$  eljárás sikeresen lefut  $p(4)$  eredménnyel (Exit)

## Eljárás-doboz modell — grafikus szemléltetés

$q(2).$   $q(4).$   $q(7).$

$p(X) \text{ :- } q(X), X > 3.$



## Eljárás-doboz modell — egyszerű nyomkövetési példa

### ● Az előző példa nyomkövetése SICStus Prologban

```
q(2). q(4). q(7).
```

```
p(X) :- q(X), X > 3.
```


```
| ?- trace, p(X).
      1      1 Call: p(_463) ?
      2      2 Call: q(_463) ?
?      2      2 Exit: q(2) ?           % ? ≡ nondeterminisztikus
kilépés
      3      2 Call: 2>3 ?
      3      2 Fail: 2>3 ?
      2      2 Redo: q(2) ?           % visszafelé menő végrehajtás
?      2      2 Exit: q(4) ?
      4      2 Call: 4>3 ?
      4      2 Exit: 4>3 ?
?      1      1 Exit: p(4) ?
X = 4 ? ;
      1      1 Redo: p(4) ?           % visszafelé menő végrehajtás
      2      2 Redo: q(4) ?           % visszafelé menő végrehajtás
      2      2 Exit: q(7) ?
      5      2 Call: 7>3 ?
      5      2 Exit: 7>3 ?
      1      1 Exit: p(7) ?
X = 7 ? ;
no
```

## Eljárás-doboz: egy összetettebb példa

$p(X,Y) \text{ :- } q(X,Z), p(Z,Y).$

$p(X,Y) \text{ :- } q(X,Y).$

$q(1,2). \quad q(2,3). \quad q(2,4).$





## Eljárás-doboz modell — „kapcsolási” alapelvek

---

- Hogyan építhető fel egy „szülő” eljárás doboza a benne hívott eljárások dobozaiból?
- Feltehető, hogy a klózfejekben (különböző) változók vannak, a fej-egyesítéseket hívás(okk)á alakítva
- Előre menő végrehajtás:
  - A szülő Hívás kapuját az első klóz első hívásának Hívás kapujára kötjük.
  - Egy rész-eljárás Kilépési kapuját
    - a következő hívás Hívás kapujára, vagy,
    - ha nincs következő hívás, akkor a szülő Kilépési kapujára kötjük
- Visszafelé menő végrehajtás:
  - Egy rész-eljárás Meghiúsulási kapuját
    - az előző hívás Újra kapujára, vagy,
    - ha nincs előző hívás, akkor a következő klóz első hívásának Hívás kapujára, vagy
    - ha nincs következő klóz, akkor a szülő Meghiúsulási kapujára kötjük
  - A szülő Újra kapuját mindegyik klóz utolsó hívásának Újra kapujára kötjük
    - mindig arra a klózra térünk vissza, amelyben legutoljára volt a vezérlés

## Eljárás-doboz modell — OO szemléletben

---

- Minden eljáráshoz tartozik egy osztály, amelynek van egy konstruktor függvénye (amely megkapja a hívási paramétereket) és egy „adj egy (következő) megoldást” metódusa.
- Az osztály nyilvántartja, hogy hányadik klózban jár a vezérlés
- A metódus első meghívásakor az első klóz első Hívás kapujára adja a vezérlést
- Amikor egy részjeljárás Hívás kapuhoz érkezünk, **létrehozunk** egy példányt a meghívandó eljárásból, majd
  - meghívjuk az eljáráspéldány „következő megoldás” metódusát (\*)
    - Ha ez sikerül, akkor a vezérlés átkerül a következő hívás Hívás kapujára, vagy a szülő Kilépési kapujára
    - Ha ez megghiúsul, akkor **megszüntetjük** az eljáráspéldányt majd ugrunk az előző hívás Újra kapujára, vagy a következő klóz elejére, stb.
- Amikor egy Újra kapuhoz érkezünk, a (\*) lépésnél folytatjuk.
- A szülő Újra kapuja (a „következő megoldás” nem első hívása) a tárolt klózsorszámnak megfelelő klózban az utolsó Újra kapura adja a vezérlést.

## OO szemléletű dobozok: p/2 „következő megoldás” metódusának C++ kódja

---

```

boolean p::next()
{ switch(clno) {
  case 0:                // entry point for the Call port
    clno = 1;            // enter clause 1:                p(X,Y) :- q(X,Z), p(Z,Y).
    qaptr = new q(x, &z); // create a new instance of subgoal q(X,Z)
redo11:
    if(!qaptr->next()) {  // if q(X,Z) fails
      delete qaptr;      // destroy it,
      goto cl2;          // and continue with clause 2 of p/2
    }
    pptr = new p(z, py); // otherwise, create a new instance of subgoal p(Z,Y)
  case 1:                // (enter here for Redo port if clno==1)
    /* redo12: */
    if(!pptr->next()) {   // if p(Z,Y) fails
      delete pptr;       // destroy it,
      goto redo11;       // and continue at redo port of q(X,Z)
    }
    return TRUE;         // otherwise, exit via the Exit port
  cl2:
    clno = 2;            // enter clause 2:                p(X,Y) :- q(X,Y).
    qbptra = new q(x, py); // create a new instance of subgoal q(X,Y)
  case 2:                // (enter here for Redo port if clno==1)
    /* redo21: */
    if(!qbptra->next()) { // if q(X,Y) fails
      delete qbptra;     // destroy it,
      return FALSE;      // and exit via the Fail port
    }
    return TRUE;         // otherwise, exit via the Exit port
} }

```

## Visszalépéses keresés — egy aritmetikai példa


---

- Példa: „jó” számok keresése
- A feladat: keressük meg azokat a kétjegyű számokat amelyek négyzete háromjegyű és a szám fordítottjával kezdődik
- A program:

```
% dec1(J): J egy pozitív decimális számjegy.  
dec1(1). dec1(2). dec1(3). dec1(4).  
dec1(5). dec1(6). dec1(7). dec1(8). dec1(9).  
  
% dec(J): J egy decimális számjegy.  
dec(0).  
dec(J) :- dec1(J).  
  
% Szam négyzete háromjegyű és a Szam fordítottjával kezdődik.  
joszam(Szam):-  
    dec1(A), dec(B),  
    Szam is A * 10 + B, Szam * Szam // 10 == B * 10 + A.
```

## Prolog végrehajtás — a 4-kapus doboz modell

```
joszam(Szam):-
    dec1(A), dec(B),
    Szam is A * 10 + B, Szam * Szam // 10 == B * 10 + A.
```



## Visszalépéses keresés — számintervallum felsorolása

---

- `dec(J)` felsorolta a 0 és 9 közötti egész számokat
- Általánosítás: soroljuk fel az  $N$  és  $M$  közötti egészeket ( $N$  és  $M$  maguk is egészek)

```
% between(M, N, I): M =< I =< N, I egész.
```

```
between(M, N, M) :-
```

```
    M =< N.
```

```
between(M, N, I) :-
```

```
    M < N,
```

```
    M1 is M+1,
```

```
    between(M1, N, I).
```

```
% dec(X): X egy decimális számjegy
```

```
dec(X) :- between(0, 9, X).
```

```
| ?- between(1, 2, _X), between(3, 4, _Y), Z is 10*_X+_Y.
```

```
Z = 13 ? ;
```

```
Z = 14 ? ;
```

```
Z = 23 ? ;
```

```
Z = 24 ? ;
```

```
no
```

## A SICStus eljárás-doboz alapú nyomkövetése — legfontosabb parancsok

---

### ● Alapvető nyomkövetési parancsok

- h <RET> (help) — parancsok listázása
- c <RET> (creep) vagy <RET> — továbblépés minden kapunál megálló nyomkövetéssel
- l <RET> (leap) — csak töréspontnál áll meg, de a dobozokat építi
- z <RET> (zip) — csak töréspontnál áll meg, dobozokat nem épít
- + <RET> ill. - <RET> — töréspont rakása/eltávolítása a kurrens predikátumra
- s <RET> (skip) — eljárástörzs átlépése (Call/Redo  $\Rightarrow$  Exit/Fail)
- o <RET> (out) — kilépés az eljárástörzsből

### ● A Prolog végrehajtást megváltoztató parancsok

- u <RET> (unify) — a kurrens hívást végrehajtás helyett egyesíti egy beolvasott kifejezéssel.
- r <RET> (retry) — újratekdi a kurrens hívás végrehajtását (ugrás a Call kapura)

### ● Információ-megjelenítő és egyéb parancsok

- w <RET> (write) — a hívás kiírása mélység-korlátozás nélkül
- b <RET> (break) — új, beágyazott Prolog interakciós szint létrehozása
- n <RET> (notrace) — nyomkövető kikapcsolása
- a <RET> (abort) — a kurrens futás abbahagyása

## TOVÁBBI VEZÉRLÉSI SZERKEZETEK





## Diszjunkció, példa: az „őse” predikátum

---

- Az „őse” reláció a „szülője” reláció tranzitív lezártja: a szülő ős (1), és az ős őse is ős (2), azaz:

```
% ose0(E, Os) : E ose Os.
ose0(E, Sz) :- szuloje(E, Sz).           % (1)
ose0(E, Os) :- ose0(E, Os0), ose0(Os0, Os). % (2)
```

- Az ose0 definíciója matematikailag helyes, de végtelen Prolog keresési teret ad:

```
szuloje(gyerek,apa). szuloje(gyerek,anya). szuloje(anya,nagyapa).

| ?- ose0(gyerek, Os).
    Os = apa ? ; Os = anya ? ; {néhány másodperc után:}
    ! Resource error: insufficient memory
```

- A végtelen rekurzió oka: Az `:- ose0(apa, X).` cél esetén az (1) klóz meghiúsul, (2) pedig egy `:- ose0(apa, Y), ose0(Y, X).` célsorozathoz vezet stb.

- A balrekurziót kiküszöbölve kapjuk:

```
ose1(E, Sz) :- szuloje(E, Sz).           % (3)
ose1(E, Os) :- szuloje(E, Sz), ose1(Sz, Os). % (4)

| ?- ose1(gyerek, Os).
Os = apa ? ; Os = anya ? ; Os = nagyapa ? ; no
```

- Ez minden `szuloje(X,Y)` részcélt kétszer hajt végre: (3)-ban és (4)-ben.

## A diszjunkció

---

- Az `ose1` predikátum hatékonyabbá tehető klózai összevonásával:

```
ose2(E, Os) :- szuloje(E, Sz), maga_vagy_ose(Sz, Os).
```

```
maga_vagy_ose(E, E). (1)
```

```
maga_vagy_ose(E, Os) :- ose2(E, Os).
```

- A `maga_vagy_ose` predikátum egy ún. **diszjunkció** bevezetésével kiküszöbölhető:

```
ose3(E, Os) :-
    szuloje(E, Sz),
    (   Os = Sz
    ;   ose3(Sz, Os)
    ).
```

- A SICStus Prolog ténylegesen úgy implementálja a fenti diszjunkciót, hogy bevezet egy `maga_vagy_ose`-vel azonos segéd-predikátumot és az `ose3` klózt `ose2`-vé alakítja.
- (Ismétlés:) Az  $x=y$  beépített predikátum a két argumentumát egyesíti.
- Az  $=/2$  eljárás egy tényállítással definiálható:  $U = V \equiv (U, V), \text{vö. (1)}.$

## A diszjunkció mint szintaktikus édesítőszer

- A diszjunkció akárhány tagú lehet. A ‘;’ művelet gyengébben köt mint a ‘,’, ezért a diszjunkciót mindig zárójelbe tesszük, míg az ágait nem kell zárójelezni. Példa, „szabványos” formázással:

```
a(X, Y, Z) :-
    p(X, U), q(Y, V),
    (    r(U, T), s(T, Z)
    ;    t(V, Z)
    ;    t(U, Z)
    ),
    u(X, Z).
```

- A diszjunkció egy segéd-predikátummal mindig kiküszöbölhető
  - Megkeressük azokat a változókat, amelyek a diszjunkcióban és azon kívül is előfordulnak
  - A segéd-predikátumnak ezek a változók lesznek az argumentumai
  - A segéd-predikátum minden klóza megfelel a diszjunkció egy ágának

```
seged(U, V, Z) :- r(U, T), s(T, Z).
seged(U, V, Z) :- t(V, Z).
seged(U, V, Z) :- t(U, Z).
```

```
a(X, Y, Z) :-
    p(X, U), q(Y, V),
    seged(U, V, Z),
    u(X, Z).
```

- A diszjunkció szemantikáját ezzel a segéd-predikátumos átalakítással definiáljuk.

## Diszjunkció — megjegyzések

---

- Az egyes klózik 'ÉS' vagy 'VAGY' kapcsolatban vannak?

- A program klózai **ÉS** kapcsolatban vannak, pl.

```
szuloje('Imre', 'István').           szuloje('Imre', 'Gizella').
```

jelentése: Imre szülője István **ÉS** Imre szülője Gizella.

- Az ÉS kapcsolatban levő klózik alternatív (VAGY kapcsolatban levő) válaszokhoz vezetnek:

```
:- szuloje('Imre' Sz). => Sz = 'István' ? ; Sz = 'Gizella' ? ; no
```

A „Ki Imre szülője?” kérdésre a válasz: István vagy Gizella.

- A fenti két klózos predikátum átalakítható egyetlen klózzá, diszjunkció segítségével:

```
szuloje('Imre', Sz) :-
    (   Sz = 'István'           (*)
    ;   Sz = 'Gizella'         (*)
    ).
```

A konjunkció ezáltal diszjunkcióvá alakult (vö. De Morgan azonosságok).

- Általánosan: tetszőleges predikátum egyklózosá alakítható:

- a klózikat átalakítjuk azonos fejűvé, új változók és egyenlőségek bevezetésével:

```
szuloje('Imre', Sz) :- Sz = 'István'.
szuloje('Imre', Sz) :- Sz = 'Gizella'.
```

- a klóztörzseket egy diszjunkcióvá fogjuk össze, amely az új predikátum törzse (lásd (\*)).

## Negáció

---

- Feladat: Keressünk (adatbázisunkban) egy olyan szülőt, aki **nem** nagyszülő!
- Ehhez negációra van szükségünk:
  - Meghiúsulós negáció: a `\+` Hívás szerkezet lefuttatja Hívást, és pontosan akkor sikerül, ha a Hívás meghiúsult.

- Egy megoldás:

```
| ?- szülője(_, X), \+ nagyszülője(_, X).
X = 'István' ? ;
X = 'Gizella' ? ;
no
```

- Egy ekvivalens megoldás:

```
| ?- szülője(_Gy, X), \+ szülője(_, _Gy).
X = 'István' ? ;
X = 'Gizella' ? ;
no
```

- Mi történik ha a két hívást megcseréljük?

```
| ?- \+ szülője(_, _Gy), szülője(_Gy, X).
no
```

## A meghíúsulások negáció (NF — Negation by Failure)

- $A \setminus +$  Hívás beépített meta-eljárás (vö.  $\nmid$  — nem bizonyítható)
  - végrehajtja a Hívás hívást,
  - ha Hívás sikeresen lefutott, akkor meghíúsul,
  - egyébként (azaz ha Hívás meghíúsult) sikerül.
- $\setminus +$  Hívás futása során Hívás legfeljebb egy megoldása áll elő
- $\setminus +$  Hívás sohasem helyettesít be változót
- Gondok a meghíúsulások negációval:
  - „zárt világ feltételezése” (CWA) — ami nem bizonyítható, az nem igaz.
 

?- $\setminus +$ szuloje('Imre', X).	----> no
?- $\setminus +$ szuloje('Géza', X).	----> true ?
  - $\setminus + H$  deklaratív szemantikája:  $\neg \exists X(H)$ , ahol  $X$  a  $H$ -ban a hívás pillanatában behelyettesítetlen változókat jelöli.
 

?- $\setminus +$ X = 1, X = 2.	----> no
?- X = 2, $\setminus +$ X = 1.	----> X = 2 ?

## Példa: együttható meghatározása lineáris kifejezésben

---

- Formula: számokból és az 'x' névkonstansból '+' és '\*' operátorokkal épül fel.
- `% :- type kif == {x} \/ number \/ {kif+kif} \/ {kif*kif}.`
- Lineáris formula: a '\*' operátor legalább egyik oldalán szám áll.

`% egyhat(Kif, E): A Kif lineáris formulában az x együtthatója E.`

`egyhat(x, 1).`

`egyhat(Kif, E) :-`

`number(Kif), E = 0.`

`egyhat(K1+K2, E) :-`

`egyhat(K1, E1),`

`egyhat(K2, E2),`

`E is E1+E2.`

`egyhat(K1*K2, E) :-`

`number(K1),`

`egyhat(K2, E0),`

`E is K1*E0.`

`egyhat(K1*K2, E) :-`

`number(K2),`

`egyhat(K1, E0),`

`E is K2*E0.`

`| ?- egyhat(((x+1)*3)+x+2*(x+x+3), E).`

`E = 8 ? ;`

`no`

`| ?- egyhat(2*3+x, E).`

`E = 1 ? ;`

`E = 1 ? ; no`

## Együttható meghatározása: többszörös megoldások kiküszöbölése

---

- negáció alkalmazásával:

```
(...)  
egyhat(K1*K2, E) :-  
    number(K1), egyhat(K2, E0), E is K1*E0.  
egyhat(K1*K2, E) :-  
    \+ number(K1),  
    number(K2), egyhat(K1, E0), E is K2*E0.
```

- hatékonyabban, feltételes kifejezéssel:

```
(...)  
egyhat(K1*K2, E) :-  
    (    number(K1) -> egyhat(K2, E0), E is K1*E0  
    ;    number(K2), egyhat(K1, E0), E is K2*E0  
    ).
```



## Feltételes kifejezések

---

- Szintaxis (felt, akkor, egyébként tetszőleges célsorozatok):

```
(...) :-  
    (...),  
    ( felt -> akkor  
    ;  egyébként  
    ),  
    (...).
```

- Deklaratív szemantika: a fenti alak jelentése megegyezik az alábbival, ha a `felt` egy egyszerű feltétel (nem oldható meg többféleképpen):

```
(...) :-  
    (...),  
    ( felt, akkor  
    ;  \+ felt, egyébként  
    ),  
    (...).
```

## Feltételes kifejezések (folyt.)

---

- **Procedurális szemantika**

A `(felt->akkor;egyébként)`, folytatás célsorozat végrehajtása:

- Végrehajtjuk a `felt` hívást.
- Ha `felt` sikeres, akkor az `akkor`, folytatás célsorozatra redukáljuk a fenti célsorozatot, a `felt` *első* megoldása által eredményezett behelyettesítésekkel. A `felt` cél többi megoldását nem keressük meg.
- Ha `felt` sikertelen, akkor az `egyébként`, folytatás célsorozatra redukáljuk, behelyettesítés nélkül.

- **Többszörös elágaztatás skatulyázott feltételes kifejezésekkel:**

<code>( felt1 -&gt; akkor1</code>	<code>( felt1 -&gt; akkor1</code>
<code>; felt2 -&gt; akkor2</code>	<code>; (felt2 -&gt; akkor2</code>
<code>; ...</code>	<code>; ...</code>
<code>)</code>	<code>...))</code>

- Az `egyébként` rész elhagyható, alapértelmezése: `fail`.
- A `\+ felt` negáció kiváltható a `( felt -> fail ; true )` feltételes kifejezéssel.

## Feltételes kifejezés — példák

---

### ● Faktoriális

```
% fakt(+N, ?F): N! = F.  
fakt(N, F) :-  
    (    N = 0 -> F = 1                                % N = 0,  F = 1  
    ;    N > 0, N1 is N-1, fakt(N1, F1), F is N*F1  
    ).
```

- Jelentése azonos a sima diszjunkciós alakkal (lásd komment), de annál hatékonyabb, mert nem hagy maga után választási pontot.

### ● Szám előjele

```
% Sign = sign(Num)  
sign(Num, Sign) :-  
    (    Num > 0 -> Sign = 1  
    ;    Num < 0 -> Sign = -1  
    ;    Sign = 0  
    ).
```

# A PROLOG SZINTAXIS

## A Prolog szintaxis összefoglalása

---

- A Prolog szintaxis alapelvei

- Minden programelem kifejezés!
- A szükséges összekötő jelek (', ', ';', ':- -->'): szabványos operátorok.
- A beolvasott kifejezést funktora alapján osztályozzuk:
  - *kérdés:*  $?- \text{Cél}.$   
*Célt* lefuttatja, és a változó-behelyettesítéseket kiírja (ez az alapértelmezés az ún. top-level interaktív felületen).
  - *parancs:*  $:- \text{Cél}.$   
 A *Célt* csendben lefuttatja. Pl. deklaráció (operátor, ...) elhelyezésére.
  - *szabály:*  $\text{Fej} :- \text{Törzs}.$   
 A szabályt felveszi a programba.
  - *nyelvtani szabály:*  $\text{Fej} --> \text{Törzs}.$   
 Prolog szabállyá alakítja és felveszi (lásd a DCG nyelvtan).
  - *tényállítási:*  $\text{Minden egyéb kifejezés}.$   
 Üres törzsű szabályként felveszi a programba.

## A Prolog nyelv-változatok

---

- A SICStus rendszer két üzemmódja
  - iso Az ISO Prolog szabványnak megfelelő.
  - sicstus Korábbi változatokkal kompatibilis.
  - Állítása: `set_prolog_flag(language, Mód)`.
  - Különbségek:
    - szintaxis-részletek, pl. a `0x1ff` szám-alak csak ISO módban,
    - beépített eljárások viselkedésének kisebb eltérései.
  - az eddig ismertetett eljárások hatása lényegében nem változik.

## Szintaktikus édesítőszerek — összefoglalás, gyakorlati tanácsok

---

### ● Operátoros kifejezések alapstruktúra alakra hozása

- Zárójelezzük be a kifejezést, az operátorok prioritása és fajtája alapján, például  $-a+b*2 \Rightarrow ((-a)+(b*2))$ .

- Hozzuk az operátoros kifejezéseket alapstruktúra alakra:

$(A \text{ Inf } B) \Rightarrow \text{Inf}(A,B)$ ,  $(\text{Pref } A) \Rightarrow \text{Pref}(A)$ ,  $(A \text{ Postf}) \Rightarrow \text{Postf}(A)$

Példa:  $((-a)+(b*2)) \Rightarrow (- (a) + * (b,2)) \Rightarrow + (- (a), * (b,2))$ .

- Trükkös esetek:

- A vesszőt névként idézni kell: pl.  $(pp,(qq;rr)) \Rightarrow ', '(pp,;(qq,rr))$ .

- - *Szám*  $\Rightarrow$  negatív számkonstans, de - *Egyéb*  $\Rightarrow$  prefix alak.

Példa:  $-1+2 \Rightarrow +(-1,2)$ , de  $-a+b \Rightarrow +(- (a), b)$ .

- *Név*(...)  $\Rightarrow$  struktúrakifejezés;

*Név* (...)  $\Rightarrow$  prefix operátoros kifejezés. Példák:

-  $(1,2) \Rightarrow -(1,2)$  (változatlan), de

-  $(1,2) \Rightarrow -(' , '(1,2))$ .

## Szintaktikus édesítőszerkek — listák, egyebek

---

### Listák alapstruktúra alakra hozása

#### Farok-megadás betoldása.

$[1,2] \Rightarrow [1,2|[]]$ .  $[[X|Y]] \Rightarrow [[X|Y]|[]]$

#### Vessző (ismételt) kiküszöbölése $[Elem1,Elem2\dots] \Rightarrow [Elem1|[Elem2\dots]]$ .

$[1,2|[]] \Rightarrow [1|[2|[]]]$

$[1,2,3|[]] \Rightarrow [1|[2,3|[]]] \Rightarrow [1|[2|[3|[]]]]$

#### Struktúrakifejezéssé alakítás: $[Fej|Farok] \Rightarrow .(Fej, Farok)$ .

$[1|[2|[]]] \Rightarrow .(1, .(2, []))$ ,  $[[X|Y]|[]] \Rightarrow .(. (X,Y), [])$

### Egyéb szintaktikus édesítőszerkek:

#### Karakterkód-jelölés: $0'Kar$ .

$0'a \Rightarrow 97$ ,  $0'b \Rightarrow 98$ ,  $0'c \Rightarrow 99$ ,  $0'd \Rightarrow 100$ ,  $0'e \Rightarrow 101$

#### Füzér (string): $"xyz\dots" \Rightarrow$ az $xyz\dots$ karakterek kódját tartalmazó lista

$"abc" \Rightarrow [97,98,99]$ ,  $" " \Rightarrow []$ ,  $"e" \Rightarrow [101]$

#### Kapcsos zárójelezés: $\{Kif\} \Rightarrow \{ \}(Kif)$ (egy $\{ \}$ nevű, egyargumentumú struktúra — a $\{ \}$ jelpár egy önálló lexikai elem, egy névkonstans).

#### Bináris, hexa stb. alak (csak iso módban), pl. $0b101010$ , $0x1a$ .



## Kifejezések szintaxisa — kétszintű nyelvtanok

---

- Egy részlet egy „hagyományos” nyelv kifejezés-szintaxisából:

$$\begin{aligned}
 \langle \text{kifejezés} \rangle &::= && \langle \text{tag} \rangle \\
 &| && \langle \text{kifejezés} \rangle \langle \text{additív művelet} \rangle \langle \text{tag} \rangle \\
 \langle \text{tag} \rangle &::= && \langle \text{tényező} \rangle \\
 &| && \langle \text{tag} \rangle \langle \text{multiplikatív művelet} \rangle \langle \text{tényező} \rangle \\
 \langle \text{tényező} \rangle &::= && \langle \text{szám} \rangle \mid \langle \text{azonosító} \rangle \mid ( \langle \text{kifejezés} \rangle )
 \end{aligned}$$

- Ugyanez kétszintű nyelvtannal:

$$\begin{aligned}
 \langle \text{kifejezés} \rangle &::= && \langle \text{kif } 2 \rangle \\
 \langle \text{kif } N \rangle &::= && \langle \text{kif } N-1 \rangle \\
 &| && \langle \text{kif } N \rangle \langle N \text{ prioritású művelet} \rangle \langle \text{kif } N-1 \rangle \\
 \langle \text{kif } 0 \rangle &::= && \langle \text{szám} \rangle \mid \langle \text{azonosító} \rangle \mid ( \langle \text{kif } 2 \rangle ) \\
 &&& \{ \text{az additív ill. multiplikatív műveletek prioritása } 2 \text{ ill. } 1 \}
 \end{aligned}$$

## Prolog kifejezések szintaxisa

---

$\langle \text{programelem} \rangle ::= \langle \text{kifejezés 1200} \rangle \langle \text{záró-pont} \rangle$

$\langle \text{kifejezés } N \rangle ::=$

- $\langle \text{op } N \text{ fx} \rangle \langle \text{köz} \rangle \langle \text{kifejezés } N-1 \rangle$
- $| \langle \text{op } N \text{ fy} \rangle \langle \text{köz} \rangle \langle \text{kifejezés } N \rangle$
- $| \langle \text{kifejezés } N-1 \rangle \langle \text{op } N \text{ xfx} \rangle \langle \text{kifejezés } N-1 \rangle$
- $| \langle \text{kifejezés } N-1 \rangle \langle \text{op } N \text{ xfy} \rangle \langle \text{kifejezés } N \rangle$
- $| \langle \text{kifejezés } N \rangle \langle \text{op } N \text{ yfx} \rangle \langle \text{kifejezés } N-1 \rangle$
- $| \langle \text{kifejezés } N-1 \rangle \langle \text{op } N \text{ xf} \rangle$
- $| \langle \text{kifejezés } N \rangle \langle \text{op } N \text{ yf} \rangle$
- $| \langle \text{kifejezés } N-1 \rangle$

$\langle \text{kifejezés 1000} \rangle ::= \langle \text{kifejezés 999} \rangle , \langle \text{kifejezés 1000} \rangle$

$\langle \text{kifejezés 0} \rangle ::=$

- $\langle \text{név} \rangle ( \langle \text{argumentumok} \rangle )$
- $\{ A \langle \text{név} \rangle \text{ és } a ( \text{közvetlenül egymás után áll!} ) \}$
- $| ( \langle \text{kifejezés 1200} \rangle ) \mid \{ \langle \text{kifejezés 1200} \rangle \}$
- $| \langle \text{lista} \rangle \mid \langle \text{füzér} \rangle$
- $| \langle \text{név} \rangle \mid \langle \text{szám} \rangle \mid \langle \text{változó} \rangle$

## Kifejezések szintaxisa — folytatás

---

$\langle \text{op } N \ T \rangle ::=$   $\langle \text{név} \rangle \{ \text{feltéve, hogy } \langle \text{név} \rangle \ N \text{ prioritású és } T \text{ típusú operátornak lett deklarálva} \}$

$\langle \text{argumentumok} \rangle ::=$   $\langle \text{kifejezés 999} \rangle$   
|  $\langle \text{kifejezés 999} \rangle , \langle \text{argumentumok} \rangle$

$\langle \text{lista} \rangle ::=$   $[]$   
|  $[ \langle \text{listakif} \rangle ]$

$\langle \text{listakif} \rangle ::=$   $\langle \text{kifejezés 999} \rangle$   
|  $\langle \text{kifejezés 999} \rangle , \langle \text{listakif} \rangle$   
|  $\langle \text{kifejezés 999} \rangle \mid \langle \text{kifejezés 999} \rangle$

$\langle \text{szám} \rangle ::=$   $\langle \text{előjeltelen szám} \rangle$   
|  $+ \langle \text{előjeltelen szám} \rangle$   
|  $- \langle \text{előjeltelen szám} \rangle$

$\langle \text{előjeltelen szám} \rangle ::=$   $\langle \text{természetes szám} \rangle$   
|  $\langle \text{lebegőpontos szám} \rangle$

## Kifejezések szintaxisa — megjegyzések

---

- A  $\langle \text{kifejezés } N \rangle$ -ben  $\langle \text{köz} \rangle$  csak akkor kell ha az őt követő kifejezés nyitó-zárójellel kezdődik.

```
| ?- op(500, fx, succ).
yes
| ?- write_canonical(succ (1,2)), nl, write_canonical(succ(1,2)).
succ('','(1,2))
succ(1,2)
```

- A  $\{ \langle \text{kifejezés} \rangle \}$  azonos a  $\{ \}(\langle \text{kifejezés} \rangle)$  struktúrával, ez pl. a DCG nyelvtanoknál hasznos.

```
| ?- write_canonical({a}).
{}(a)
```

- Egy  $\langle \text{füzér} \rangle$  " jelek közé zárt karaktersorozat, általában a karakterek kódjainak listájával azonos.

```
| ?- write("baba").
[98,97,98,97]
```

## A Prolog lexikai elemei 1. (ismétlés)

---

### ● $\langle \text{név} \rangle$

- kisbetűvel kezdődő alfanumerikus jelsorozat (ebben megengedve kis- és nagybetűt, számjegyeket és aláhúzásjelet);
- egy vagy több ún. speciális jelből (+-\* /\\$^<>='~: .?@#&) álló jelsorozat;
- az önmagában álló ! vagy ; jel;
- a [] {} jelpárok;
- idézőjelek ( ' ) közé zárt tetszőleges jelsorozat, amelyben \ jellel kezdődő escape-szekvenciákat is elhelyezhetünk.

### ● $\langle \text{változó} \rangle$

- nagybetűvel vagy aláhúzással kezdődő alfanumerikus jelsorozat.
- az azonos jelsorozattal jelölt változók egy klózon belül azonosaknak, különböző klózokban különbözőeknek tekintődnek;
- kivétel: a semmis változók (\_) minden előfordulása különböző.

## A Prolog lexikai elemei 2.

---

- $\langle$  természetes szám  $\rangle$

- (decimális) számjegysorozat;
- 2, 8 ill. 16 alapú számrendszerben felírt szám, ilyenkor a számjegyeket rendre a 0b, 0o, 0x karakterekkel kell prefixálni (csak iso módban)
- karakterkód-konstans 0'c alakban, ahol c egyetlen karakter (vagy egy ilyet jelölő escape-szekvencia)

- $\langle$  lebegőpontos szám  $\rangle$

- mindenképpen tartalmaz tizedespontot
- mindkét oldalán legalább egy (decimális) számjeggyel
- e vagy E betűvel jelzett esetleges exponens

## Megjegyzések és formázó-karakterek

---

- Megjegyzések (comment)

- A % százalékjeltől a sor végéig
- A /\* jelpártól a legközelebbi \*/ jelpárig.

- Formázó elemek

- szóköz, újsor, tabulátor stb. (nem látható karakterek)
- megjegyzés

- A programszöveg formázása

- formázó elemek (szóköz, újsor stb.) szabadon elhelyezhetők;
- kivétel: struktúrakifejezés neve után nem szabad formázó elemet tenni;
- prefix operátor és ( közé kötelező formázó elemet tenni;
- ⟨ záró-pont ⟩: egy . karakter amit egy formázó elem követ.

# PROLOG PÉLDÁK





## A régi jegyzet bevezető példája: útvonalkeresés

---

- A feladat:

- Tekintsük (autóbusz)járatok egy halmazát.
- Mindegyik járhoz a két végpont és az útvonal hossza van megadva.
- Írjunk Prolog eljárást, amellyel megállapítható, hogy két pont összeköthető-e pontosan N csatlakozó járáttal!

- Átfogalmazás: egy súlyozott irányítatlan gráfban két pont közötti utat keresünk. Élek:

```
% járat(A, B, H): Az A és B városok között van járat, és hossza H km.  
járat('Budapest', 'Prága', 515).  
járat('Budapest', 'Bécs', 245).  
járat('Bécs', 'Berlin', 635).  
járat('Bécs', 'Párizs', 1265).
```

- Irányított élek:

```
% útszakasz(A, B, H): A-ból B-be eljuthatunk egy H úthosszú járáttal.  
útszakasz(Kezdet, Cél, H) :-  
    (    járat(Kezdet, Cél, H)  
    ;    járat(Cél, Kezdet, H)  
    ).
```

## Az útvonalkeresési feladat — folytatás

---

### ● Adott lépekszámú útvonal (él-sorozat) és hossza:

```
% útvonal(N, A, B, H): A és B között van (pontosan)
% N szakaszból álló útvonal, amelynek összhossza H.
útvonal(0, Hová, Hová, 0).
útvonal(N, Honnan, Hová, H) :-
    N > 0,
    N1 is N-1,
    útszakasz(Honnan, Közben, H1),
    útvonal(N1, Közben, Hová, H2),
    H is H1+H2.
```

### ● Futási példa:

```
| ?- útvonal(2, 'Párizs', Hová, H).
    H = 1900, Hová = 'Berlin' ? ;
    H = 2530, Hová = 'Párizs' ? ;
    H = 1510, Hová = 'Budapest' ? ;
    no
```

## Körmentes út keresése

---

- Könyvtár betöltése, adott funktorú eljárások importálásával:

```
:- use_module(library(lists), [member/2]).
```

- Segéd-argumentum: az érintett városok listája, fordított sorrendben

```
% útvonala_2(N, A, B, H): A és B között van (pontosan)
% N szakaszból álló körmentes útvonala, amelynek összhossza H.
útvonala_2(N, Honnan, Hová, H) :-
    útvonala_2(N, Honnan, Hová, [Honnan], H).

% útvonala_2(N, A, B, Kizártak, H): A és B között van pontosan
% N szakaszból álló körmentes, Kizártak elemein át nem menő H hosszú út.
útvonala_2(0, Hová, Hová, Kizártak, 0).
útvonala_2(N, Honnan, Hová, Kizártak, H) :-
    N > 0, N1 is N-1, útszakasz(Honnan, Közben, H1),
    \+ member(Közben, Kizártak),
    útvonala_2(N1, Közben, Hová, [Közben/Kizártak], H2), H is H1+H2.
```

- Példa-futás:

```
| ?- útvonala_2(2, 'Párizs', Hová, H).
    H = 1900, Hová = 'Berlin' ? ;
    H = 1510, Hová = 'Budapest' ? ; no
```

## Továbbfejlesztés: körmentes út keresése, útvonal-gyűjtéssel

- Az alapötlet: a Kizártak listában gyűlik a (fordított) útvonal.
- A rekurzív eljárásban szükséges egy **új argumentum**, hogy az útvonalat kiadjuk!

```
:- use_module(library(lists), [member/2, reverse/2]).

% útvonal_3(N, A, B, Út, H): A és B között van (pontosan)
% N szakaszból álló körmentes Út útvonal, amelynek összhossza H.
útvonal_3(N, Honnan, Hová, Út, H) :-
    útvonal_3(N, Honnan, Hová, [Honnan], FÚt, H),
    reverse(FÚt, Út).

% útvonal_3(N, A, B, FÚt0, FÚt, H): A és B között van pontosan
% N szakaszból álló körmentes, FÚt0 elemein át nem menő H hosszú út.
% FÚt = (az A → B útvonal megfordítása) ⊕ FÚt0.
útvonal_3(0, Hová, Hová, FordÚt, FordÚt, 0).
útvonal_3(N, Honnan, Hová, FordÚt0, FordÚt, H) :-
    N > 0, N1 is N-1, útszakasz(Honnan, Közben, H1),
    \+ member(Közben, FordÚt0),
    útvonal_3(N1, Közben, Hová, [Közben|FordÚt0], FordÚt, H2), H is H1+H2.

| ?- útvonal_3(2, 'Párizs', _, Út, H).
    H = 1900, Út = ['Párizs','Bécs','Berlin'] ? ;
    H = 1510, Út = ['Párizs','Bécs','Budapest'] ? ; no
```

## Súlyozott gráf ábrázolása éllistával

---

### ● A gráf ábrázolása

- a gráf élek listája,
- az él egy három-argumentumú struktúra,
- argumentumai: a két végpont és a súly.

### ● Típus-definíció

```
% :- type él ---> él(pont, pont, súly).  
% :- type pont == atom.  
% :- type súly == int.  
% :- type gráf == list(él).
```

### ● Példa

```
hálózat([él('Budapest', 'Bécs', 245),  
         él('Budapest', 'Prága', 515),  
         él('Bécs', 'Berlin', 635),  
         él('Bécs', 'Párizs', 1265)]).
```

## Ismétlődésmentes útvonal keresése listával ábrázolt gráfban

---

```
:- use_module(library(lists), [select/3]).

% útvonal_4(N, G, A, B, L, H): A G gráfban van egy A-ból
% B-be menő N szakaszból álló L út, melynek összhossza H.
útvonal_4(0, _Gráf, Hová, Hová, [Hová], 0).
útvonal_4(N, Gráf, Honnan, Hová, [Honnan|Út], H) :-
    N > 0, N1 is N-1,
    select(Él, Gráf, Gráf1),
    él_végpontok_hossz(Él, Honnan, Közben, H1),
    útvonal_4(N1, Gráf1, Közben, Hová, Út, H2),
    H is H1+H2.

% él_végpontok_hossz(Él, A, B, H): Az Él irányítatlan él
% végpontjai A és B, hossza H.
él_végpontok_hossz(él(A,B,H), A, B, H).
él_végpontok_hossz(él(A,B,H), B, A, H).

| ?- hálózat(_Gráf), útvonal_4(2, _Gráf, 'Budapest', _, Út, H).
    H = 880, Út = ['Budapest','Bécs','Berlin'] ? ;
    H = 1510, Út = ['Budapest','Bécs','Párizs'] ? ;
    no
```

## Bináris fákra vonatkozó példasor — fa levele

- Ismétlés: egészekből álló bináris fa:

```
:- type itree == {node(itree, itree)} \/ {leaf(int)}.
:- type itree ---> node(itree, itree) | leaf(int).
```

- Írjunk egy predikátumot annak eldöntésére, hogy egy adott érték szerepel-e egy fa levelében (vö. member/2)!

- *% fa\_levele(Fa, Ertek): A Fa bináris fa levelében szerepel az Ertek szám.*  
`fa_levele(leaf(V), V). % ha a fa egyetlen levélből áll és a levélbeli`  
`% érték megegyezik a keresettel, akkor ‘siker’`  
`fa_levele(node(L,_), V) :-`  
`fa_levele(L, V). % ha a bal részében van, akkor az egészben is`  
`fa_levele(node(_,R), V) :-`  
`fa_levele(R, V). % ha a jobb részében van, akkor az egészben is`

- Az aláhúzásjel egy ún. semmis (void) változó, ennek minden előfordulása különböző változó!

- Példák: ellenőrzés (1), adott fa leveleinek felsorolása (2),  
adott levelű fák felsorolása, (3) ( $\infty$  keresési tér).

```
| ?- fa_levele(node(node(leaf(1),leaf(2)),leaf(7)), 2). ==> yes (1)
| ?- fa_levele(node(node(leaf(1),leaf(2)),leaf(7)), 3). ==> no (1)
| ?- fa_levele(node(leaf(1),leaf(7)), E). ==> E = 1 ? ; E = 7 ? ; no (2)
| ?- fa_levele(Fa, 3). ==> Fa = leaf(3) ? ; Fa = node(leaf(3),_A) ? ; ... (3)
```

## Összetett adatstruktúrák konjunktív és diszjunktív bejárása

- Prologban egy összetett adatstruktúrát kétféleképpen lehet bejárni:

- konjunktívan: a részek bejárása ÉS kapcsolatban van, általában egy eredményt ad

- pl. fa összegzése (sum\_tree), fa ellenőrzése (itree), fa kiírása:

```
% faki(Fa): Fa kiírható (mindig teljesül :-). Mellékhatásként kiírja a Fa fát.
faki(leaf(V)) :-
    write('@'), write(V).    % A write(X) beépített pred. kiírja az X kifejezést.
faki(node(L,R)) :-
    write('('), faki(L), write(' -- '), faki(R), write(')').

| ?- faki(node(node(leaf(1),leaf(8)),leaf(7))).    ⇒ ((@1 -- @8) -- @7)
yes
```

- diszjunktívan: a részek bejárása VAGY kapcsolatban van, visszalépéskor új eredmény

- pl. fa leveleinek felsorolása (fa\_levele)

- A diszjunktív, felsoroló bejárás könnyen kiegészíthető további feltételekkel

- Keressük egy fának az (5,10) intervallumba eső leveleit:

```
| ?- _Fa = node(node(leaf(1),leaf(8)),leaf(7)), fa_levele(_Fa, E), 5 < E, E < 10.
    ⇒ E = 8 ? ; E = 7 ? ; no
| ?- _Fa = (...), fa_levele(_Fa, E), 5 < E, E < 10, write(E), write(' '), fail.
    ⇒ 8 7 ⇒ no
```

- A fail beépített predikátum mindig megghiúsul, pl. ún. visszalépéses ciklus szervezésére jó.



## Levél elhagyása bináris fából

- Írjunk egy predikátumot annak eldöntésére, hogy egy adott érték szerepel-e egy összetett fa levelében! A predikátum adja vissza a levél elhagyása után fennmaradó fát!

```
% flm(Fa, Ertek, Marad): A Fa összetett bináris fa egy Ertek értékű
% levelének elhagyása után marad a Marad fa. (flm = fa_level_maradek)
flm(node(leaf(V),T), V, T).    % ha a bal részfa a keresett levél
                                % akkor a jobb részfa a maradék
flm(node(T,leaf(V)), V, T).    % ugyanez jobboldali levél esetére
flm(node(L0,R), V, node(L,R)) :-
    flm(L0, V, L).            % ha a bal részfából elhagyható a levél
                                % akkor ennek maradéka, kiegészítve
                                % a jobb részfával, lesz a teljes fa maradéka
flm(node(L,R0), V, node(L,R1)) :-
    flm(R0, V, R1).           % ugyanez jobb részfa esetére
```

- Az flm/3 predikátum használható ellenőrzése, de fa szétbontására is:

```
| ?- flm(node(leaf(1),node(leaf(2),leaf(3))), 2, T). ==>
    T = node(leaf(1),leaf(3)) ? ; no
| ?- flm(node(leaf(1),node(leaf(2),leaf(3))), 7, T). ==> no
| ?- flm(node(leaf(1),node(leaf(2),leaf(3))), X, T). ==>
    T = node(leaf(2),leaf(3)), X = 1 ? ;
    T = node(leaf(1),leaf(3)), X = 2 ? ;
    T = node(leaf(1),leaf(2)), X = 3 ? ; no
```

## Levél beszúrása bináris fába

- Írjunk egy predikátumot arra, hogy egy adott értékű levelet egy fába minden lehetséges módon beszúrjon!
- Nem kell írunk, már megírtuk! Az `flm` predikátum erre is jó:

```
% flm(Fa, Ertek, Marad): A Fa összetett bináris fa egy Ertek értékű
% levelének elhagyása után marad a Marad fa. Röviden: Fa - Ertek = Marad.

% flm(Fa, Ertek, Marad): A Fa (összetett) bináris fa úgy áll elő, hogy
% a Marad fába beszúrunk egy E értékű levelet. Fa = Marad + Ertek.
flm(node(leaf(V),T), V, T).    % Egy T fába beszúrhatunk egy levelet
(...)                        % úgy, hogy az egylevelű fát T elé tesszük
```

- Példák:

```
| ?- flm(Fa, 2, leaf(1)), faki(Fa), write(' '), fail.
(@2 -- @1) (@1 -- @2)                ==> no
| ?- flm(Fa0, 2, leaf(1)), flm(Fa, 3, Fa0), faki(Fa), write(' '), fail.
(@3 -- (@2 -- @1)) ((@2 -- @1) -- @3) ((@3 -- @2) -- @1) ((@2 -- @3) -- @1)
(@2 -- (@3 -- @1)) (@2 -- (@1 -- @3)) (@3 -- (@1 -- @2)) ((@1 -- @2) -- @3)
((@3 -- @1) -- @2) ((@1 -- @3) -- @2) (@1 -- (@3 -- @2)) (@1 -- (@2 -- @3)) ==> no

negylevelu(X, Y, Z, U, Fa) :- % Fa az X, Y, Z, U levelekből áll
    flm(Fa0, Y, leaf(X)), flm(Fa1, Z, Fa0), flm(Fa, U, Fa1).

| ?- findall(Fa, negylevelu(1,3,4,6,Fa), Fak), length(Fak,Db). ==> Db = 120, Fak = (...)
```

## Példa: adott értékű kifejezés előállítása

---

- A feladat: írjunk Prolog programot a következő feladvány megoldására:
  - Az 1, 3, 4, 6 számokból a négy alpművelet felhasználásával állítsuk elő a 24 számértéket!
  - Mind a négy számot fel kell használni, tetszőleges sorrendben.
  - Tetszőleges alpműveletek használhatók, tetszőleges zárójelezéssel.
- Már van egy predikátumunk (`negylevelu/5`), amely adott számokból tetszőleges fát épít.
- Definiáljunk egy predikátumot, amely egy fának megfelelő aritmetikai kifejezéseket készít!

```
% fa_kif(Fa, Kif): Kif a Fa fával azonos alakú, azonos számokból álló
% aritmetikai kifejezés, amelyben a négy alpművelet fordulhat elő.
fa_kif(leaf(V), V).
fa_kif(node(L,R), Exp) :-
    fa_kif(L, E1),
    fa_kif(R, E2),
    alap4(E1, E2, Exp).

% alap4(X, Y, Kif): Kif az X és Y kifejezésekből a négy alpművelet egyikével áll elő.
alap4(X, Y, X+Y).          alap4(X, Y, X-Y).
alap4(X, Y, X*Y).          alap4(X, Y, X/Y).

| ?- fa_kif(node(leaf(1),node(leaf(2),leaf(3))), Kif).
Kif = 1+(2+3) ? ; Kif = 1-(2+3) ? ; Kif = 1*(2+3) ? ; Kif = 1/(2+3) ? ;
(...)
Kif = 1+2/3 ? ; Kif = 1-2/3 ? ; Kif = 1*(2/3) ? ; Kif = 1/(2/3) ? ; no
```

## Példa: adott értékű kifejezés előállítása (folyt.)

---

- Korábban elkészített predikátumok:

- adott számokból álló fákat felsoroló `negylevelu/5`
- adott fával azonos szerkezetű aritmetikai kifejezéseket felsoroló `fa_kif/2`

- Ezekre építve könnyen megírható a feladvány megoldására használható predikátum:

```
% Kif egy a négy alapművelettel az X, Y, Z, U számokból
% felépített kifejezés, amelynek értéke Ertek.
negylevelu_erteke(X, Y, Z, U, Ertek, Kif) :-
    negylevelu(X, Y, Z, U, Fa),
    fa_kif(Fa, Kif),
    Kif ::= Ertek.

| ?- negylevelu_erteke(1,3,4,6,24,Kif).
...
```

- Megjegyzések

- Az aritmetikai eljárásokban a változók nem csak számokra, hanem tömör aritmetikai kifejezésekre is be lehetnek helyettesítve.
- A `negylevelu_erteke` eljárás utolsó hívása helyett **nem** lenne jó: `Ertek is Kif. Miért?`

Szándékosan üres

---

SZÁNDÉKOSAN ÜRES

## II. RÉSZ



## 4. fejezet: Prolog programozási módszerek

---

- Az előző előadás-blokk (jegyzetbeli 3. fejezet) célja volt:
  - a Prolog nyelv alapjainak bemutatása,
  - a logikailag „tisztá” résznyelvre koncentrálni.
- A jelen előadás-blokk (jegyzetben a 4. fejezet) célja: olyan
  - beépített eljárások,
  - programozási technikákbemutatása, amelyekkel
  - hatékony Prolog programok készíthetők,
  - esetleg a tiszta logikán túlmutató eszközök alkalmazásával.

## Prolog programozási módszerek: tartalomjegyzék

---

- A keresési tér szűkítése
- Vezérlési eljárások
- Determinizmus és indexelés
- A Prolog megvalósítási módszereiről
- Jobbrekurzió és akkumulátorok
- Megoldásgyűjtő eljárások
- Meta-logikai eljárások
- Algoritmusok Prologban
- Megoldások gyűjtése és felsorolása
- Modularitás
- Magasabbrendű eljárások
- Dinamikus adatbáziskezelés
- Nyelvtani elemzés
- „Hagyományos” beépített eljárások



# A KERESÉSI TÉR SZŰKÍTÉSE

## Prolog nyelvi eszközök a keresési tér szűkítésére

---

### • Eszközök

- a vágó beépített eljárás: ! (az első Prolog rendszerektől kezdve)
- feltételes diszjunktív szerkezet (későbbi kiterjesztés): ( if -> then ; else )

### • Feltételes szerkezet — procedurális szemantika (ismétlés)

A (felt->akkor;egyébként), folyt célsorozat végrehajtása:

- Végrehajtjuk a felt hívást (egy önálló végrehajtási környezetben).
- Ha felt sikeres, akkor az akkor, folyt célsorozatra redukáljuk a fenti célsorozatot, a felt **első** megoldása által eredményezett behelyettesítésekkel. A felt cél **többi megoldását nem keressük meg**.
- Ha felt sikertelen, akkor az egyébként, folyt célsorozatra redukáljuk.

### • Feltételes szerkezet — alternatív procedurális szemantika:

- A feltételes szerkezetet egy speciális diszjunkciónak tekintjük:

```
(  felt, {vágás}, akkor
  ;  egyébként
)
```

- A {vágás} jelentése: megszünteti a felt-beli választási pontokat, és egyébként választását is letiltja.

## Feltételes szerkezet: választási pontok a feltételben

---

- Eddig a főleg determinisztikus (választásmentes) feltételeket mutattunk.

- Példafeladat: `első_poz_elem(L, P)`: `P` az `L` lista első pozitív eleme.

- Első megoldás, rekurzióval (mérnöki :-))

```
első_poz_elem([EP|_], EP) :- EP > 0.
első_poz_elem([X|L], EP) :- X =< 0, első_poz_elem(L, EP).
```

- Második megoldás, visszalépéses kereséssel (matematikus :-))

```
első_poz_elem(L, EP) :-
    append(Nk, [EP|_], L), EP > 0, \+ van_poz_eleme(Nk).
```

```
van_poz_eleme(L) :- member(P, L), P > 0.
```

- Harmadik megoldás, feltételes szerkezettel (gyorsprogramozás — Prolog hekker :-))

```
első_poz_elem(L, EP) :-
    (    member(EP, L), EP > 0 -> true
    ;    fail                                     % ez a sor elhagyható
    ).
```

- Figyelem: a harmadik megoldás épít a `member/2` felsorolási sorrendjére!

## A vágó eljárás

---

- A vágó beépített eljárás (neve: !) végrehajtása: letiltja a a többi klóz választását és megszünteti az összes választási pontot a klóztörzsben őt megelőző eljáráshívásokban.
- Példák a vágó használatára (lista első pozitív eleme)
  - Mérnöki megoldás:
 

```
első_poz_elem([EP|_], EP) :- EP > 0, !.
első_poz_elem([X|L], EP) :- X =< 0, első_poz_elem(L, EP).
```
  - Prolog hekker megoldása:
 

```
első_poz_elem(L, EP) :- member(EP, L), EP > 0, !.
```
- Miért vágunk le ágakat a keresési térben?
  - mert mi tudjuk, hogy ott nincs megoldás, de a Prolog megvalósítás nem — zöld vágás, szemantikailag „ártalmatlan”
    - (Például, a legtöbb Prolog megvalósítás „nem tudja”, hogy a  $X > 0$  és  $X \leq 0$  feltételek kizárják egymást, lásd indexelés.)
  - ténylegesen eldobunk megoldásokat — vörös vágás, a program jelentését megváltoztatja
    - (Vörös vágás sokszor úgy keletkezik, hogy egy zöld vágót tartalmazó programban a „felesleges” feltételeket elhagyjuk (pl. az  $X =< 0$  feltételt a fenti 2. klózban)

## Példák a vágó eljárás használatára

---

```
% fakt(+N, ?F): N! = F.
fakt(0, 1) :- !.                                     % zöld vágó
fakt(N, F) :- N > 0, N1 is N-1, fakt(N1, F1), F is N*F1.

% last(+L, ?E): L utolsó eleme E. (lists könyvtárbeli)
last([E], E) :- !.                                   % zöld vágó
last(_|L, E) :- last(L, E).

% pozitívak(+L, -P): P az L pozitív elemeiből áll.
pozitívak([], []).
pozitívak([E|Ek], [E|Pk]) :-
    E > 0, !,                                         % vörös vágó
    pozitívak(Ek, Pk).
pozitívak(_|Ek, Pk) :-
    /* \+ _E > 0, */ pozitívak(Ek, Pk).
```

Figyelem: a fenti példák nem tökéletesek, hatékonyabb ill. általánosabban használható változatukat később ismertetjük!

## A vágó definíciója

---

- Segédfogalom

- Egy cél **szülője** az a cél, amelyik az őt tartalmazó klóz fejével illesztődött.
- Pl. a `last([E], E) :- !.` klózbeli vágó szülője lehet a `last([7], X)` hívás.
- A `g(ancestors)` nyomkövetési parancs kiírja a kurrens cél őseit (szülőjét, annak szülőjét stb.)

- A vágó végrehajtása:

- mindig sikerül; és a végrehajtás adott állapotától visszafelé egészen a szülő célig, azt is beleértve, minden választási pontot megszüntet.

- A vágás kétféle választási pontot szüntet meg:

```

r(X) :- s(X), !.    % az s(X)-beli választási pontokat --- a vágót megelőző
                   % cél(ok)nak az első megoldására szorítkozunk
r(X) :- t(X).       % az r(X) többi klózának választását --- a vágót tartalmazó
                   % klóz mellett kötelezzük el magunkat (commit)

```

- A vágó szemléltetése a 4-kapus doboz modellben: a vágó **Újra** kapujából egyenesesen a körülvető (**szülő**) doboz **Meghiúsulási** kapujára megyünk.

## A vágó által megszüntetett választási pontok

% vágó nélküli példa

q(X):- s(X).

q(X):- t(X).

% ugyanaz a példa vágóval

r(X):- s(X), !.

r(X):- t(X).

s(a).          s(b).          t(c).

% a vágó nélküli példa futása


:- q(X), write(X), fail.

--->          abc

% a vágót tartalmazó példa futása

:- r(X), write(X), fail.

--->          a



## A diszjunktív feltételes szerkezet megvalósítása a vágó segítségével

- A diszjunktív feltételes szerkezet, a diszjunkcióhoz hasonlóan egy segédeljárással váltható ki:

<pre> p :-     ...     ( felt1 -&gt; akkor1     ; felt2 -&gt; akkor2     ; ...     ; egyébként     )     ... </pre>	$\Rightarrow$	<pre> p :-     ...     segéd(...)     ...     segéd(...) :- felt1, !, akkor1.     segéd(...) :- felt2, !, akkor2.     ...     segéd(...) :- egyébként. </pre>
---	---------------	---

- Az egyébként alternatíva elmaradhat, ilyenkor a megfelelő klóz is elmarad.
- SICStus módban a felt részekben vágó nem lehet, ISO módban lehet, de hatásköre (szülője) a felt rész.
- Az akkor részekben lehet vágó. Ennek hatásköre, a  $\rightarrow$  nyílból generált vágóval ellentétben, a teljes p predikátum (ilyenkor a Prolog megvalósítás egy speciális, ún. távolbaható vágót használ).
- Vágót rendkívül ritkán szükséges feltételes szerkezetben szerepeltetni.



## Példák a diszjunktív feltételes szerkezet használatára

---

```
% fakt(+N, ?F):  $N! = F$ .
fakt(N, F) :-
    (    N = 0 -> F = 1
    ;    N > 0, N1 is N-1, fakt(N1, F1), F is N*F1
    ).

% last(+L, ?E): az L nem üres lista utolsó eleme E.
last([E|L], Last) :-
    (    L = [] -> Last = E
    ;    last(L, Last)
    ).

% pozitívak(+L, ?Pk): Pk az L pozitív elemeiből áll.
pozitívak([], []).
pozitívak([E|Ek], Pk) :-
    (    E > 0 -> Pk = [E|Pk0]
    ;    Pk = Pk0
    ),
    pozitívak(Ek, Pk0).
```

## A vágás első alapesete — klóz mellett való elkötelezés

---

- A klóz melletti elkötelezés általában egyszerű feltételes szerkezetet jelent.

szülő :- feltétel, !, akkor.

szülő :- egyébként.

- A vágó szükségtelenné teszi a feltétel negációjának végrehajtását a többi klózban. A logikailag tiszta, de nem hatékony alak:

szülő :- feltétel, akkor.

szülő :- \+ feltétel, egyébként.

A fenti két alak csak akkor ekvivalens, ha feltétel egyszerű, nincs benne választás.

- Analógia: ha  $a$ ,  $b$  és  $c$  Boole-értékű változók, akkor

$\text{if } a \text{ then } b \text{ else } c \equiv a \wedge b \vee \neg a \wedge c$

- A vágó által kiváltott negált feltételt célszerű kommentként jelezni:

szülő :- feltétel, !, akkor.

szülő :- /\* \+ feltétel, \*/ egyébként.

## Feltételes szerkezetek

---

### Feltételes szerkezet — példa

```
% abs(X, A): A az X abszolút értéke.
abs(X, A)    :- X < 0, !, A is -X.
abs(X, X) /* :- X >= 0 */.
```

### Diszjunktív feltételes szerkezet

```
abs(X, A) :-
    (   X < 0 -> A is -X
    ;   A = X
    ).
```

### Általános alak

```
p :- felt1, !, akkor1.
p :- felt2, !, akkor2.
...
p :- egyébként.
```

### Általános alak

```
p :-
    (   felt1 -> akkor1
    ;   felt2 -> akkor2
    ;   ...
    ;   egyébként
    ).
```

## Feltételes szerkezetek és fejillesztés

---

- Vigyázat: a tényleges feltétel részét képezik a fejbeli egyesítések!

% a vágó előtt fej-egyesítés:	% az egyesítés explicitté téve:
<code>abs(X, X) :- X &gt;= 0, !.</code>	<code>abs(X, A) :- A = X, X &gt;= 0, !.</code>
<code>abs(X, A) :- A is -X.</code>	<code>abs(X, A) :- A is -X.</code>

- A fej-egyesítés gondot okozhat, ha az eljárást ellenőrzésre használjuk:

```
| ?- abs(10, -10). ---> yes
```

- A megoldás a **vágás alapszabálya**:

- A kimenő paraméterek értékadását mindig a vágó után végezzük!

```
abs(X, A) :- X >= 0, !, A = X.
abs(X, A) :- A is -X.
```

- Ez nemcsak általánosabban használható, hanem hatékonyabb kódot is ad: csak akkor helyettesíti be a kimenő paramétert, ha már tudja, mi az értéke (nincs „előre-behelyettesítés”, mint a fenti első két példában).
- („**kimenő**” paraméterek — vágó alkalmazásakor általában nincs többirányú használat :-)

## A bevezető példának a vágás alapszabályát betartó változata

---

```
% fakt(+N, ?F):  $N! = F$ .  
fakt(0, F) :- !, F = 1.  
fakt(N, F) :-  $N > 0$ , N1 is N-1, fakt(N1, F1), F is N*F1.
```

```
% last(+L, ?E): az L nem üres lista utolsó eleme E.  
last([E], Last) :- !, Last = E.  
last(_|L, E) :- last(L, E).
```

```
% pozitívak(+L, ?Pk): Pk az L pozitív elemeiből áll.  
pozitívak([], []).  
pozitívak([E|Ek], Pk) :-  
    E > 0, !, Pk = [E|Pk0], pozitívak(Ek, Pk0).  
pozitívak(_|Ek, Pk) :-  
    /* \+ _E > 0, */ pozitívak(Ek, Pk).
```

**Megjegyzés:** a diszjunktív alakban a feltételek eleve explicitek, nincs fejillesztési probléma, ezért **a diszjunktív feltételes szerkezet használatát javasoljuk a vágó helyett.**

## Példasor: $\text{max}(X, Y, Z)$ : $X$ és $Y$ maximuma $Z$ .

---

- 1. változat, tiszta Prolog. Lassú (előre-behelyettesítés, két hasonlítás), választási pontot hagy.

$\text{max}(X, Y, X) :- X \geq Y.$

$\text{max}(X, Y, Y) :- Y > X.$

- 2. változat, zöld vágóval. Lassú (előre-behelyettesítés, két hasonlítás), nem hagy választási pontot.

$\text{max}(X, Y, X) :- X \geq Y, !.$

$\text{max}(X, Y, Y) :- Y > X.$

- 3. változat, vörös vágóval. Gyorsabb (előre-behelyettesítés, egy hasonlítás), nem hagy választási pontot, de nem használható ellenőrzésre, pl. `| ?- max(10, 1, 1)` sikerül.

$\text{max}(X, Y, X) :- X \geq Y, !.$

$\text{max}(X, Y, Y).$

- 4. változat, vörös vágóval. Helyes, nagyon gyors (egy hasonlítás, nincs előre-behelyettesítés) és nem is hoz létre választási pontot.

$\text{max}(X, Y, Z) :- X \geq Y, !, Z = X.$

$\text{max}(X, Y, Y) /* :- Y > X */.$

## A vágás második alapesete — első megoldásra való megszorítás

---

- Mikor használjuk az első megoldásra megszorító vágót?
  - behelyettesítést nem okozó, eldöntendő kérdés esetén;
  - feladatspecifikus optimalizálásra;
  - végtelen választási pontot létrehozó eljárások hasznosítására.

- Eldöntendő kérdés: eljáráshívás csupa bemenő paraméterrel

```
% van_elég_hosszú_út(+N, +A, +B, +Min):  
% A és B között van N lépéses út,  
% amelynek összhossza legalább Min km.  
van_elég_hosszú_út(N, A, B, Min) :-  
    útvonal(N, A, B, Hossz), Hossz >= Min, !.
```

- Eldöntendő kérdés esetén általában nincs értelme többszörös választ adni/várni.

## Feladatspecifikus optimalizálás

---

- A feladat: megkeresendő egy lista elején álló „plató” hossza (platónak hívjuk a csupa azonos elemből álló folytonos részlistát).

```
% Az L lista első eleme H-szor ismétlődik
% a lista kezdőszeleteként.
kezdethossz(L, H) :-
    L = [E|_], append(Ek, Farok, L),
    \+ Farok = [E|_], !,                % vörös vágó
    /* egyformák(Ek, E), */
    length(Ek, H).

/*
% egyformák(Ek, E): Az Ek lista minden eleme E.
egyformák([], _).
egyformák([E|Ek], E) :-
    egyformák(Ek, E).
*/
| ?- kezdethossz([1,1,1,2,3,5], H).
H = 3 ? ; no
```



## Végtelen választás megszelidítése: memberchk (lists könyvtár)

---

### ● memberchk/2 definíciója:

% memberchk(X, L): "X eleme az L listának" kérdés első megoldása.

% 1. változat

```
memberchk(X, L):-
    member(X, L), !.
```

% 2. ekvivalens változat

```
memberchk(X, [X|_]) :- !.
memberchk(X, [_|L]) :-
    memberchk(X, L).
```

### ● memberchk/2 használata

#### ● Eldöntő kérdésben (visszalépéskor nem keresi végig a lista maradékát.)

```
| ?- memberchk(1, [1,2,3,4,5,6,7,8,9]).
```

#### ● Nyílt végű lista elemévé tesz, pl.:

```
| ?- memberchk(1,L), memberchk(2,L), memberchk(1,L).
    L = [1,2|_A] ?
```

## Nyílt végű listák kezelése memberchk segítségével: szótárprogram

---

```
szótaraz(Sz):-
    read(M-A), !,
    % A read(X) beépített eljárás egy kifejezést
    % olvas be és egyesíti X-szel
    memberchk(M-A,Sz),
    write(M-A), nl,
    szótaraz(Sz).
```

szótaraz(\_).

Egy futása:

```
| ?- szótaraz(Sz).
|: alma-apple.           |: alma-X.
alma-apple               alma-apple
|: korte-pear.           |: X-pear.
korte-pear               korte-pear
|: vege.                 % nem egyesíthető M-A-val
```

```
Sz = [alma-apple,korte-pear|_A] ?
```

# VEZÉRLÉSI ELJÁRÁSOK



## Vezérlési eljárások, a `call/1` beépített eljárás

---

- Vezérlési eljárás: A Prolog végrehajtáshoz kapcsolódó beépített eljárás (pl. vágó, if-then-else).
- A vezérlési eljárások többsége **magasabbrendű** eljárás, azaz olyan eljárás, amely egy vagy több argumentumát eljáráshívásként értelmezi. (A magasabbrendű Prolog eljárásokat szokás **meta-eljárásnak** is hívni.)
- A meta-eljárások fő képviselője és alapvető építőeleme a `call/1`:
  - Hívási minta: `call(+Cél)`
  - Cél egy „meghívható kifejezés” (callable, vö. `callable/1`), azaz struktúra, vagy névkonstans.
  - Jelentése (deklaratív szemantika): Cél igaz.
  - Hatása (procedurális szemantika): a Cél kifejezést eljáráshívássá alakítja és meghívja.
- A klóztörzsben célként megengedett egy `X` változó használata, ezt a rendszer egy `call(X)` hívássá alakítja át.

```
| kétszer(Hívás) :- call(Hívás), Hívás.
```

```
| ?- kétszer(write(ba)), nl.          ---> baba
```

```
| ?- listing(kétszer).               ---> kétszer(A) :-
                                         call(user:A), call(user:A).
```

## Vezérlési szerkezetek mint eljárások

---

- A `call/1` argumentumában szerepelhetnek vezérlési szerkezetek is, mert ezek maguk beépített eljárásként is jelen vannak a Prolog rendszerben:
  - `(', ')/2`: konjunkció.
  - `(;)/2`: diszjunkció.
  - `(->)/2`: if-then.
  - `(;)/2`: if-then-else.
- A `call`-ban szereplő vezérlési szerkezetek lényegében ugyanúgy futnak, mint az interpretált (`consult`-tal betöltött) kód.
- Példák:

```
| ?- _Cél = (kétszer(write(ba)), write(' ')), kétszer(_Cél), nl.  
baba baba  
| ?- kétszer((member(X, [a,b,c,d]), write(X), fail ; nl)).  
abcd  
abcd
```

## call/1 példa: futási időt mérő meta-eljárás

---

```
% Kiírja Goal első megoldásának előállításához vagy a megghiúsuláshoz
% szükséges időt, a Txt szöveg kíséretében (lásd: peldak/call_koltsege.pl).
time(Txt, Goal) :-
    statistics(runtime, [T0,_]), % T0 az indítás óta eltelt CPU idő,
                                % msec-ban (szemétgyűjtés nélkül).
    (
        call(Goal) -> Res = true
    ;   Res = false
    ),
    statistics(runtime, [T1,_]), T is T1-T0,
    format('~w futási idő = ~3d sec\n', [Txt,T]),
           % ~w formázó: kiírás a write/1 segítségével
           % ~3d formázó: I egész kiírása I/1000-ként, 3 tizedesre
    Res = true.
```

A call/1 viszonylag költséges: egy 1414 hosszú lista megfordítása nrev-vel (kb. 1 millió append hívás), minden append körül egy felesleges call-lal ill. anélkül:

	call nélkül	call-lal	Lassulás
lefordítva	0.140 sec	1.680 sec	12.00
interpretálva	1.710 sec	3.520 sec	2.06

## További beépített vezérlési eljárások

---

- `\+` Cél: Cél „nem bizonyítható”. A beépített eljárás definíciója:

```
\+ X :- call(X), !, fail.  
\+ _X.
```

- `once(Cél)`: Cél igaz, és csak az első megoldását kérjük. Definíciója:

```
once(X) :- call(X), !.
```

- `true`: azonosan igaz (mindig sikerül), `fail`: azonosan hamis (mindig meghiúsul).

- `repeat`: végtelen sokszor igaz (egy végtelen választási pontot hoz létre). Definíciója:

```
repeat.  
repeat :- repeat.
```

- A `repeat` eljárást legtöbbször egy mellékhatásos eljárás ismétlésére használjuk. A végtelen választási pontot kötelező egy vágóval semlegesíteni.

- Példa (egyszerű kalkulátor):

```
bc :- repeat, read(Expr),  
      (  
        Expr = end_of_file -> true  
        ; Res is Expr, write(Expr = Res), nl, fail  
      ),  
      !.
```

## Példa: magasabbrendű reláció definiálása

---

- Az implikáció ( $P \Rightarrow Q$ ) megvalósítása negáció segítségével:

```
% P minden megoldása esetén Q igaz.
forall(P, Q) :-
    \+ (P, \+Q). % Szintaktikus emlékeztető:
                % az első \+ után kötelező a szóköz!

| ?- _L = [1,2,3],
    % _L minden eleme pozitív:
    forall(member(X, _L), X > 0).
true ?

| ?- _L = [1,-2,3], forall(member(X, _L), X > 0).
no

| ?- _L = [1,2,3],
    % _L szigorúan monoton növő:
    forall(append(_, [A,B|_], _L), A < B).
true ?
```

- forall/2 csak eldöntendő kérdés esetén használható.



# DETERMINIZMUS ÉS INDEXELÉS

# Determinizmus

- Egy eljáráshívás **determinisztikus**, ha (legfeljebb) egyféleképpen sikerülhet.
- Egy eljáráshívásnak egy sikeres végrehajtása **determinisztikusan futott le**:
  - ha nem hagyott választási pontot a híváshoz tartozó részében, azaz
    - vagy választásmentesen futott le, azaz létre sem hozott választási pontot (figyelem: ez a Prolog megvalósítástól függ!);
    - vagy létrehozott ugyan választási pontot, de megszüntette (kimerítette, levágta).
- A SICStus Prolog nyomkövetésében ? jelzi a **nem**determinisztikus lefutást:

p(1, a).		?- p(1, X).	% det. hívás,
p(2, b).		1 1 Exit: p(1,a)	% det. lefutás
p(3, b).		?- p(Y, a).	% det. hívás,
		? 1 1 Exit: p(1,a)	% nemdet. lefutás
		?- p(Y, b), Y > 2.	% nemdet. hívás
		? 1 1 Exit: p(2,b)	% nemdet. lefutás
		1 1 Exit: p(3,b)	% det. lefutás

## A determinisztikus lefutás

---

- Mi a determinisztikus lefutás haszna?
  - a futás gyorsabb lesz,
  - a tárigény csökken,
  - más optimalizálások (pl. jobbrekurzió) alkalmazható.
- Hogyan ismeri fel a fordító azt, hogy nem kell választási pontot?
  - indexelés (indexing)
  - vágók és feltételes szerkezetek
- Az alábbi definíciók esetén a  $p(\textit{Nonvar}, Y)$  hívás nem hoz létre választási pontot:

$p(1, a).$ $p(2, b).$	$p(1, Y) :- !,$ $\quad Y = a.$ $p(\_, b).$	$p(X, Y) :-$ $\quad ( \quad X ::= 1 \rightarrow Y = a$ $\quad ; \quad Y = b$ $\quad ).$
--------------------------	--	--

## Indexelés — ismétlés

---

- Mi az indexelés?
  - egy adott hívásra illeszthető klózik gyors kiválasztása,
  - egy eljárás klózikainak fordítási idejű csoportosításával.
- A legtöbb Prolog rendszer, így a SICStus Prolog is, az első fej-argumentum alapján indexel (first argument indexing).
- Az indexelés alapja az első fejargumentum külső funktora:
  - C szám vagy névkonstans esetén C/0;
  - R nevű és N argumentumú struktúra esetén R/N;
  - változó esetén nem értelmezett.
- Az indexelés megvalósítása:
  - Fordításkor a funktorokhoz elkészítjük az illeszthető klózik részhalmazát.
  - Futáskor lényegében konstans idő alatt választunk a részhalmazok közül.
  - **Fontos:** ha egyelemű a részhalmaz, nem hozunk létre választási pontot!

## Példa indexelésre

p(0, a).	/* (1) */	q(1).
p(X, t) :- q(X).	/* (2) */	q(2).
p(s(0), b).	/* (3) */	
p(s(1), c).	/* (4) */	
p(9, z).	/* (5) */	

● A  $p(A, B)$  hívással illesztendő klózhalmaz:

- $\{(1) (2) (3) (4) (5)\}$  ha A változó;
- $\{(1) (2)\}$  ha  $A = 0$ ;
- $\{(2) (3) (4)\}$  ha A fő funktora  $s/1$ ;
- $\{(2) (5)\}$  ha  $A = 9$ ;
- $\{(2)\}$  minden más esetben.

● Példák hívásokra:

- $p(1, Y)$  nem hoz létre választási pontot.
- $p(s(1), Y)$  létrehoz választási pontot, de determinisztikusan fut le.
- $p(s(0), Y)$  nemdeterminisztikusan fut le.

## Struktúrák, változók a fejargumentumban

### Azonos funktorú struktúrák az első fejargumentumban:

- Ha a klózek szétválasztásához szükség van az első (struktúra) argumentum részeire is, akkor érdemes segédjeljárást bevezetni.
- Például  $p/2$  és  $q/2$  ekvivalens, de  $q(\text{Nonvar}, Y)$  determinisztikusan fut le!

$p(0, a).$	$q(0, a).$	$q\_seged(0, b).$
$p(s(0), b).$	$q(s(X), Y) :-$	$q\_seged(1, c).$
$p(s(1), c).$	$q\_seged(X, Y).$	
$p(9, z).$	$q(9, z).$	

### Fejlesztés kiváltása egyenlőséggel (vö. rétegelt minta)

- Az indexelés figyelembe veszi a törzs elején szereplő egyenlőséget:  
 $p(X, \dots) :- X = \text{Kif}, \dots$  esetén Kif funktora szerint indexel.
- Példa: lista hosszának reciproka, üres lista esetén 0:

```
rhossz([], 0).
rhossz(L, RH) :- L = [_|_], length(L, H), RH is 1/H.
% rhossz([X|L], RH) :- length([X|L], H), RH is 1/H.
% kevésbé hatékony, mert újra felépíti az [X|L] listát.
% rhossz(L, RH) :- L \= [], length(L, H), RH is 1/H.
% kevésbé hatékony, mert L=[] esetben választási pontot hagy.
```

## Indexelés — további tudnivalók

---

### ● Indexelés és aritmetika

- Az indexelés nem foglalkozik aritmetikai vizsgálatokkal.
- Pl. az  $N = 0$  és  $N > 0$  feltételek nem „zárják ki” egymást.
- Az alábbi  $\text{fakt}/2$  eljárás lefutása nem-determinisztikus:

$\text{fakt}(0, 1).$

$\text{fakt}(N, F) :- N > 0, N1 \text{ is } N-1, \text{fakt}(N1, F1), F \text{ is } N * F1.$

### ● Indexelés és listák

- Gyakran kell az üres és nem-üres lista esetét szétválasztani.
- A bemenő lista-argumentumot célszerű az első argumentum-pozícióba tenni.
- Az  $[]$  és  $[\dots | \dots]$  eseteket az indexelés megkülönbözteti (funktoruk:  $'[]'$  / 0 ill.  $'.'$  / 2).
- A két klóz sorrendje nem érdekes (feltéve, hogy zárt listával hívjuk az első pozíción) — de azért tegyük a leálló klózt mindig előre.

## Listakezelő eljárások indexelése: példák

---

- Az `append/3` választásmentesen fut le, ha első argumentuma zárt végű lista.

```
append([], L, L).
append([X|L1], L2, [X|L3]) :- append(L1, L2, L3).
```

- A `last/2` közvetlen megfogalmazása nemdeterminisztikusan fut le:

```
% last(L, E): Az L lista utolsó eleme E.
last([E], E).
last(_|L, E) :- last(L, E).
```

- Érdemes segédeljárást bevezetni, `last2/2` választásmentesen fut

```
last2([X|L], E) :- last2(L, X, E).

% last2(L, X, E): Az [X|L] lista utolsó eleme E.
last2([], E, E).
last2([X|L], _, E) :- last2(L, X, E).
```

- Az utolsó listaelemet választásmentesen felsoroló `member` (`lists` könyvtárból):

```
member(E, [H|T]) :- member_(T, H, E).

% member_(L, X, E): Az [X|L] lista eleme E.
member_(_, E, E).
member_([H|T], _, E) :- member_(T, H, E).
```



## Az indexelés és a vágó kölcsönhatása

- Hogyan vehető figyelembe a vágó az indexelés fordításakor?

- Példa: a  $p(1, A)$  hívás választásmentes, de a  $q(1, A)$  nem!

$p(1, Y) :- !, Y = 2. \quad \% (1)$	$q(1, 2) :- !. \quad \% (1)$
$p(X, X). \quad \% (2)$	$q(X, X). \quad \% (2)$
$\text{Arg1}=1 \rightarrow (1), \text{Arg1} \neq 1 \rightarrow (2)$	$\text{Arg1}=1 \rightarrow \{(1), (2)\}, \text{Arg1} \neq 1 \rightarrow (2)$

- A fordító figyelembe veszi a vágót az indexelésben, ha garantált, hogy egy adott fő funktor esetén a vágót elérjük. Ennek feltételei:

- az első argumentum változó, konstans, vagy csak változókat tartalmazó struktúra legyen,
- a további argumentumok változók legyenek,
- a fejben az összes változóelőfordulás különböző legyen,
- a törzs első hívása a vágó (megengedve a fejillesztést kiváltó egyenlőséget).

- Ilyenkor a fordító az adott funktorhoz tartozó listából kihagyja a vágót követő klózokat.

- Példa:  $p(X, D, E) :- X = s(A, B, C), !, \dots \quad p(X, Y, Z) :- \dots$

- Ez egy újabb érv a vágás alapszabálya mellett:

**A kimenő paraméterek értékadását mindig a vágó után végezzük!**

## A vágó és az indexelés hatékonysága

- Egy Fibonacci-szerű sorozat:  $f_1 = 1$ ;  $f_2 = 2$ ;  $f_n = f_{\lfloor 3n/4 \rfloor} + f_{\lfloor 2n/3 \rfloor}$ ,  $n > 2$

% determinisztikus	% determ. lefutású	% választásmentes
fib(1, 1).	fibc(1, 1) :- !.	fibci(1, F) :- !, F = 1.
fib(2, 2).	fibc(2, 2) :- !.	fibci(2, F) :- !, F = 2.
fib(N, F) :-	fibc(N, F) :-	fibci(N, F) :-
N > 2,	N > 2,	N > 2,
N2 is N*3//4,	N2 is N*3//4,	N2 is N*3//4,
N3 is N*2//3,	N3 is N*2//3,	N3 is N*2//3,
fib(N2, F2),	fibc(N2, F2),	fibci(N2, F2),
fib(N3, F3),	fibc(N3, F3),	fibci(N3, F3),
F is F2+F3.	F is F2+F3.	F is F2+F3.

- Futási idők  $N = 2000$  esetén

	fib	fibc	fibci
futási idő	990 msec	890 msec	830 msec
meghiúsulási idő	440 msec	30 msec	0 msec
összesen	1430 msec	920 msec	830 msec
nyom-verem mérete	4.1Mbyte	2.0 Mbyte	256 byte

- fibc esetén a meghiúsulási idő azért nem 0, mert a rendszer a nyom-vermet (trail-stack) dolgozza fel. A nyom-verem tárolja a változó-értékadások visszacsinálási információit.

## Választás-mentesség diszjunktív feltételes szerkezetek esetén

- Feltételes szerkezet végrehajtásakor általában választási pont jön létre.
- A **SICStus Prolog** a „( felt -> akkor ; egyébként )” szerkezetet választásmentesen hajtja végre, ha a felt konjunkció tagjai csak:
  - aritmetikai összehasonlító eljárashívások (pl. <, =<, ==), és/vagy
  - kifejezés-típust ellenőrző eljárashívások (pl. atom, number), és/vagy
  - általános összehasonlító eljárashívások (ld. később, pl. @<, @=<, ==).
- Analóg módon választásmentes kód keletkezik a „fej :- felt, !, akkor.” klózból, ha fej argumentumai különböző változók, és felt olyan mint fent.
- Például választásmentes kód keletkezik az alábbi definíciókból:

```
vektorfajta(X, Y, Fajta) :-
    (   X == 0, Y == 0
      % X = 0, Y = 0 nem lenne jó
    -> Fajta = null
    ;   Fajta = nem_null
    ).
```

```
vektorfajta(X, Y, Fajta) :-
    X == 0, Y == 0, !,
    Fajta = null.
vektorfajta(_X, _Y, nem_null).
```

# A PROLOG MEGVALÓSÍTÁSI MÓDSZEREIRŐL

## A Prolog megvalósítás néhány mérföldköve

---

- 1973: Marseille Prolog (A. Colmerauer et al.)
  - értelmező (interpreter), Fortran nyelven
  - kifejezések ábrázolása: struktúra-osztásos (structure-sharing)
  - veremszervezés: egyetlen verem (csak visszalépéskor szabadul fel)
- 1977: DEC-10 Prolog (D. H. D. Warren)
  - fordítóprogram Prolog és assembly nyelven (+ értelmező Prologban)
  - kifejezések ábrázolása: struktúra-osztásos
  - veremszervezés: három verem (visszalépéskor mindhárom felszabadul)
    - globális verem (global stack): globális (struktúra-beli) változók, szemétgyűjtött
    - fő verem (local stack): eljárások, választási pontok, változók, det. lefutáskor felszabadul
    - nyom verem (trail): változó-behelyettesítések tárolása (vágónál felszabadítható)
- 1983: WAM — Warren Abstract Machine (D. H. D. Warren)
  - absztrakt gép Prolog programok végrehajtására
  - kifejezések ábrázolása: struktúra-másolósos (structure-copying)
  - három verem, mint DEC-10 Prologban, a globális verem tárolja a struktúrákat
  - A legtöbb mai Prolog WAM alapú (SICStus, SWI, GNU Prolog, ...)

## Struktúrák ábrázolása

- A kétféle kifejezés-ábrázolás összehasonlítása:

	struktúra-osztásos	struktúra-másolósos
tárigény:	a változók számával arányos	a struktúra méretével arányos
struktúra-építés időigénye	konstans	a struktúra méretével arányos
struktúra-szétszedés	költségesebb	kevésbé költséges

- Struktúra **építése**: egy változónak és egy **programszövegbeli** struktúrának az egyesítése
- FONTOS: egy változó értékeként megjelenő struktúra egyesítése egy behelyettesíthető változóval mindenképpen konstans költségű!

- Példa:

```

hosszabbít(L, [1,2,3,...,n|L]).
sokszoroz(0, L) :- !, L = [].
sokszoroz(N, L) :-
    hosszabbít(L0, L), N1 is N-1, sokszoroz(N1, L0).

```

- $\text{sokszoroz}(n, L)$  költsége és tárigénye struktúra-osztásnál  $O(n)$ , struktúra-másolásnál  $O(n^2)$
- A gyakorlatban mégis a struktúra-másolósos megoldás bizonyult hatékonyabbnak.

## WAM: Prolog kifejezések tárolása

- A WAM-ban javasolt kifejezés-ábrázolás (LBT: low bit tagging scheme)

	<i>globális/lokális</i>	<i>globális verem</i>
● Behelyettesítetlen változó:	saját cím	REF
● Másik változóra/kifejezésre való utalás:	másik kif. címe	REF
● Névkonstans	atom tábla index	A CON
● Egész szám	egész érték	I CON
● Lista	cím	LIST
	cím:	fej-kifejezés farok-kifejezés
● Struktúra	cím	STRU
	cím:	funktor tábla index argumentum-kif. ...

- A SICStus 3.x rendszer a 4 legmagasabb helyiértékű biten tárolja jelzőket (tag) — ezért a veremterületek mérete 256 Mbyte-ban korlátozott. (SICStus 4-ben már LBT séma lesz.)

## WAM: néhány további részlet

---

- Változók kezelése
  - Két változó egyesítése: a fiatalabbik az öregebbikre utaló **REF** értéket kap
  - **Utalástalanítás**: az (esetleg többtagú) REF-lánc követése
  - Behelyettesíthetetlen változó  $\equiv$  önmagára mutató utalás  $\Rightarrow$  egyszerűbb utalástalanítás
- Visszalépés
  - **Feltételes változó**: behelyettesíthetetlen változó, öregebb mint a legfrissebb választási pont
  - Feltételes változó behelyettesítése esetén a változó címét beírjuk a nyom-verembe
  - Visszalépéskor a nyom alapján „visszacsináljuk” a változó-behelyettesítéseket, majd a vermeket visszahúzzuk
- SICStus programok WAM utasítás-sorozatra fordíthatók (*File.pl*  $\Rightarrow$  *File.wam*):  

```
| ?- prolog:fshell_files(File, wam, []).
```
- A WAM bemutatása (tutorial): <http://www.vanx.org/archive/wam/wam.html>



# JOBBREKURZIÓ ÉS AKKUMULÁTOROK

## Jobbrekurzió (farok-rekurzió, tail-recursion) optimalizálás

---

- Az általános rekurzió költséges, helyben és időben is.
- Jobbrekurzióról beszélünk, ha
  - a rekurzív hívás a klóztörzs utolsó helyén van, vagy az utolsó helyen szereplő diszjunkció egyik ágának utolsó helyén stb., és
  - a rekurzív hívás pillanatában nincs választási pont a predikátumban (a rekurzív hívást megelőző célok determinisztikusan futottak le, nem maradt nyitott diszjunkciós ág).
- Jobbrekurzió optimalizálás: az utolsó hívás végrehajtása **előtt** a predikátum által lefoglalt hely felszabadul ill. szemétgyűjtésre alkalmassá válik.
- Ez az optimalizálás nemcsak rekurzív hívás esetén, hanem minden **utolsó** hívás esetén megvalósul — a pontos név: utolsó hívás optimalizálás (last call optimisation).
- A jobbrekurzió így tehát nem növeli a memória-igényt, korlátlan mélységig futhat — mint a ciklusok az imperatív nyelvekben. Példa:

```
ciklus(Állapot) :- lépés(Állapot, Állapot1), !, ciklus(Állapot1).  
ciklus(_Állapot).
```

## Predikátumok jobbrekurzív alakra hozása — listaösszeg

---

- A listaösszegzés „természetes”, nem jobbrekurzív definíciója:

```
% sum(+L, ?S): Az L számlista elemeinek összege S ( $S = 0 + L_n + L_{n-1} + \dots + L_1$ ).
sum([], 0).
sum([X|L], S):- sum(L,S0), S is S0+X.
```

- Első jobbrekurzív változat, csak ellenőrzésre használható:

```
% sum1(+L, +S): Az L számlista elemeinek összege S ( $S - L_1 - L_2 - \dots - L_n = 0$ ).
sum1([], 0).
sum1([X|L], S) :- /* S is S0+X helyett: */ S0 is S-X, sum1(L, S0).
```

- Második jobbrekurzív változat, csak kiírni tudja az eredményt:

```
% sum2(+L): Az L számlista elemeinek összegét ( $0 + L_1 + L_2 + \dots + L_n$ ) kiírja.
sum2(L):- sum2(L, 0).
```

```
% sum2(+L, +S0): Az L lista S0-lal növelt összegét kiírja.
sum2([], S) :- write(S), nl.
sum2([X|L], S0):- S1 is S0+X, sum2(L, S1).
```

- Ahhoz, hogy az összeget **eredményként** ki tudjuk adni, szükséges egy további, kimenő argumentum.

## Jobbrekurzív listaösszeg — akkumulátorpár segítségével

---

- Harmadik változat: teljes értékű jobbrekurzív lista-összegző:

```
% sum3(+L, ?S): Az L számlista elemeinek összege S.
sum3(L, S):- sum3(L, 0, S).
```

```
% sum3(+L, +S0, ?S): L elemeit hozzáadva S0-hoz kapjuk S-et. ( $\equiv \sigma L = S - S0$ )
sum3([], S, S).
sum3([X|L], S0, S):-
    S1 is S0+X, sum3(L, S1, S).
```

- A jobbrekurzív `sum3` eljárás több mint **3-szor gyorsabb** mint a nem jobbrekurzív `sum`!
- Az **akkumulátor** az imperatív (azaz megváltoztatható értékű) változó fogalmának deklaratív megfelelője:
  - A `sum3(L, S0, S)` predikátumban az `S0` és `S` argumentumok egy akkumulátorpárt alkotnak.
  - Az akkumulátorpár két része az adott változó mennyiség (a példában az összeg) különböző időpontokban vett értékeit mutatja:
    - `S0` az összeg értéke a `sum3/3` **meghívásakor**: az összegző változó kezdőértéke;
    - `S` az összeg értéke a `sum3/3` **lefutása után**: összegző változó végértéke.

## Az akkumulátorok használata

---

- Az akkumulátorokkal általánosan több egymás utáni változtatást is leírhatunk:

```
p(..., A0, A) :-  
    q0(..., A0, A1), ...,  
    q1(..., A1, A2), ...,  
    qn(..., An, A).
```

- A sum3/3 második klóza ilyen alakra hozva:

```
sum3([X|L], S0, S) :- plus(X, S0, S1), sum3(L, S1, S).
```

```
plus(X, S0, S) :- S is S0+X.
```

- Akkumulátorváltozók elnevezési konvenciója: kezdőérték: *Vál t0*; közbülső értékek: *Vál t1*, ..., *Vál tn*; végérték: *Vál t*.
- A Prolog akkumulátorpár nem más mint a funkcionális programozásból ismert gyűjtőargumentum és a függvény eredményének együttese.

## Akkumulátorok használata — folytatás

---

### ● Három lista összege

```
% sum_3_lists(+L, +LL, +LLL, +S0, ?S): Az L, LL, LLL számlisták
% összegeinek összege S-S0
sum_3_lists(L, LL, LLL, S0, S) :-
    sum3(L, S0, S1), sum3(LL, S1, S2), sum3(LLL, S2, S).
```

Előrebocsátott megjegyzés: a fenti szabály DCG (Definite Clause Grammar) formája

```
sum_3_lists(L, LL, LLL) --> sum3(L), sum3(LL), sum3(LLL).
```

### ● Többszörös akkumulálás — listák összege és négyzetösszege

```
% sum12(+L, +S0, ?S, +Q0, ?Q):  $S-S0 = \sum Li$ ,  $Q-Q0 = \sum Li*Li$ 
sum12([], S, S, Q, Q).
sum12([X|L], S0, S, Q0, Q) :-
    S1 is S0+X, Q1 is Q0+X*X, sum12(L, S1, S, Q1, Q).
```

### ● Többszörös akkumulátorok összevonása

```
% sum12(+L, +S0/Q0, ?S/Q):  $S-S0 = \sum Li$ ,  $Q-Q0 = \sum Li*Li$ 
sum12([], SQ, SQ).
sum12([X|L], S0/Q0, SQ) :-
    S1 is S0+X, Q1 is Q0+X*X, sum12(L, S1/Q1, SQ).
```

## Különbséglisták

---

- A revapp mint akkumuláló eljárás

```
% revapp(Xs, L0, L): Xs megfordítását L0 elé fűzve kapjuk L-t.
% Másképpen: Xs megfordítása L-L0.
revapp([], L, L).
revapp([X|Xs], L0, L) :-
    L1 = [X|L0], revapp(Xs, L1, L).
```

- Az L-L0 jelölés (különbséglista): azt a listát nevezi meg, amelyet úgy kapunk, hogy L végéről elhagyjuk L0-t (feltéve, hogy L0 szuffixuma L-nek).

- Például az [1,2,3] listának megfelelő különbséglisták:

- [1,2,3,4] - [4], [1,2,3,a,b] - [a,b], [1,2,3] - [], ...

- A legáltalánosabb (nyílt) különbséglistában a „kivonandó” változó: [1,2,3|L] - L

- Egy nyílt különbséglista konstans időben összefűzhető egy másikkal:

```
% app_dl(DL1, DL2, DL3): DL1 és DL2 különbséglisták összefűzése DL3.
app_dl(L-L0, L0-L1, L-L1).
| ?- app_dl([1,2,3|L0]-L0, [4,5|L1]-L1, DL).
    => DL = [1,2,3,4,5|L1]-L1, L0 = [4,5|L1]
```

- A nyílt különbséglista „egyszer használatos”, egy hozzáfűzés után már nem lesz nyílt!

## Különbséglisták (folyt.)

---

- Példa: lineáris idejű listafordítás, `nrev` stílusában, különbséglistával:

```
% nrev(L, DR): Az L lista megfordítása a DR különbséglista.
nrev_dl([], L-L).           % L-L  $\equiv$  üres különbséglista
nrev_dl([X|L], DR) :-
    nrev_dl(L, DR0),
    app_dl(DR0, [X|T]-T, DR). % [X|T]-T  $\equiv$  egyelemű különbséglista

% app_dl(DL1, DL2, DL3): DL1 és DL2 különbséglisták összefűzése DL3.
app_dl(L-L0, L0-L1, L-L1).

% Az L lista megfordítása R
rev(L, R) :-
    nrev_dl(L, R-[]).
```

- Az `nrev_dl/2` eljárás törzsében érdemes a két hívást megcserélni (jobbrekurzió!).
- `nrev_dl(L, R-R0)  $\implies$  rev2(L, R0, R)` átalakítással és `app_dl` kiküszöbölésével a fenti `nrev_dl/2` eljárásból kapunk egy `rev2/3`-t, amely azonos `revapp/3`-mal!
- Ettől az átalakítástól kb **3-szor gyorsabb** lesz a program  $\implies$  érdemes a különbséglisták helyett akkumulátorpárokat használni!
- A továbbiakban a különbséglista jelölést csak a fejkommentek megfogalmazásában használjuk.



## Az append mint akkumuláló eljárás

---

- Írjunk egy `eleje_marad(Eleje, L, Marad)` eljárást!

```
% eleje_marad(Eleje, L, Marad): Az L lista kezdetén az Eleje lista áll,  
% annak L-ből való elhagyása után marad a Marad lista.  
eleje_marad([], L, L).  
eleje_marad([X|Xs], L0, L) :-  
    L0 = [X|L1],  
    eleje_marad(Xs, L1, L).
```

- Az akkumulálási lépés:  $L0 = [X|L1]$ , egy elem **elhagyása** a lista elejéről.
- A 2. és 3. argumentum felcserélésével az `eleje_marad` eljárás átalakul az `append` eljárássá!
- Tehát az `append` is tekinthető akkumuláló eljárásnak (a 2. és 3. argumentum a szokásos akkumulátorpárokhoz képest fel van cserélve):

```
% append(Xs, L, L0): L0 elejéről Xs elemeit hagyva marad L.  
% Másképpen: Xs = L0-L.  
append([], L, L).  
append([X|Xs], L, L0) :-  
    L0 = [X|L1], append(Xs, L, L1).
```

- Az akkumulálási lépés: az `L0` változó értékül kap egy listát, melynek farka `L1`, az akkumulálált mennyiség: az a változó, amelyben az összefűzés eredményét várjuk.

## Egy mintafeladat: $a^n b^n$ alakú sorozat előállítása

---

### ● Első megoldás, $3n$ lépés

```
% anbn(N, L): Az L lista N db a-ból
% és azt követő N db b-ből áll.
anbn(N, L) :-
    an(N, a, AN),
    an(N, b, BN),
    append(AN, BN, L).

% an(N, A, L): L az A elemet N-szer
% tartalmazó lista
an(0, _A, L) :- !, L = [].
an(N, A, [A|L]) :-
    N > 0,
    N1 is N-1,
    an(N1, A, L).
```

### ● Második megoldás, $2n$ lépés

```
anbn(N, L) :-
    an(N, b, [], BN),
    an(N, a, BN, L).

% an(N, A, L0, L): L-L0 az A
% elemet N-szer tartalmazó lista
an(0, _A, L0, L) :- !, L = L0.
an(N, A, L0, [A|L]) :-
    N > 0,
    N1 is N-1,
    an(N1, A, L0, L).
```

## $a^n b^n$ alakú sorozatok (folyt.)

---

### ● Harmadik megoldás, $n$ lépés

```
anbn(N, L) :-
    anbn(N, [], L).

% anbn(N, L0, L): Az L-L0 lista N db a-ból és azt követő N db b-ből áll.
anbn(0, L0, L) :- !, L = L0.
anbn(N, L0, [a|L]) :-
    N > 0,
    N1 is N-1,
    anbn(N1, [b|L0], L).
```

### ● A második klóz nem jobbrekurzív változata

```
anbn(N, L0, L) :-
    N > 0, N1 is N-1,
    L1 = [b|L0],           % 1. lépés: L0 elé b => L1
    anbn(N1, L1, L2),      % 2. lépés: L1 elé a^N1 b^N1 => L2
    L = [a|L2].            % 3. lépés: L2 elé a => L
```

## $a^n b^n$ alakú sorozatok — más nyelvű megoldások

---

### ● C++ megoldás

```
link *anbn(unsigned n) {  
    link *l = 0, *b = 0;    // ez elé építjük a b-ket  
    link **a = &l;          // ebbe tesszük az a-kat  
    for (; n > 0; --n) {  
        *a = new link('a'); // előlről  
        a = &(*a)->next;    // hátra épít  
        b = new link('b', b); // hátulról előre épít  
    }  
    *a = b; return l;  
}
```

## Összetettebb adatstruktúrák akkumulálása

---

- Az adatstruktúra:  
`% :- type bfa --> ures ; bfa(int, bfa, bfa).`
- A fa csomópontjaiban tároljuk a számértékeket, a levelek nem tárolnak információt.
- Egészek gyűjtése **rendezett** bináris fában
  - `beszur(BFa0, E, BFa)`: Az E egész számnak a BFa0 fába való beszúrása a BFa bináris fát eredményezi.
  - Itt BFa0 és BFa egy akkumulátorpár, de az indexelés érdekében BFa0 az első argumentum-pozícióba kerül.
- Példafutás:

```
| ?- beszur(ures, 3, Fa0),  
      beszur(Fa0, 1, Fa1),  
      beszur(Fa1, 5, Fa2).
```

```
Fa0 = bfa(3,ures,ures),  
Fa1 = bfa(3,bfa(1,ures,ures),ures),  
Fa2 = bfa(3,bfa(1,ures,ures),bfa(5,ures,ures)) ?
```

## Akkumulálás bináris fákkal

---

### ● Elem beszúrása bináris fába

```
% beszur(BF0, E, BF): E beszúrása BF0 rendezett fába
% a BF rendezett fát adja
% :- pred beszur(bfa::in, int::in, bfa::out).
beszur(ures, Elem, bfa(Elem, ures, ures)).
beszur(BF0, Elem, BF):-
    BF0 = bfa(E,B,J), % az indexelés működik!
    (   Elem == E -> BF = BF0
    ;   Elem < E ->
        BF = bfa(E,B1,J),
        beszur(B, Elem, B1)
    ;   BF = bfa(E,B,J1),
        beszur(J, Elem, J1)
    ).
```

## Akkumulálás bináris fákkal — folyt.

---

### ● Lista konverziója bináris fává

```
% lista_bfa(L, BF0, BF): L elemeit beszúrva BF0-ba kapjuk BF-t.  
% :- pred lista_bfa(list(int)::in, bfa::in, bfa::out).  
lista_bfa([], BF, BF).  
lista_bfa([E|L], BF0, BF):-  
    beszur(BF0, E, BF1),  
    lista_bfa(L, BF1, BF).
```

```
| ?- lista_bfa([3,1,5], ures, BF).  
BF = bfa(3,bfa(1,ures,ures),bfa(5,ures,ures)) ? ;  
no
```

```
| ?- lista_bfa([3,1,5,1,2,4], ures, BF).  
BF = bfa(3,bfa(1,ures,bfa(2,ures,ures)),  
        bfa(5,bfa(4,ures,ures),ures)) ? ;  
no
```

## Akkumulálás bináris fákkal — folyt.

---

### ● Bináris fa konverziója listává

```
% bfa_lista(BF, L0, L): A BF fa levelei az L-L0 listát adják.
% :- pred bfa_lista(bfa::in, list(int)::in,
%               list(int)::out).
bfa_lista(ures, L, L).
bfa_lista(bfa(E, B, J), L0, L) :-
    bfa_lista(J, L0, L1),
    bfa_lista(B, [E|L1], L).
```

### ● Rendezés bináris fával

```
% L lista rendezettje R.
% :- pred rendez(list(int)::in, list(int)::out).
rendez(L, R):-
    lista_bfa(L, ures, BF), bfa_lista(BF, [], R).

| ?- rendez([1,5,3,1,2,4], R).
R = [1,2,3,4,5] ? ;
no
```



# MEGOLDÁSGYŰJTŐ BEÉPÍTETT ELJÁRÁSOK

## Keresési feladat Prologban — felsorolás vagy gyűjtés?

- Keresési feladat: bizonyos feltételeknek megfelelő dolgok meghatározása.
- Prolog nyelven egy ilyen feladat alapvetően kétféle módon oldható meg:
  - gyűjtés — az összes megoldás összegyűjtése, pl. egy listába;
  - felsorolás — a megoldások visszalépéses felsorolása: egyszerre egy megoldást kapunk, de visszalépés esetén sorra előáll minden megoldás.
- Egyszerű példa: egy lista páros elemeinek megkeresése:

### % Gyűjtés:

```
% páros_elemei(L, Pk): Pk az L
% lista páros elemeinek listája.
páros_elemei([], []).
páros_elemei([X|L], Pk) :-
    X mod 2 =\= 0, !,
    páros_elemei(L, Pk).
páros_elemei([P|L], [P|Pk]) :-
    páros_elemei(L, Pk).
```

### % Felsorolás:

```
% páros_eleme(L, P): P egy páros
% eleme az L listának.
páros_eleme([X|L], P) :-
    X mod 2 == 0, P = X.
páros_eleme([_X|L], P) :-
    % _X akár páros, akár páratlan
    % folytatjuk a felsorolást:
    páros_eleme(L, P).

% egyszerűbb megoldás:
páros_eleme2(L, P) :-
    member(P, L), P mod 2 == 0.
```

## Gyűjtés és felsorolás kapcsolata

---

- Vizsgáljuk meg, hogyan lehet egy felsoroló eljárást visszavezetni a gyűjtőre, és fordítva:

- felsorolás gyűjtésből: a member/2 könyvtári eljárás segítségével, pl.

```
páros_eleme(L, P) :-  
    páros_elemei(L, Pk), member(P, Pk).
```

Természetesen ez így nem hatékony!

- gyűjtés felsorolásból: a megoldásgyűjtő beépített eljárások segítségével, pl.

```
páros_elemei(L, Pk) :-  
    findall(P, páros_eleme(L, P), Pk).  
% A páros_eleme(L, P) cél  
% összes P megoldásának listája Pk.
```

## A findall(?Gyűjtő, :Cél, ?Lista) beépített eljárás

---

- Az eljárás végrehajtása (procedurális szemantikája):
  - a Cél kifejezést eljáráshívásként értelmezi, meghívja  
(A : annotáció meta- (azaz eljárás) argumentumot jelez);
  - minden egyes megoldásához előállítja Gyűjtő egy *másolatát*, azaz a megoldásbeli változókat, ha vannak, szisztematikusan újakkal helyettesíti;
  - Az összes Gyűjtő értéket egy listába összegyűjti, és ezt egyesíti Lista-val.

- Példák az eljárás használatára:

```
| ?- findall(X, (member(X, [1,7,8,3,2,4]), X>3), L).
```

```
    => L = [7,8,4] ? ; no
```

```
| ?- findall(X-Y, (between(1, 3, X), between(1, X, Y)), L).
```

```
    => L = [1-1,2-1,2-2,3-1,3-2,3-3] ? ; no
```

- Az eljárás jelentése (deklaratív szemantikája):

$Lista = \{ \text{Gyűjtő másolat} \mid (\exists X \dots Z) \text{Cél igaz} \}$

ahol  $X, \dots, Z$  a findall hívásban levő szabad változók (azaz olyan, a hívás pillanatában behelyettesítetlen változók, amelyek a Cél-ban előfordulnak de a Gyűjtő-ben nem).

## A bagof(?Gyűjtő, :Cél, ?Lista) beépített eljárás

---

- Az eljárás végrehajtása (procedurális szemantikája):
  - a Cél kifejezést eljáráshívásként értelmezi, meghívja;
  - összegyűjti a megoldásait (a Gyűjtő-t és a szabad változók behelyettesítéseit);
  - a szabad változók összes behelyettesítését *felsorolja* és mindegyikhez a Lista-ban megadja az összes hozzá tartozó Gyűjtő értéket.

- Példák az eljárás használatára:

gráf([a-b, a-c, b-c, c-d, b-d]).

```
| ?- gráf(_G), findall(B, member(A-B, _G), VegP).
       $\implies$  VegP = [b,c,c,d,d] ? ; no
| ?- gráf(_G), bagof(B, member(A-B, _G), VegP).
       $\implies$  A = a, VegP = [b,c] ? ;
              A = b, VegP = [c,d] ? ;
              A = c, VegP = [d] ? ; no
```

- A bagof eljárás jelentése (deklaratív szemantikája):

$\text{Lista} = \{ \text{Gyűjtő} \mid \text{Cél igaz} \}, \text{Lista} \neq []$ .

## A bagof megoldásgyűjtő eljárás (folyt.)

### ● Explicit kvantorok

- `bagof(Gyűjtő, V1 ^ ... ^ Vn ^ Cél, Lista)` alakú hívása a  $V1, \dots, Vn$  változókat egzisztenciálisan kötöttnek tekinti, nem sorolja fel.
- jelentése:  $Lista = \{ Gyűjtő \mid (\exists V1, \dots, Vn) Cél \text{ igaz} \} \neq []$ .  
 $| \text{?- gráf}(_G), \text{bagof}(B, A \text{^member}(A-B, _G), VegP).$   
 $\implies VegP = [b,c,c,d,d] ? ; no$

### ● Egymásba ágyazott gyűjtések

- szabad változók esetén a `bagof` nemdeterminisztikus lehet, így skatulyázható:

```
% A G irányított gráf fokszámlistája FL:
% FL = { A - N | N = |{ V | A - V ∈ G }| }
fokszámai(G, FL) :-
    bagof(A-N, Vk(bagof(V, member(A-V, G), Vk),
                    length(Vk, N)
                    ), FL).

| ?- gráf(_G), fokszámai(_G, FL).
     $\implies FL = [a-2,b-2,c-1] ? ; no$ 
```

## A bagof megoldásgyűjtő eljárás (folyt.)

---

- Fokszámlista hatékonyabb előállítás

- a vezérlési szerkezeteket célszerű elkerülni a meta-argumentumokban
- segéd eljárás bevezetésével a kvantor is szükségtelenné válik:

*% Az A pont foka a G irányított gráfban N, N>0.*

pont\_foka(A, G, N) :-

bagof(V, member(A-V, G), Vks), length(Vks, N).

*% A G irányított gráf fokszámlistája FL:*

fokszámai(G, FL) :- bagof(A-N, pont\_foka(A, G, N), FL).

- Példák a bagof/3 és findall/3 közötti kisebb különbségekre:

| ?- findall(X, (between(1, 5, X), X<0), L).  $\implies$  L = [] ? ; no

| ?- bagof(X, (between(1, 5, X), X<0), L).  $\implies$  no

| ?- findall(S, member(S, [f(X,X),g(X,Y)]), L).

$\implies$  L = [f(\_A,\_A),g(\_B,\_C)] ? ; no

| ?- bagof(S, member(S, [f(X,X),g(X,Y)]), L).

$\implies$  L = [f(X,X),g(X,Y)] ? ; no

- A bagof/3 logikailag tisztább mint a findall/3, de időigényesebb!

## A setof(?Gyűjtő, :Cél, ?Lista) beépített eljárás

---

- az eljárás végrehajtása:

- ugyanaz mint: bagof(Gyűjtő, Cél, L0), sort(L0, Lista),
- itt sort/2 egy univerzális rendező eljárás (lásd később), amely
- az eredménylistát rendezzi (az ismétlődések kiszűrésével).

- Példa a setof/3 eljárás használatára:

```
gráf([a-b,a-c,b-c,c-d,b-d]).
```

```
% Gráf egy pontja P.
```

```
pontja(P, Gráf) :- member(A-B, Gráf), ( P = A ; P = B ).
```

```
% A G gráf pontjainak listája Pk.
```

```
gráf_pontjai(G, Pk) :- setof(P, pontja(P, G), Pk).
```

```
| ?- gráf(_G), gráf_pontjai(_G, Pk).  $\implies$  Pk = [a,b,c,d] ? ; no
```



# META-LOGIKAI ELJÁRÁSOK

## A meta-logikai, azaz a logikán túlmutató eljárások fajtái:

---

- A Prolog kifejezések pillanatnyi behelyettesítettségi állapotát vizsgáló eljárások (értelemszerűen sorrendfüggők):

- kifejezések osztályozása (1)

```
| ?- var(X) /* X változó? */, X = 1.  $\implies$  X = 1
| ?- X = 1, var(X).  $\implies$  no
```

- kifejezések rendezése (4)

```
| ?- X @< 3 /* X megelőzi 3-t? */, X = 4.  $\implies$  X = 4
    % a változók megelőzik a nem változó kifejezéseket
| ?- X = 4, X @< 3.  $\implies$  no
```

- Prolog kifejezéseket szétszedő vagy összerakó eljárások:

- (struktúra) kifejezés  $\iff$  név és argumentumok (2)


```
| ?- X = f(alma,körte), X =.. L  $\implies$  L = [f,alma,körte]
```

- névkonstansok és számok  $\iff$  karaktereik (3)

```
| ?- atom_codes(A, [0'a,0'b,0'a])  $\implies$  A = aba
```

## Kifejezések osztályozása

- Kifejezés-osztályok fastruktúrája — osztályozó beépített eljárások (ismétlés)



var(X)	X változó
nonvar(X)	X nem változó
atomic(X)	X konstans
compound(X)	X struktúra
atom(X)	X atom
number(X)	X szám
integer(X)	X egész szám
float(X)	X lebegőpontos szám

- SICStus-specifikus osztályozó eljárások:

- simple(X): X nem összetett (konstans vagy változó);
- ground(X): X tömör, azaz nem tartalmaz behelyettesítetlen változót.

- Az osztályozó eljárások használata — példák

- var, nonvar — többirányú eljárásokban a különböző irányok elágaztatása
- number, atom, ... — nem-megkülönböztetett úniók feldolgozása (pl. szimbolikus deriválás)

## Osztályozó eljárások: elágaztatás behelyettesítettség alapján

---

- Példa: a `length/2` beépített eljárás megvalósítása (SICStus kód!)

*% length(?L, ?N): Az L lista N hosszú.*

`length(L, N) :- var(N), !, length(L, 0, N).`

`length(L, N) :- dlength(L, 0, N).`

*% length(?L, +I0, -I):*

*% Az L lista I-I0 hosszú.*

`length([], I, I).`

`length([_|L], I0, I) :-`

`I1 is I0+1,`

`length(L, I1, I).`

*% dlength(?L, +I0, +I):*

*% Az L lista I-I0 hosszú.*

`dlength([], I, I) :- !.`

`dlength([_|L], I0, I) :-`

`I0 < I, I1 is I0+1,`

`dlength(L, I1, I).`

| `?- length([1,2], Len).` (*length/3*)  $\implies$  `Len = 2 ? ; no`

| `?- length([1,2], 3).` (*dlength/3*)  $\implies$  `no`

| `?- length(L, 3).` (*dlength/3*)  $\implies$  `L = [_A,_B,_C] ? ; no`

| `?- length(L, Len).` (*length/3*)  $\implies$  `L = [], Len = 0 ? ;`

`L = [_A], Len = 1 ? ; L = [_A,_B], Len = 2 ?`

## Struktúrák szétszedése és összerakása: az *univ* eljárás

● Az *univ* eljárás hívási mintái: ●  $+Kif = \dots ?Lista$

●  $-Kif = \dots +Lista$

● Az eljárás jelentése: Igaz, ha

●  $Kif = Fun(A_1, \dots, A_n)$  és  $Lista = [Fun, A_1, \dots, A_n]$ ,  
ahol *Fun* egy névkonstans és  $A_1, \dots, A_n$  tetszőleges kifejezések; vagy

●  $Kif = C$  és  $Lista = [C]$ , ahol *C* egy konstans.

● Példák

?- el(a,b,10) =.. L.	$\implies$	L = [el,a,b,10]
?- Kif =.. [el,a,b,10].	$\implies$	Kif = el(a,b,10)
?- alma =.. L.	$\implies$	L = [alma]
?- Kif =.. [1234].	$\implies$	Kif = 1234
?- Kif =.. L.	$\implies$	<b>hiba</b>
?- f(a,g(10,20)) =.. L.	$\implies$	L = [f,a,g(10,20)]
?- Kif =.. [/ ,X,2+X].	$\implies$	Kif = X/(2+X)
?- [a,b,c] =.. L.	$\implies$	L = ['.',a,[b,c]]

## Struktúrák szétszedése és összerakása: a functor eljárás

- functor/3: kifejezés funktorának, adott funktorú kifejezésnek az előállítás
- Hívási minták: functor(-Kif, +Név, +Argszám)  
functor(+Kif, ?Név, ?Argszám)
- Jelentése: igaz, ha Kif egy Név/Argszám funktorú kifejezés.
  - A konstansok 0-argumentumú kifejezésnek számítanak.
  - Ha Kif kimenő, az adott funktorú legáltalánosabb kifejezéssel egyesíti (argumentumaiban csupa különböző változóval).

### ● Példák:

?- functor(el(a,b,1), F, N).	⇒ F = el, N = 3
?- functor(E, el, 3).	⇒ E = el(_A,_B,_C)
?- functor(alma, F, N).	⇒ F = alma, N = 0
?- functor(Kif, 122, 0).	⇒ Kif = 122
?- functor(Kif, el, N).	⇒ <b>hiba</b>
?- functor(Kif, 122, 1).	⇒ <b>hiba</b>
?- functor([1,2,3], F, N).	⇒ F = '.', N = 2
?- functor(Kif, ., 2).	⇒ Kif = [_A _B]

## Struktúrák szétszedése és összerakása: az arg eljárás

- `arg/3`: kifejezés adott sorszámú argumentuma.
  - Hívási minta: `arg(+Sorszám, +StrKif, ?Arg)`
  - Jelentése: A `StrKif` struktúra `Sorszám`-adik argumentuma `Arg`.
  - Végrehajtása: `Arg`-ot az adott sorszámú argumentummal **egyesíti**.
  - Az `arg/3` eljárás így nem csak egy argumentum elővételére, hanem a struktúra változó-argumentumának behelyettesítésére is használható (ld. a 2. példát alább).

- Példák:

```
| ?- arg(3, el(a, b, 23), Arg).    => Arg = 23
| ?- K=el(_,_,_), arg(1, K, a),
      arg(2, K, b), arg(3, K, 23). => K = el(a,b,23)
| ?- arg(1, [1,2,3], A).          => A = 1
| ?- arg(2, [1,2,3], B).          => B = [2,3]
```

- Az *univ* visszavezethető a functor és arg eljárásokra (és viszont), például:

```
Kif =.. [F,A1,A2]    <=>    functor(Kif, F, 2),
                           arg(1, Kif, A1), arg(2, Kif, A2)
```

## Az *univ* alkalmazása: ismétlődő sémák összevonása

- A feladat: egy szimbolikus aritmetikai kifejezésben a kiértékelhető (infix) részkifejezések helyettesítése az értékükkel.

- 1. megoldás, *univ* nélkül:

```
% Az X szimbolikus kifejezés egyszerűsítése EX.
egysz0(X, EX) :-
    atomic(X), !, EX = X.
egysz0(U+V, EKif) :-
    egysz0(U, EU), egysz0(V, EV),
    kiszamol(EU+EV, EU, EV, EKif).
egysz0(U*V, EKif) :-
    egysz0(U, EU), egysz0(V, EV),
    kiszamol(EU*EV, EU, EV, EKif).
%...
% EU és EV részekből képzett EUV egyszerűsítése EKif.
kiszamol(EUV, EU, EV, EKif) :-
    number(EU), number(EV), !, EKif is EUV.
kiszamol(EUV, _, _, EUV).

| ?- deriv((x+y)*(2+x), x, D), egysz0(D, ED).
    => D = (1+0)*(2+x)+(x+y)*(0+1), ED = 1*(2+x)+(x+y)*1 ? ; no
```



## Az *univ* alkalmazása: ismétlődő sémák összevonása (folyt.)

---

### ● Kifejezés-egyszerűsítés, 2. megoldás, *univ* segítségével

```
egysz(X, EX) :-
    atomic(X), !, EX = X.
egysz(Kif, EKif) :-
    Kif =.. [Muv,U,V],    % Kif = Muv(U,V)
    egysz(U, EU), egysz(V, EV),
    EUV =.. [Muv,EU,EV], % EUV = Muv(EU,EV)
    kiszamol(EUV, EU, EV, EKif).
```

### ● Kifejezés-egyszerűsítés, általánosítás tetszőleges *tömör* kifejezésre:

```
egysz1(Kif, EKif) :-
    Kif =.. [M|ArgL], egysz1_lista(ArgL, EArgL), EKif0 =.. [M|EArgL],
    % catch(:Cél,?Kiv,:KCél): ha Cél kivételt dob, KCél-t futtatja:
    catch(EKif is EKif0, _, EKif = EKif0).

egysz1_lista([], []).
egysz1_lista([K|Kk], [E|Ek]) :-
    egysz1(K, E), egysz1_lista(Kk, Ek).
```

```
| ?- egysz1(f(1+2+a, exp(3,2), a+1+2), E). ==> E = f(3+a,9.0,a+1+2)
```

## Univ alkalmazása általános kifejezés-bejárásra: kiiratás

---

- A feladat: egy tetszőleges kifejezés kiiratása úgy, hogy
  - a kétargumentumú operátorok zárójelezett infix formában,
  - minden más alap-struktúra alakban jelenjék meg.

```

ki(Kif) :-
    compound(Kif), !, Kif =.. [Func, A1|ArgL],
    ( % kétargumentumú kifejezés, funktora infix operátor
      ArgL = [A2], current_op(_, Kind, Func), infix_fajta(Kind)
    -> write('('), ki(A1),
        write(' '), write(Func), write(' '), ki(A2), write(')')
      ; write(Func),
        write('('), ki(A1), arglistaki(ArgL), write(')')
    ).
ki(Kif) :- write(Kif).

% infix_fajta(F): F egy infix operátorfajta.
infix_fajta(xfx). infix_fajta(xfy). infix_fajta(yfx).

% Az [A1,...,An] listát ",A1,...,An" alakban kiírja.
arglistaki([]).
arglistaki([A|AL]) :- write(', '), ki(A), arglistaki(AL).

| ?- ki(f(+a, X*c*X, e)). ==> f(+a),((_117 * c) * _117),e

```

## Univ alkalmazása általános kifejezés-bejárásra: változómentesítés

---

- A SICStus Prologban beépített `numbervars(?Kif, +NO, ?N)` eljárás hatása:
  - A tetszőleges `Kif` minden változóját `'$VAR'(I)` alakú kifejezéssel helyettesíti,  
 $I = NO, \dots, N-1$  (azaz `Kif`-ben  $N-NO$  különböző változó van).
- A `'$VAR'(0), '$VAR'(1), \dots` kifejezések `write`-tal való kiíráskor változónévként (`A, B \dots`) jelennek meg.
- A `write_term(Kif, Opciók)` beépített eljárás kiírja a `Kif` kifejezést, az `Opciók` által meghatározott módon.
- A `numbervars/3` által létrehozott `'$VAR'/1` struktúrák „eredetiben” is megjeleníthetők:
 

```
| ?- _K = [f(_X),g(_),_X], numbervars(_K, 0, N), write(_K), nl,
          write_term(_K, [quoted(true),numbervars(false)]), nl.
===>    [f(A),g(B),A]
          [f('$VAR'(0)),g('$VAR'(1)),$VAR'(0)]
          N = 2
```
- A feladat: elkészítendő egy `numbervars1/3` eljárás, amely `'$VAR'` helyett `'$myvar'` funktort használ.

## Általános kifejezés-bejárás *univ*-val: változómentesítés

---

### ● A változómentesítés egy saját megvalósítása:

```
% A Term kifejezésben levő változókat '$myvar(I)' stb.
% struktúrákkal helyettesíti be, I = NO, ... N-1.
numbervars1(Term, NO, N) :-
    var(Term), !,
    Term = '$myvar'(NO), N is NO+1.
numbervars1(Term, NO, N) :-
    Term =.. [_|Args],
    numbervars1_list(Args, NO, N).

% numbervars1_list(L, NO, N): Az L listában levő változókat
% '$myvar(I)' stb. struktúrákkal helyettesíti be, I = NO, ... N-1.
numbervars1_list([], N, N).
numbervars1_list([A|As], NO, N) :-
    numbervars1(A, NO, N1), numbervars1_list(As, N1, N).

| ?- Kif = [f(_X),g(_),_X], numbervars1(Kif, 0, N).
====>      N = 2,
           Kif = [f('$myvar'(0)),g('$myvar'(1)),$myvar'(0)]
```

## numbervars1 egy alkalmazása

---

### Két kifejezés azonossága

- A kifejezések azonosak, ha változó-behelyettesítés *nélkül* egyesíthetők;
- azaz, ha az egyik változót tartalmaz, akkor a másik ugyanott ugyanazt a változót tartalmazza.
- `azonos/2 == néven, nem_azonos/2 \== néven` szabványos beépített eljárás és operátor.

```

nem_azonos(X, Y) :-
    (   numbervars1(X, 0, N), numbervars1(Y, N, _), X = Y -> fail
    ;   true
    ).

azonos(X, Y) :-
    \+ nem_azonos(X, Y).

% azonos2/2 és azonos/2 teljesen ekvivalens.
% \+ \+ X : csakkor sikeres amikor X, de változóbehelyettesítést nem okoz
azonos2(X, Y) :-
    \+ \+ (numbervars1(foo(X,Y), 0, _), X = Y).

| ?- azonos(X, 1).                ----> no
| ?- azonos(X, Y).                ----> no
| ?- azonos(X, X).                ----> true ?
| ?- append([], L1, L2), azonos(L1, L2). ----> L2 = L1 ?

```

## Univ alkalmazása: részkifejezések keresése

- A feladat: egy tetszőleges kifejezéshez soroljuk fel a benne levő számokat, és minden szám esetén adjuk meg annak a *kiválasztóját*!
- Egy részkifejezés kiválasztója egy olyan lista, amely megadja, mely argumentumpozíciók mentén juthatunk el hozzá.
- Az  $[i_1, i_2, \dots, i_k]$  lista egy Kif-ből az  $i_1$ -edik argumentum  $i_2$ -edik argumentumának,  $\dots$   $i_k$ -edik argumentumát választja ki.
- Pl.  $a*b+f(1,2,3)/c$ -ben  $b$  kiválasztója  $[1,2]$ ,  $3$  kiválasztója  $[2,1,3]$ .

*% kif\_szám(?Kif, ?N, ?Kiv): Kif Kiv kiválasztójú része az N szám.*

kif\_szám(X, N, Kiv) :-

    number(X), !, N = X, Kiv = [].

kif\_szám(X, N, [I|Kiv]) :-

    compound(X), *% a változó kizárása miatt fontos!*

    functor(X, \_F, ArgNo), between(1, ArgNo, I), arg(I, X, X1),

    kif\_szám(X1, N, Kiv).

| ?- kif\_szám(f(1,[b,2]), N, K).

====> K = [1], N = 1 ? ;

    K = [2,2,1], N = 2 ? ; no

## Atomok szétszedése és összerakása

---

- `atom_codes/2`: névkonstans és karakterkód-lista közötti átalakítás
  - Hívási minták: `atom_codes(+Atom, ?KódLista)`  
`atom_codes(-Atom, +KódLista)`
  - Jelentése: Igaz, ha `Atom` karakterkódjainak a listája `KódLista`.
  - Végrehajtása:
    - Ha `Atom` adott (bemenő), és a  $c_1c_2\dots c_n$  karakterekből áll, akkor `KódLista`-t egyesíti a  $[k_1, k_2, \dots, k_n]$  listával, ahol  $k_i$  a  $c_i$  karakter kódja.
    - Ha `KódLista` egy adott karakterkód-lista, akkor ezekből a karakterekből összerak egy névkonstanst, és azt egyesíti `Atom`-mal.

- Példák:

```
| ?- atom_codes(ab, Cs).            $\implies$  Cs = [97,98]
| ?- atom_codes(ab, [0'a|L]).       $\implies$  L = [98]
| ?- Cs="bc", atom_codes(Atom, Cs).  $\implies$  Cs = [98,99], Atom = bc
| ?- atom_codes(Atom, [0'a|L]).     $\implies$  hiba
```

## Atomok szétszedése és összerakása — alkalmazási példák

---

### ● Keresés névkonstansokban

*% Atom-ban a Rész nem üres részatom kétszer ismétlődik.*

`dadogó_rész(Atom, Rész) :-`

`atom_codes(Atom, Cs), dadogó(Cs, Ds), atom_codes(Rész, Ds).`

*% L-ben a D nem üres részlista kétszer ismétlődik (lásd korábban).*

`dadogó(L, D) :- D = [_|_],`

`append(_, Farok, L), append(D, Vég, Farok), append(D, _, Vég).`

`| ?- dadogó_rész(babaruhaha, R).       $\implies$    R = ba ? ; R = ha ? ; no`

### ● Atomok összefűzése

*% atom\_concat(+A, +B, ?C): A és B névkonstansok összefűzése C.*

*% (Szabványos beépített eljárás atom\_concat(?A, ?B, +C) módban is.)*

`atom_concat(A, B, C) :-`

`atom_codes(A, Ak), atom_codes(B, Bk),`

`append(Ak, Bk, Ck),`

`atom_codes(C, Ck).`

`| ?- atom_concat(abra, kadabra, A).  $\implies$  A = abrakadabra ?`



## Számok szétszedése és összerakása

---

- `number_codes/2`: szám és karakterkód-lista közötti átalakítás
  - Hívási minták: `number_codes(+Szám, ?KódLista)`  
`number_codes(-Szám, +KódLista)`
  - Jelentése: Igaz, ha Szám tizes számrendszerbeli alakja a KódLista karakterkód-listának felel meg.
  - Végrehajtása:
    - Ha Szám adott (bemenő), és a  $c_1c_2\dots c_n$  karakterekből áll, akkor KódLista-t egyesíti a  $[k_1, k_2, \dots, k_n]$  kifejezéssel, ahol  $k_i$  a  $c_i$  karakter kódja.
    - Ha KódLista egy adott karakterkód-lista, akkor ezekből a karakterekből összerak egy számot (ha nem lehet, hibát jelez), és azt egyesíti Szám-mal.

- Példák:

?- <code>number_codes(12, Cs).</code>	$\implies$ Cs = [49,50]
?- <code>number_codes(0123, [0'1 L]).</code>	$\implies$ L = [50,51]
?- <code>number_codes(N, " - 12.0e1").</code>	$\implies$ N = -120.0
?- <code>number_codes(N, "12e1").</code>	$\implies$ <b>hiba</b> ( <b>nincs</b> .0)
?- <code>number_codes(120.0, "12e1").</code>	$\implies$ no (a szám adott! :-)

## Kifejezések rendezése: szabványos sorrend

---

- A Prolog szabvány definiálja két tetszőleges Prolog kifejezés szabványos sorrendjét.
- Jelölés:  $X \prec Y$  — az  $X$  kifejezés megelőzi az  $Y$  kifejezést a szabványos sorrendben.
- A szabványos sorrend definíciója:
  1. Ha  $X$  és  $Y$  azonos, akkor sem  $X \prec Y$  sem  $Y \prec X$  nem igaz és fordítva.
  2. Ha  $X$  és  $Y$  különböző kifejezésosztályba tartozik, akkor az osztály dönt:  
*változó*  $\prec$  *lebegőpontos szám*  $\prec$  *egész szám*  $\prec$  *név*  $\prec$  *struktúra*.
  3. Ha  $X$  és  $Y$  változó, akkor az eredmény rendszerfüggő.
  4. Ha  $X$  és  $Y$  lebegőpontos vagy egész szám, akkor  $X \prec Y \Leftrightarrow X < Y$ .
  5. Ha  $X$  és  $Y$  név, akkor sorrendjük megegyezik a lexikografikus (abc) sorrenddel.
  6. Ha  $X$  és  $Y$  struktúrák:
    - 6.1. Ha  $X$  és  $Y$  aritása ( $\equiv$  argumentumszáma) különböző,  $X \prec Y \Leftrightarrow X$  aritása kisebb mint  $Y$  aritása.
    - 6.2. Egyébként, ha a rekordok neve különböző,  $X \prec Y \Leftrightarrow X$  neve  $\prec Y$  neve.
    - 6.3. Egyébként (azonos név, azonos aritás) balról az első nem azonos argumentum dönt.
- (A SICStus Prologban kiterjesztésként megengedett végtelen (ciklikus) kifejezésekre a fenti rendezés nem érvényes.)

## Kifejezések összehasonlítása — beépített eljárások

- Két tetszőleges kifejezés összehasonlítását végző eljárások:

hívás	igaz, ha
$Kif1 == Kif2$	$Kif1 \not\prec Kif2 \wedge Kif2 \not\prec Kif1$
$Kif1 \backslash == Kif2$	$Kif1 \prec Kif2 \vee Kif2 \prec Kif1$
$Kif1 @< Kif2$	$Kif1 \prec Kif2$
$Kif1 @=< Kif2$	$Kif2 \not\prec Kif1$
$Kif1 @> Kif2$	$Kif2 \prec Kif1$
$Kif1 @>= Kif2$	$Kif1 \not\prec Kif2$

- Az összehasonlító eljárások logikailag nem tiszták:

| ?-  $X @< 3, X = 4. \implies X = 4$   
 | ?-  $X = 4, X @< 3. \implies \text{no}$

- Az összehasonlítás mindig a belső ábrázolás szerint történik:

| ?-  $[1, 2, 3, 4] @< \text{struktúra}(1, 2, 3). \implies \text{sikerül (6.1 szabály)}$

## A meta-logikai eljárások egy komplex alkalmazása: $\prec$ megvalósítása

*% T1 megelőzi T2-t a szabványos sorrendben. (Ekvivalens  $T1 @< T2$  -vel, kivéve  
% a változókat, ezek rendezése a T1-T2-beli előfordulásuk szerint történik.)*

`precedes(T1, T2) :-`

`\+ \+ (numbervars(T1-T2, 0, _), prec(T1, T2)).`

*% class(+T, -C): A T kifejezés a C-edik kifejezésosztályba tartozik.*

`class(T, C) :-`

```
(   T='$VAR'(_) -> C=0           % változó
;   float(T)    -> C=1           % lebegőpontos szám
;   integer(T)  -> C=2           % egész szám
;   atom(T)     -> C=3           % névkonstans
;   compound(T) -> C=4           % összetett kifejezés
).
```

*% T1 megelőzi T2-t, a változók már '\$VAR'(n) struktúrákra vannak lecserélve.*

`prec(T1, T2) :-`

```
class(T1, C1), class(T2, C2),
(   C1 == C2 ->
    (   C1 == 1 -> T1 < T2      % 4. szabály (lebegőpontos szám)
    ;   C1 == 2 -> T1 < T2      % 4. szabály (egész szám)
    ;   struct_prec(T1, T2)     % 3., 5. és 6. szabály
    )                               % (változó, név, struktúra)
;   C1 < C2                       % 2. szabály
).
```

## A $\prec$ reláció megvalósítása (folyt.)

---

*% S1 megelőzi S2-t (S1 és S2 struktúra-kifejezés vagy névkonstans).*

```
struct_prec(S1, S2) :-
    functor(S1, F1, N1), functor(S2, F2, N2),
    (   N1 < N2 -> true
    ;   N1 = N2,
        (   F1 = F2 -> args_prec(1, N1, S1, S2)
        ;   atom_prec(F1, F2)
        )
    ).
```

*% Az S1 struktúra-kifejezés N0, ..., N sorszámú argumentumai  
% lexicografikusan megelőzik S2 azonos sorszámú argumentumait.*

```
args_prec(N0, N, S1, S2) :-
    N0 =< N,
    arg(N0, S1, A1), arg(N0, S2, A2),
    (   A1 = A2 -> N1 is N0+1, args_prec(N1, N, S1, S2)
    ;   prec(A1, A2)
    ).
```

*% Az A1 névkonstans megelőzi az A2 névkonstanst.*

```
atom_prec(A1, A2) :-
    atom_codes(A1, C1), atom_codes(A2, C2), struct_prec(C1, C2).
```

# EGYENLŐSÉGFAJTÁK — ÖSSZEFOGLALÁS



## A Prolog egyenlőség-szerű beépített eljárásai

<ul style="list-style-type: none"> <li>• <math>U = V</math>: <math>U</math> egyesítendő <math>V</math>-vel. Soha sem jelez hibát.</li> </ul>	<pre>  ?- X = 1+2.    ==&gt; X = 1+2   ?- 3 = 1+2.    ==&gt; no</pre>
<ul style="list-style-type: none"> <li>• <math>U == V</math>: <math>U</math> azonos <math>V</math>-vel. Soha sem jelez hibát és soha sem helyettesít be.</li> </ul>	<pre>  ?- X == 1+2.    ==&gt; no   ?- 3 == 1+2.    ==&gt; no   ?- +(1,2)==1+2 ==&gt; yes</pre>
<ul style="list-style-type: none"> <li>• <math>U ::= V</math>: Az <math>U</math> és <math>V</math> aritmetikai kifejezések értéke megegyezik. Hibát jelez, ha <math>U</math> vagy <math>V</math> nem (tömör) aritmetikai kifejezés.</li> </ul>	<pre>  ?- X ::= 1+2.    ==&gt; <b>hiba</b>   ?- 1+2 ::= X.    ==&gt; <b>hiba</b>   ?- 2+1 ::= 1+2. ==&gt; yes   ?- 2.0 ::= 1+1. ==&gt; yes</pre>
<ul style="list-style-type: none"> <li>• <math>U \text{ is } V</math>: <math>U</math> egyesítendő a <math>V</math> aritmetikai kifejezés értékével. Hiba, ha <math>V</math> nem (tömör) aritmetikai kifejezés.</li> </ul>	<pre>  ?- 2.0 is 1+1. ==&gt; no   ?- X is 1+2.    ==&gt; X = 3   ?- 1+2 is X.    ==&gt; <b>hiba</b>   ?- 3 is 1+2.    ==&gt; yes   ?- 1+2 is 1+2. ==&gt; no</pre>
<ul style="list-style-type: none"> <li>• <math>(U =.. V</math>: <math>U</math> „szétszedettje” a <math>V</math> lista)</li> </ul>	<pre>  ?- 1+2 =.. X.    ==&gt; X = [+ ,1 ,2]   ?- X =.. [f ,1] .==&gt; X = f(1)</pre>

## A Prolog nem-egyenlőség jellegű beépített eljárásai

- A nem-egyenlőség jellegű eljárások soha sem helyettesítenek be változót!

- $U \backslash= V$ :  $U$  nem egyesíthető  $V$ -vel.

Soha sem jelez hibát.

?- $X \backslash= 1+2.$	$\implies$ no
?- $+(1,2) \backslash= 1+2.$	$\implies$ no

- $U \backslash== V$ :  $U$  nem azonos  $V$ -vel.

Soha sem jelez hibát.

?- $X \backslash== 1+2.$	$\implies$ yes
?- $3 \backslash== 1+2.$	$\implies$ yes
?- $+(1,2) \backslash== 1+2$	$\implies$ no

- $U =\backslash= V$ : Az  $U$  és  $V$  aritmetikai kifejezések értéke különbözik.

Hibát jelez, ha  $U$  vagy  $V$  nem (tömör) aritmetikai kifejezés.

?- $X =\backslash= 1+2.$	$\implies$ <b>hiba</b>
?- $1+2 =\backslash= X.$	$\implies$ <b>hiba</b>
?- $2+1 =\backslash= 1+2.$	$\implies$ no
?- $2.0 =\backslash= 1+1.$	$\implies$ no



## A Prolog (nem-)egyenlőség jellegű beépített eljárásai — példák

		<i>Egyesítés</i>		<i>Azonosság</i>		<i>Aritmetika</i>		
$U$	$V$	$U = V$	$U \backslash = V$	$U == V$	$U \backslash == V$	$U := V$	$U \backslash = V$	$U \text{ is } V$
1	2	<i>no</i>	<i>yes</i>	<i>no</i>	<i>yes</i>	<i>no</i>	<i>yes</i>	<i>no</i>
a	b	<i>no</i>	<i>yes</i>	<i>no</i>	<i>yes</i>	error	error	error
1+2	+(1, 2)	<i>yes</i>	<i>no</i>	<i>yes</i>	<i>no</i>	<i>yes</i>	<i>no</i>	<i>no</i>
1+2	2+1	<i>no</i>	<i>yes</i>	<i>no</i>	<i>yes</i>	<i>yes</i>	<i>no</i>	<i>no</i>
1+2	3	<i>no</i>	<i>yes</i>	<i>no</i>	<i>yes</i>	<i>yes</i>	<i>no</i>	<i>no</i>
3	1+2	<i>no</i>	<i>yes</i>	<i>no</i>	<i>yes</i>	<i>yes</i>	<i>no</i>	<i>yes</i>
X	1+2	$X=1+2$	<i>no</i>	<i>no</i>	<i>yes</i>	error	error	$X=3$
X	Y	$X=Y$	<i>no</i>	<i>no</i>	<i>yes</i>	error	error	error
X	X	<i>yes</i>	<i>no</i>	<i>yes</i>	<i>no</i>	error	error	error

Jelmagyarázat: *yes* — siker; *no* — meghiúsulás, error — hiba.

# IMPERATÍV PROGRAMOK ÁTÍRÁSA PROLOGBA



## Hogyan írjunk át imperatív nyelvű algoritmust Prolog programmá?

---

- Példafeladat: Hatékony hatványozási algoritmus

- Alaplépés: a kitevő felezése, az alap négyzetre emelése.
- Lényegében a kitevő kettes számrendszerbeli alakja szerint hatványoz.

- Az algoritmust megvalósító C nyelvű függvény:

```
/* hatv(a, h) = a**h */  
int hatv(int a, unsigned h)  
{  
    int e = 1;  
    while (h > 0)  
    {  
        if (h & 1) e *= a;  
        h >>= 1; a *= a;  
    }  
    return e;  
}
```

- Az algoritmusban három változó van: a, h, e:

- a és h végértékére nincs szükség,
- e végső értéke szükséges (ez a függvény eredménye).

## A hatv C függvénynek megfelelő Prolog eljárás

- Egy kétargumentumú C függvénynek egy 2+1-argumentumú Prolog eljárás felel meg.
- A függvény eredménye a reláció utolsó argumentuma lesz:  $\text{hatv}(+A, +H, ?E): A^H = E$ .
- A ciklusnak segédeljárás felel meg:  $\text{hatv}(+A0, +H0, +E0, ?E): A0^{H0} * E0 = E$ .
- Az »a« és »h« C változóknak az »+A« és »+H« bemenő *paraméterek* (nem kell a végérték), az »e« C változónak az »+E0, ?E« *akkumulátorpár* felel meg (kezdőérték, végérték).

```

hatv(A, H, E) :-
    hatv(A, H, 1, E).

hatv(A0, H0, E0, E) :- H0 > 0, !,
    (   H0 /\ 1 == 1
        % /\ ≡ bitenkénti "és"
    ->  E1 is E0*A0
    ;   E1 = E0
    ),
    H1 is H0 >> 1,
    A1 is A0*A0,
    hatv(A1, H1, E1, E).
hatv(_, _, E, E).

```

```

int hatv(int a, unsigned h)
{
    int e = 1;

    ism:  if (h > 0)
        {   if (h & 1)
            e *= a;

            h >>= 1;
            a *= a;
            goto ism;
        } else return e;
}

```

## A C ciklus és a Prolog eljárás kapcsolata

---

- A ciklust megvalósító Prolog eljárás minden pontján minden C változónak megfeleltetethető egy Prolog változó (pl. h-nak  $H0$ ,  $H1$ , ...):
  - A ciklusmag elején a C változók a megfelelő Prolog argumentumban levő változónak felelnek meg.
  - Egy C értékadásnak egy új Prolog változó bevezetése felel meg, az ez után következő kódban az új változó felel meg a C változónak.
  - Ha a diszjunkció egyik ága megváltoztat egy változót, akkor a többi ágon is be kell vezetni az új Prolog változót, a régivel azonos értékkel (ld. `if (h & 1) ...`).
- A C ciklusmag végén a Prolog eljárást vissza kell hívni, argumentumaiban az egyes C változóknak pillanatnyilag megfeleltetett Prolog változóval.
- A C ciklus **ciklus-invariánsa** nem más mint a Prolog eljárás fejkommentje, a példában:
 

```
% hatv(+A0, +H0, +E0, ?E):  $A0^{H0} * E0 = E$ .
```

## Programhelyesség-bizonyítás

---

- Egy algoritmus (függvény) specifikációja:
  - **előfeltételek:** a bemenő paramétereknek teljesíteniük kell ezeket,
  - **utófeltételek:** a paraméterek és az eredmény kapcsolatát írják le.
- Egy algoritmus **helyes**, ha minden, az előfeltételeket kielégítő adatra a függvény hibátlanul lefut, és eredményére fennállnak az utófeltételek.
- Példa:  $x = \text{mfoku\_gyok}(a, b, c)$ 
  - előfeltételek:  $b^2 - 4ac \geq 0$ ,  $a \neq 0$
  - utófeltétel:  $ax^2 + bx + c = 0$
  - a program:

```
double mfoku_gyok(a, b, c)
double a, b, c;
{ double d = sqrt(b*b-4*a*c);
  return (-b+d)/2/a;
}
```
- A program helyességének bizonyítása lineáris kódra viszonylag egyszerű.

## Ciklikus programok helyességének bizonyítása

- A ciklusokat „fel kell vágni” egy **ciklus-invariánssal**, amely:
  - az előfeltételekből és a ciklust megelőző értékadásokból következik,
  - ha a ciklus elején fennáll, akkor a ciklus végén is (indukció),
  - belőle és a leállási feltételből következik a ciklus utófeltétele.

```
int hatv(int a0, unsigned h0) /* utófeltétel:  $\text{hatv}(a0, h0) = a0^{h0}$  */
{
  int e = 1, a = a0, h = h0;
  while /* ciklus-invariáns:  $a0^{h0} == e * a^h$  */ (h > 0)
  {
    /* induláskor a kezdőértékek alapján triviálisan fennáll */
    if (h & 1) e *= a;          /*  $e' = e * a^{h \& 1}$  */
    h >>= 1;                  /*  $h' = (h - (h \& 1)) / 2$  */
    a *= a;                   /*  $a' = a * a$  */
  }                          /* indukció:  $e' * a'^{h'} = \dots = e * a^h$  */
  return e;
  /* Az invariánsból  $h = 0$  miatt következik az utófeltétel */
}
```

## Második példa: Fibonacci sorozat tagjainak hatékony számítása

---

### ● A C függvény

```
unsigned fib(unsigned n)
{ unsigned f = 0, fnxt = 1, t;
  while (n > 0) t = fnxt, fnxt += f, f = t, --n; /* (1) */
  return f;
}
```

● Az (1) ciklusnak bemenő változói:  $n$ ,  $f$ ,  $fnxt$ , kimenő változója:  $f$ .

● A ciklusnak megfeleltetett Prolog eljárás:  $\text{fib}(N, F0, FNXT, F)$ :  
az  $F0$  és  $FNXT$  kezdőértékű Fibonacci sorozat  $N$ -edik tagja  $F$ .

```
% "betű szerinti" Prolog átírás:
fib(N, F0, FNXT, F) :- N > 0, !,
    T = FNXT, FNXT1 is FNXT+F0,
    F1 = T, N1 is N-1,
    fib(N1, F1, FNXT1, F).
fib(_, F0, _, F0).
```

```
% Leegyszerűsített alak:
fib(N, F0, FNXT, F) :- N > 0, !,
    FNXT1 is FNXT+F0,
    N1 is N-1,
    fib(N1, FNXT, FNXT1, F).
fib(_, F0, _, F0).
```



## Fibonacci sorozat — Prolog stílusban

---

- A Fibonacci sorozat teljes Prolog megvalósítása, és az ennek megfeleltethető C kód:

```

fib(N, F) :-                                % unsigned fib(unsigned N)
    fib(N, 0, 1, F).                        % { unsigned F0=0, F1=1, F2;
                                           %
fib(N, F0, F1, F) :-                        % ism:
    N > 0, !,                               %   if (N > 0)
    N1 is N-1,                             %   {   --N;
    F2 is F0+F1,                           %       F2 = F0+F1;
                                           %       F0 = F1; F1 = F2;
    fib(N1, F1, F2, F).                    %       goto ism;
                                           %   }
fib(_, F0, _, F0).                         %   return F0;
                                           % }

```

# MEGOLDÁSOK GYŰJTÉSE ÉS FELSOROLÁSA

---

## Keresési feladat Prologban — felsorolás vagy gyűjtés (ism.)?

- Keresési feladat: bizonyos feltételeknek megfelelő dolgok meghatározása.
- Prolog nyelven egy ilyen feladat alapvetően kétféle módon oldható meg:
  - gyűjtés — az összes megoldás összegyűjtése, pl. egy listába;
  - felsorolás — a megoldások visszalépéses felsorolása: egyszerre egy megoldást kapunk, de visszalépés esetén sorra előáll minden megoldás.
- Egyszerű példa: egy lista páros elemeinek megkeresése:

### % Gyűjtés:

```
% páros_elemei(L, Pk): Pk az L
% lista páros elemeinek listája.
páros_elemei([], []).
páros_elemei([X|L], Pk) :-
    X mod 2 =\= 0, !,
    páros_elemei(L, Pk).
páros_elemei([P|L], [P|Pk]) :-
    páros_elemei(L, Pk).
```

### % Felsorolás:

```
% páros_eleme(L, P): P egy páros
% eleme az L listának.
páros_eleme([X|L], P) :-
    X mod 2 == 0, P = X.
páros_eleme([_X|L], P) :-
    % _X akár páros, akár páratlan
    % folytatjuk a felsorolást:
    páros_eleme(L, P).

% egyszerűbb megoldás:
páros_eleme2(L, P) :-
    member(P, L), P mod 2 == 0.
```

## Mi a közös a felsoroló és gyűjtő eljárásokban?

---

- Keressük meg a közös részt a `páros_elemei` és `páros_eleme` eljárásokban!
- Mindkettőben át kell lépni a páratlan elemeket, és meg kell keresni az első páros elemet a listában:

```
% köv_páros(L0, P, L) :- Az L0 első páros eleme P, a maradék L.
köv_páros([X|L0], P, L) :-
    X mod 2 =\= 0, !, köv_páros(L0, P, L).
köv_páros([P|L], P, L).
```

- A `köv_páros` eljárásra épülő gyűjtő és felsoroló eljárások:

```
% páros_elemei(L, Pk): Pk az L
% lista páros elemeinek listája.
páros_elemei(L0, Pk) :-
    köv_páros(L0, P, L1), !,
    Pk = [P|Pk1],
    páros_elemei(L1, Pk1).
páros_elemei(_, []).
```

```
% páros_eleme(L, P): P egy páros
% eleme az L listának.
páros_eleme(L0, P) :-
    köv_páros(L0, P0, L1),
    ( P = P0
    ; páros_eleme(L1, P)
    ).
```

## A gyűjtő és felsoroló sémák összehasonlítása

---

- A páros elemeket gyűjtő ill. felsoroló eljárások alapján adjunk meg egy általános sémát a kétféle eljárástípusra!
- Az általános esetben a keresésnek lehet egy vagy több Param paramétere. Például, kereshetjük a Param-mal osztható elemeket.
- A közös építőelem:  $\text{következő}(V0, \text{Param}, E, V1)$ : A  $V0$  kifejezéssel jellemzett keresési térben az első megoldás  $E$ , és a fennmaradó keresési tér  $V1$ , a Param paraméter-érték mellett.

A gyűjtő séma:

```
% A V0 keresési térben a Param
% paraméterű megoldások listája L.
megoldások(V0, Param, L) :-
    következő(V0, Param, E, V1), !,
    L = [E|L1],
    megoldások(V1, Param, L1).
megoldások(_, _, []).
```

A felsoroló séma:

```
% A V0 keresési térben E egy
% Param paraméterű megoldás.
megoldás(V0, Param, E) :-
    következő(V0, Param, E0, V1),
    ( E = E0
    ;   megoldás(V1, Param, E)
    ).
```

## Egy összetettebb példa: fennsíkok felsorolása

---

- Egy listában fennsíknak nevezünk:
  - egy csupa azonos elemből álló, legalább kételemű, folytonos részlistát;
  - amely az ilyenek között maximális (egyik irányba sem kiterjeszthető).
- A feladat: felsorolandók egy lista fennsíkjai és kezdőpozíciójuk.
- `fennsík(L, F, H)`: Az L listában az F (1-től számozott) pozíción egy H hosszú fennsík van.
- Egy gyorsprogramozási módszerrel készült (Prolog hekker) megoldás:

```
fennsík0(L, F, H) :-
    Teste = [E,E|_],
    append(Eleje, Teste, L),
    \+ last(Eleje, E),
    length(Eleje, F0), F is F0+1,
    kezdehossz(Teste, H).
% kezdehossz/2 definícióját
% lásd korábban
```

```
fennsík1(L, F, H) :-
    Teste = [E,E|_],
    append(Eleje, Teste, L),
    \+ last(Eleje, E),
    length(Eleje, F0), F is F0+1,
    % kezdehossz/2 kifejtve:
    (    append(Ek, Farok, Teste),
        \+ Farok = [E|_] ->
        length(Ek, H)
    ).
```

## Fennsíkok felsorolása — 2., hatékony megoldás

---

● Használjuk a megoldás-felsoroló sémát:  $\text{megoldás}(V0, Param, E)!$

●  $V0$ : »L, P«, a bejárando lista és első elemének pozíciója;

●  $Param$ : üres;

●  $E$ : »F, H«, a megoldás-fennsík kezdőpozíciója és hossza.

% Az L listában az F pozíción egy H hosszú fennsík van.

```
fennsík(L, F, H) :-  
    fennsík(L, 1, F, H).
```

% A P0-tól számozott L0 listában az F pozíción

% egy H hosszú fennsík van.

```
fennsík(L0, P0, F, H) :-  
    % az első fennsík jellemzői F0 és H0,  
    % a fennsík utáni maradéklista L1:  
    első_fennsík(L0, P0, F0, H0, L1),  
    ( F = F0, H = H0  
    ; P1 is F0+H0, % L1 kezdőpozíciója, P1, nem más mint  
                  % az előző megoldás kezdőpozíciója+hossza  
      fennsík(L1, P1, F, H)  
    ).
```

## Fennsíkok felsorolása — 2., hatékony megoldás (folyt.)

---

### ● Az első fennsík előállítása:

```
% első_fennsík(+L0, +P0, -F, -H, -L): A P0-tól számozott L0 listában az
% első fennsík az F. pozíción van és hossza H, a fennsík után fennmaradó
% rész pedig az L lista.
első_fennsík([E,E|L1], P0, F, H, L) :-
    !, F = P0, azonosak(L1, E, 2, H, L).
első_fennsík(_|L1, P0, F, H, L) :-
    P1 is P0+1,
    első_fennsík(L1, P1, F, H, L).

% azonosak(+L0, +E, +H0, -H, -L): Az L0 lista elejéről a maximális számú
% E-vel azonos elemet hagyva marad L, a hagyott elemek száma H-H0.
azonosak([X|L0], E, H0, H, L) :-
    E = X, !,
    H1 is H0+1,
    azonosak(L0, E, H1, H, L).
azonosak(L, _, H, H, L).
```



# MODULARITÁS



## Modulok definiálása SICStus Prolog nyelven

---

- A SICStus Prolog modulfogalmának jellemzői:

- Minden modul külön állományba kell kerüljön.

- Az állomány első programeleme egy modul-parancs kell legyen:

```
:- module( Modulnév, [ExpFunktor1, ExpFunktor2, ...] ).
```

- *ExpFunktor* = az exportálandó eljárás funktora (név/argumentumszám)

- Példa:

```
:- module(platók, [fennsík/3]).           % plato állomány első sora
```

- Modul-betöltésre szolgáló beépített eljárások:

- `use_module(ÁllományNév)`

- `use_module(ÁllományNév, [ImpFunktor1, ImpFunktor2, ...])`

*ImpFunktor* — az importálandó eljárás funktora

- *ÁllományNév* lehet névkonstans, vagy pl. `library(KönyvtárNév):`

```
:- use_module(plato).                    % a fenti modul betöltése
```

```
:- use_module(library(lists), [last/2]). % csak last/2 importált
```

- Modulqualifikált hívási forma: *Modul:Hívás* a *Modul*-ban futtatja *Hívás*-t.

- A modulfogalom nem szigorú, egy nem exportált eljárás is meghívható modulqualifikált formában, pl. `platók:első_fennsík(...)`.

## Meta-eljárások modularizált programban

---

- Eljárásparaméterek átadása gondot okozhat, ha modulközi hívásról van szó:

modul1.pl állomány:

```
:- module(modul1, [kétszer/1]).

% :- meta_predicate kétszer(:).  (*)
kétszer(X) :-
    X, X.

p :- write(bu).
```

modul2.pl állomány:

```
:- module(modul2, [q/0,r/0]).

:- use_module(modul1).

q :- kétszer(p).

r :- kétszer(modul2:p).

p :- write(ba).
```

- Futtatás:

```
| ?- [modul1,modul2].
| ?- q.  => bubu
| ?- r.  => baba
```

- Automatikus modul-kvalifikáció meta-predikátum deklarációval:

Ha modul1.pl-ben elhagyjuk a (\*)-gal jelzett sor előtti % kommentjelet, akkor

```
| ?- q.  => baba!
```

## Meta-predikátum deklaráció, modulnév-kiterjesztés

### ● Meta-predikátum deklaráció

#### ● Formája:

`:- meta_predicate <eljárásnév>(<módspec1>, ..., <módspecn>), ....`

#### ● $\langle \text{módspec}_i \rangle$ lehet ‘:’, ‘+’, ‘-’, vagy ‘?’.

#### ● A ‘:’ mód azt jelzi, hogy az adott argumentumot **betöltéskor** ún. modulnév-kiterjesztésnek kell alávetni. (A többi mód hatása azonos, be/kimenő irányt jelezhetünk segítségükkel.)

### ● Egy $Kif$ kifejezés modulnév-kiterjesztése a következő átalakítást jelenti:

#### ● ha $Kif M:X$ alakú, vagy egy olyan változó, amely az adott eljárás fejében meta-argumentum pozíción szerepelt, akkor változatlanul hagyjuk;

#### ● egyébként helyettesítjük $CurMod:Kif$ -fel, ahol $CurMod$ a kurrens modul.

### ● Példa folyt. (tfh. a modul1-beli kétszer meta-predikátumnak deklarált!)

```
:- module(modul2, [négyszer/1,q/0]).
```

```
:- use_module(modul1).
```

```
q :- kétszer(p).
```

% tárolt alak:

$\Rightarrow$  `q :- kétszer(modul2:p).`

```
:- meta_predicate négyszer(:).
```

```
négyszer(X) :- kétszer(X), kétszer(X).
```

$\Rightarrow$  **változatlan**

# MAGASABBRENDŰ ELJÁRÁSOK

## Magasabbrendű eljárások — listakezelés

---

- Magasabbrendű (vagy meta-eljárás) egy eljárás,
  - ha eljárásként értelmezi egy vagy több argumentumát
  - pl. `call/1`, `findall/3`, `\+ /1` stb.

- Listafeldolgozás `findall` segítségével — példák

- Páros elemek kiválasztása

*% Az L egész-lista páros elemeinek listája Pk.*

`páros_elemei(L, Pk) :-`

`findall(X, (member(X, L), X mod 2 == 0), Pk).`

| `?- páros_elemei([1,2,3,4], Pk).`  $\implies$  `Pk = [2,4]`

- A listaelemek négyzetre emelése

*% Az L számlista elemei négyzeteinek listája Nk.*

`négyzetei(L, Nk) :-`

`findall(Y, (member(X, L), Y is X*X), Nk).`

| `?- négyzetei([1,2,3,4], Nk).`  $\implies$  `Nk = [1,4,9,16]`

## Általános listakezelő meta-eljárások, findall/3-ra építve

---

- Lista szűrése (vö. a filter Erlang függvénnyel!)

*% Az L lista X elemeinek Pred szerinti szűrése FL.*

```
:- meta_predicate filter(+, ?, :, -).
```

```
filter(L, X, Pred, FL) :-
```

```
    findall(X, (member(X, L), call(Pred)), FL).
```

| ?- filter([1,2,3,4], X, X mod 2 == 0, Pk).  $\implies$  Pk = [2,4]

- Lista leképezése (vö. a map Erlang függvénnyel!)

*% Az L lista X elemeit Pred-del Y-ba képezve*

*% kapjuk az ML listát.*

```
:- meta_predicate map(+, ?, :, ?, -).
```

```
map(L, X, Pred, Y, ML) :-
```

```
    findall(Y, (member(X, L), Pred), ML).
```

| ?- map([1,2,3,4], X, Y is X\*X, Y, Nk).  $\implies$  Nk = [1,4,9,16]

- A példákban a szűrést az  $\langle X, \text{Pred} \rangle$  argumentumpár, a leképezést az  $\langle X, \text{Pred}, Y \rangle$  hármas határozza meg. Ezek egy egy- ill. kétargumentumú predikátumot írnak le (vö. a funkcionális nyelvek  $\lambda$ -kifejezéseivel).

## Részlegesen paraméterezett eljáráshívások

---

- A listát elemenként négyzetreemelő eljárás egy másik változata:

```
négyzete(X, Y) :- Y is X*X.
```

```
négyzeteik(Xk, Yk) :- map(Xk, X, négyzete(X,Y), Y, Yk).
```

- A lista elemeire az  $x \rightarrow x^2 + Px + Q$  hozzárendelést alkalmazó eljárás:

```
másodfokú_képe(P, Q, X, Y) :- Y is X*X + P*X + Q.
```

```
másodfokú_képeik(P, Q, Xk, Yk) :- map(Xk, X, másodfokú_képe(P,Q,X,Y), Y, Yk).
```

- Konvenció: a meta-alkalmazásban változó paramétereket az eljárás végére tesszük — így egyszerűsíthető a meta-eljárás hívása.

- Példa: A map/5 eljárásból elhagyjuk az X és Y argumentumokat, és az eljárás-argumentumban sem szerepeltetjük ezeket:

```
másodfokú_képeik(P, Q, Xk, Yk) :- map(Xk, másodfokú_képe(P,Q), Yk).
```

```
map(Xk, RészlPred, Yk) :-
```

```
    % A RészlPred részlegesen paraméterezett hívás kiegészítése Pred-dé:
```

```
    RészlPred =.. L0, append(L0, [X,Y], L), Pred =.. L,    (*)
```

```
    findall(Y, (member(X, Xk), Pred), Yk).
```



## Részlegesen paraméterezett eljáráshívások — segédeszközök

---

- A másodfokú\_képe(P,Q) kifejezés itt a másodfokú\_képe/4 **részlegesen paraméterezett** hívásának tekinthető.
- Ilyen hívások kiegészítésére és meghívására szolgálnak a call/N eljárások.
- call(RPred, A1, A2, ...) végrehajtása: az RPred hívást kiegészíti az A1, A2, ... argumentumokkal, és meghívja.
- A call/N eljárások sok Prologban beépítettek, SICStusban definiálандók:

```
:- meta_predicate call(:, ?), call(:, ?, ?), ....
```

```
% Pred az A utolsó argumentummal meghívva igaz.
```

```
call(M:Pred, A) :-
```

```
    Pred =.. FAs0, append(FAs0, [A], FAs1),
```

```
    Pred1 =.. FAs1, call(M:Pred1).
```

```
% Pred az A és B utolsó argumentumokkal meghívva igaz.
```

```
call(M:Pred, A, B) :-
```

```
    Pred =.. FAs0, append(FAs0, [A,B], FAs2),
```

```
    Pred2 =.. FAs2, call(M:Pred2).
```

```
...
```

## Részlegesen paraméterezett eljárások — rekurzív map/3

---

- A részleges paraméterezés segítségével a map/3 meta-eljárás rekurzívan is definiálható, findall/3 nélkül:

```
% map(Xs, Pred, Ys): Az Xs lista elemeire a Pred transzformációt
% alkalmazva kapjuk az Ys listát.
map([X|Xs], Pred, [Y|Ys]) :-
    call(Pred, X, Y), map(Xs, Pred, Ys).
map([], _, []).
```

- Példák:

```
| ?- map([1,2,3,4], négyzete, L).            $\implies$  L = [1,4,9,16]
| ?- map([1,2,3,4], másodfokú_képe(2,1), L).  $\implies$  L = [4,9,16,25]
```

- A call/N-re épülő megoldás előnyei:

- általánosabb és hatékonyabb lehet, mint a findall-ra épülő;
- alkalmazható akkor is, ha az elemekre elvégzendő műveletek nem függetlenek, pl. foldl.

## Rekurzív meta-eljárások — foldl és foldr

---

● *% foldl(+Xs, :Pred, +Y0, -Y): Y0-ból indulva, az Xs elemeire balról jobbra  
% sorra alkalmazva a Pred által leírt kétargumentumú függvényt kapjuk Y-t.*  
foldl([X|Xs], Pred, Y0, Y) :-  
    call(Pred, X, Y0, Y1), foldl(Xs, Pred, Y1, Y).  
foldl([], \_, Y, Y).

jegyhozzá(Alap, Jegy, Szam0, Szam) :- Szam is Szam0\*Alap+Jegy.

| ?- foldl([1,2,3], jegyhozzá(10), 0, E).  $\implies$  E = 123

● *% foldr(+Xs, :Pred, +Y0, -Y): Y0-ból indulva, az Xs elemeire jobbról balra  
% sorra alkalmazva a Pred kétargumentumú függvényt kapjuk Y-t.*  
foldr([X|Xs], Pred, Y0, Y) :-  
    foldr(Xs, Pred, Y0, Y1), call(Pred, X, Y1, Y).  
foldr([], \_, Y, Y).

| ?- foldr([1,2,3], jegyhozzá(10), 0, E).  $\implies$  E = 321

# DINAMIKUS ADATBÁZISKEZELÉS

## Dinamikus predikátumok

---

- A dinamikus predikátum jellemzői:
  - a program szövegében lehet 0 vagy több klóza;
  - futási időben hozzáadhatunk és elvehetünk klózokat belőle;
  - végrehajtása mindenképpen interpretált.
- Létrehozása
  - programszövegbeli deklarációval:  
:- dynamic(Eljárásnév/Argumentumszám) .  
(ha van klóza a programban, akkor az első előtt — ilyenkor kötelező);
  - futási időben, adatbáziskezelő beépített eljárással
- Adatbáziskezelő eljárások („adatbázis” = a program klózainak összessége):
  - klóz felvétele első, utolsó helyre: asserta/1, assertz/1
  - klóz törlése (illesztéssel, többszörösen sikerülhet): retract/1
  - klóz lekérdezése (illesztéssel, többszörösen sikerülhet): clause/2
- A klózfelvétel ill. törlés **tartós** mellékhatás, visszalépéskor **nem** áll vissza a korábbi állapot.

## Klóz felvétele: asserta/1, assertz/1

---

### ● asserta(:@Klóz)

- A Klóz kifejezést klózként értelmezve felveszi a programba az adott predikátum *első* klózaként. A Klózban levő változók szisztematikusan újakra cserélődnek.
- A '@' mód jelentése: tisztán bemenő paraméter, az eljárás a paraméterbeli változókat nem helyettesíti be (a '+' mód speciális esete).
- A ':' mód modul-kvalifikált paramétert jelez.

### ● assertz(:@Klóz)

- Ugyanaz mint asserta, csak a Klóz kifejezést az adott predikátum *utolsó* klózaként veszi fel.

### ● Példa:

?- assertz((p(1,X):-q(X))), asserta(p(2,0)),	⇒	p(2, 0).
assertz((p(2,Z):-r(Z))), listing(p).	⇒	p(1, A) :- q(A).
	⇒	p(2, A) :- r(A).

```
| ?- assert(s(X,X)), s(U,V), U == V, X \== U.
V = U ? ; no
```

## Klóz törlése: retract/1

---

- `retract(:@Klóz)`

- A Klóz klóz-kifejezésből megállapítja a predikátum funktorát.
- Az adott predikátum klózeit sorra megpróbálja illeszteni Klóz-zal.
- Ha az illesztés sikerült, akkor kitörli a klózt és sikeresen lefut.
- Visszalépés esetén folytatja a keresést (illeszt, töröl, sikerül stb.)

- Példa (folytatás):

| ?- listing(p), retract((p(2,\_):-\_)), listing(p), fail.  $\implies$  no

- A futás kimenete:

p(2, 0).	p(1, A) :-	p(1, A) :-
p(1, A) :-	q(A).	q(A).
q(A).	p(2, A) :-	
p(2, A) :-	r(A).	
r(A).		

## Alkalmazási példa — egyszerűsített findall

---

- A findall1/3 eljárás hatása megegyezik a beépített findall-lal, de
- Nem működik helyesen, ha a Cél-ban újabb findall1 hívás van.

```
:- dynamic(megoldás/1).
```

```
% findall1(Minta, Cél, L): Cél összes megoldására Minták listája L.
```

```
findall1(Minta, Cél, _MegoldL) :-
```

```
    call(Cél),
```

```
    asserta(megoldás(Minta)), % fordított sorrendben vesszük fel!
```

```
    fail.
```

```
findall1(_Minta, _Cél, MegoldL) :-
```

```
    megoldás_lista([], MegoldL).
```

```
% A megoldás/1 tényállításokban tárolt kifejezések fordított listája L-L0.
```

```
megoldás_lista(L0, L) :-
```

```
    retract(megoldás(M)), !,
```

```
    megoldás_lista([M|L0], L).
```

```
megoldás_lista(L, L).
```

```
| ?- findall1(Y, (member(X, [1,2,3]), Y is X*X), ML).  $\implies$  ML = [1,4,9]
```



## Klóz lekérdezése: `clause/2`

---

● `clause(:@Fej, ?Törzs)`

- A `Fej` alapján megállapítja a predikátum funktorát.
- Az adott predikátum klózeit sorra megpróbálja illeszteni a  
`Fej :- Törzs` kifejezéssel (tényállítás esetén `Törzs = true`).
- Ha az illesztés sikerült, akkor sikeresen lefut.
- Visszalépés esetén folytatja a keresést (illeszt, sikerül stb.)

● Példa:

```
:- listing(p), clause(p(2, 0), T).
```

```
p(2, 0).
p(1, A) :-
    q(A).
p(2, A) :-
    r(A).
```

```
T = true ? ;
T = r(0) ? ;
no
```

## A clause eljárás alkalmazása: egyszerű nyomkövető interpreter

---

- Az alábbi interpreter csak „tisztá”, beépített eljárást nem alkalmazó Prolog programok futtatására alkalmas.

```
% interp(G, D): A G cél futását D bekezdésű nyomkövetéssel mutatja.
interp(true, _) :- !.
interp((G1, G2), D) :- !,
    interp(G1, D), interp(G2, D).
interp(G, D) :-
    (   trace(G, D, call)
    ;   trace(G, D, fail), fail    % követi a fail kaput, tovább-hiúsul
    ),
    D2 is D+2,
    clause(G, B), interp(B, D2),
    (   trace(G, D, exit)
    ;   trace(G, D, redo), fail    % követi a redo kaput, tovább-hiúsul
    ).

% A G cél áthaladását a Port kapun D bekezdésű nyomkövetéssel mutatja.
trace(G, D, Port) :-
    /*D szóközt ír ki:*/ tab(D),
    write(Port), write(' '), write(G), nl.
```

## Nyomkövető interpreter - példafutás

```
:- dynamic app/3, app/4.  % (*)
```

```
app([], L, L).
```

```
app([X|L1], L2, [X|L3]) :-  
    app(L1, L2, L3).
```

```
app(L1, L2, L3, L123) :-  
    app(L1, L23, L123),  
    app(L2, L3, L23).
```

- A (\*) sor elhagyható, ha a fenti (mondjuk app34) állományt az alábbi (SICStus-specifikus) beépített eljárással töltjük be:

```
| ?- load_files(app34,  
    compilation_mode(  
        assert_all)).
```

```
| ?- interp(app(_, [b,c], L, [c,b,c,b]), 0).  
call: app(_203, [b,c], _253, [c,b,c,b])  
    call: app(_203, _666, [c,b,c,b])  
    exit: app([], [c,b,c,b], [c,b,c,b])  
    call: app([b,c], _253, [c,b,c,b])  
    fail: app([b,c], _253, [c,b,c,b])  
    redo: app([], [c,b,c,b], [c,b,c,b])  
        call: app(_873, _666, [b,c,b])  
        exit: app([], [b,c,b], [b,c,b])  
    exit: app([c], [b,c,b], [c,b,c,b])  
    call: app([b,c], _253, [b,c,b])  
        call: app([c], _253, [c,b])  
            call: app([], _253, [b])  
            exit: app([], [b], [b])  
        exit: app([c], [b], [c,b])  
    exit: app([b,c], [b], [b,c,b])  
    exit: app([c], [b,c], [b], [c,b,c,b])  
L = [b] ?
```

# NYELVTANI ELEMZÉS PROLOGBAN

## Egy egyszerű nyelvtani elemzési példa

---

- Bináris számok nyelvtana

$$\begin{aligned} \langle \text{szám} \rangle &::= \langle \text{számjegy} \rangle \langle \text{számmaradék} \rangle \\ \langle \text{számmaradék} \rangle &::= \langle \text{számjegy} \rangle \langle \text{számmaradék} \rangle \mid \epsilon \\ \langle \text{számjegy} \rangle &::= 0 \mid 1 \end{aligned}$$

- Ugyanez DCG (Definite Clause Grammar) jelöléssel:

```
szám --> számjegy, számmaradék.
számmaradék --> számjegy, számmaradék | "".
számjegy --> "0" | "1".
```

- A definit klóz nyelvtan (DCG):

- egy általános nyelvtani formalizmus,
- amely egyszerűen Prologra fordítható,
- a legtöbb Prolog rendszer része (bár a szabványnak nem).

## Nyelvtani elemzés „bevetítése” Prologba

---

- Nyelvtani elemzés: annak eldöntése, hogy egy (Prolog listában tárolt) jelsorozat megfelel-e egy adott nem-terminális nyelvtani fogalomnak.
- A lista tetszőleges elemekből állhat, pl. karakterkódok listája, lexikai elemek (token-ek) listája.
- A nem-terminálisoknak kétargumentumú Prolog szabályok felelnek meg, pl.

```
szám -->      számjegy,      számmaradék.
szám(L0, L) :- számjegy(L0, L1), számmaradék(L1, L).
% Az L0 kódlistáról "leelemezhető" egy <szám>, marad L ha
%           L0-ról leelemezhető egy <számjegy>, marad L1, és
%           L1-ről leelemezhető egy <számmaradék>, marad L.
```

- Általánosan: az adott nem-terminálisnak megfelelő jelsorozatot „leelemezve” (lehagyva) egy L0 lista elejéről marad egy L lista.
- Terminális szimbólumok esetén egyetlen elemet kell le hagyni a listáról, erre szolgál a 'C'/3 beépített eljárás. Definíciója: 'C'(L0, X, L) :- L0 = [X|L].  
(A SICStus fordító a 'C'/3 hívást ténylegesen a fenti egyenlőséggel helyettesíti.)
- A „leelemzés” tulajdonképpen akkumulálási folyamat, ahol az elemi akkumulálási lépés: egy terminális le hagyása a lista elejéről ('C'/3).

## A DCG szabályok lefordított alakja

---

### ● A korábbi DCG példa:

```
szám -->           számjegy, számmaradék.           % A | B ≡ A ; B
számmaradék -->    számjegy, számmaradék | "".      % "" ≡ []
számjegy -->       "0" | "1".                       % "0" ≡ [48]
```

### ● A fenti DCG szabályok betöltésekor a következő Prolog kód keletkezik:

```
szám(L0, L) :-
    számjegy(L0, L1), számmaradék(L1, L).

számmaradék(L0, L) :-
    (    számjegy(L0, L1), számmaradék(L1, L)
    ;    L = L0
    ).

számjegy(L0, L) :-
    (    'C'(L0, 48, L)
    ;    'C'(L0, 49, L)
    ).
```

### ● A DCG elemző futtatása:

```
| ?- szám("101", ""). => yes                % "101" ≡ [0'1,0'0,0'1]
| ?- szám("102", L).  => L = "2" ; L = "02" ; no % Valójában L = [50] ; ...
```

## Vezérlési szerkezetek DCG szabályokban

---

- DCG szabályokban használható: vágó, diszjunkció, negáció és feltételes diszjunktív szerkezet.
- Ezek változtatás nélkül átkerülnek a Prolog alakba. Példák:

*% Leelemezhető számjegyek egy MAXIMÁLIS (esetleg üres) listája.*

számmaradék -->

( számjegy -> számmaradék

; []

*% Vigyázat: [] helyett true nem jó!*

).

*% Ugyanez vágóval*

számmaradék --> számjegy, !, számmaradék.

számmaradék --> [] .

*% Figyelem: nincsenek DCG tényállítások!*

*% Az utóbbi Prolog alakja:*

számmaradék(L0, L) :-

számjegy(L0, L1), !, számmaradék(L1, L).

számmaradék(L0, L) :-

L = L0.

| ?- számmaradék("102", L).  $\implies$  L = "2" ; no



## Prolog hívás beillesztése DCG szabályba

---

### ● Általánosabb példa: decimális számjegyek elemzése

```
számjegy --> "0" ; "1" ; "2" ; "3" ; "4" ;  
             "5" ; "6" ; "7" ; "8" ; "9".
```

*% Ugyanez általánosabban és egyszerűbben:*

```
számjegy -->  
    [K],                % K a következő terminális  
    {decimális_jegy_kódja(K)}. % Prolog hívás
```

*% K egy számjegy kódja.*

```
decimális_jegy_kódja(K):-  
    K >= 0'0, K =< 0'9.
```

### ● A fenti DCG szabály Prolog megfelelője:

*% Leelemezhető egy számjegy kódja.*

```
számjegy(L0, L) :-  
    'C'(L0, K, L),      % K a következő terminális  
    decimális_jegy_kódja(K). % megfelelő-e a K?
```

## Az elemző kiegészítése argumentumokkal

- Egy DCG szabály az elemzéssel párhuzamosan további (kimenő) argumentum(ok)ban felépítheti a kielemezett dolog „jelentését”, pl. egy elemzési fát, vagy annak egy kiértékelését.

- Példa: szám elemzése és értékének kiszámítása:

```
% leelemezhető egy Sz értékű decimálisszámjegy-sorozat
szám(Sz) --> számjegy(J), számmaradék(J, Sz).

% leelemezhető számjegyek egy esetleg üres listája, amelynek
% az eddig leelemzett Sz0-val együtt vett értéke Sz.
számmaradék(Sz0, Sz) -->
    számjegy(J), !, {Sz1 is Sz0*10+J}, számmaradék(Sz1, Sz).
számmaradék(Sz0, Sz0) --> [].

% leelemezhető egy J értékű számjegy.
számjegy(J) --> [K], {decimális_jegy_kódja(K), J is K-0'0}.

| ?- szám(Sz, "102 56", L). ==> L = " 56", Sz = 102; no
```

- A számmaradék DCG szabály Prolog alakja:

```
számmaradék(Sz0, Sz, L0,L) :-
    számjegy(J, L0,L1), !, Sz1 is Sz0*10+J, számmaradék(Sz1, Sz, L1,L).
számmaradék(Sz0, Sz0, L0,L) :- L=L0.
```

- Vegyük észre, hogy itt két akkumulátorpár van, egy „kézi” (Sz) és egy DCG-ből generált (L).

## A DCG nyelvtani szabályok szerkezete — összefoglalás

- A DCG szabály alakja:  $\langle \textit{Baloldal} \rangle \text{ --> } \langle \textit{Jobboldal} \rangle$ .
- $\langle \textit{Baloldal} \rangle$ : egy nem-terminális(, amit esetleg terminálisok listája követ).
- $\langle \textit{Jobboldal} \rangle$ : konjunkció (, ), diszjunkció (;), ha-akkor (->) és negáció (\+) segítségével épül fel terminálisokból, nem-terminálisokból és Prolog hívásokból.
- Nem-terminális: tetszőleges *hívható* kifejezés (névkonstans vagy struktúra).
- Terminális: *tetszőleges* Prolog kifejezés; 0, 1 vagy több terminális jel sorozata *listaként* helyezhető el a DCG szabályokban.
- Prolog hívás: {} zárójelekbe zárva helyezhető el (vágó köré nem kell zárójel).
- A DCG egy darab „automatikus” akkumulátort biztosít (az akkumulálási lépés: 'C', egy elem levétele):

$$p(A, \dots) \text{ --> } q_0(B, \dots), \dots, [X], \dots, q_i(C, \dots), \dots, \{\text{Cél}\}, \dots, q_n(D, \dots).$$

$$p(A, \dots, L_0, L) :- q_0(B, \dots, L_0, L_1), \dots, 'C'(L_{i-1}, X, L_i), q_i(C, \dots, L_i, L_{i+1}), \dots, \text{Cél}, \dots, q_n(D, \dots, L_n, L).$$

## DCG példa: kifejezés kiértékelése

### ● Egyszerű aritmetikai kifejezés elemzése és kiértékelése.

*% kif(Z, L0, L): L0 elején egy Z értékű aritmetikai kifejezés áll, marad L.*

kif(Z) --> tag(X), "+", kif(Y), {Z is X + Y}.

kif(Z) --> tag(X), "-", kif(Y), {Z is X - Y}.

kif(X) --> tag(X).

*% tag(Z, L0, L): L0-ból leelemezhető egy Z értékű tag, marad L.*

tag(Z) --> szám(X), "\*", tag(Y), {Z is X \* Y}.

tag(Z) --> szám(X), "/", tag(Y), {Z is X / Y}.

tag(X) --> szám(X).

| ?- kif(Z, "10\*10-6\*6", "").  $\implies$  Z = 64 ; no

| ?- kif(Z, "10\*10-6\*6", L).  $\implies$  L = [], Z = 64 ; L = "\*6", Z = 94 ; ...

| ?- kif(Z, "4-2+1", []).  $\implies$  Z = 1 **Probléma: jobbról balra elemez!**

### ● Egy lehetséges javítás

kif(Z) --> tag(X), kifmaradék(X, Z).

kifmaradék(X, Z) --> "+", tag(Y), W is X + Y, kifmaradék(W, Z).

kifmaradék(X, Z) --> "-", tag(Y), W is X - Y, kifmaradék(W, Z).

kifmaradék(X, X) --> [].

...

## Egy nagyobb DCG példa: „természetes” nyelvű beszélgetés

---

```
:- use_module(library(lists)).

% mondat(Alany, Áll, L0, L): L0-L kielemezhető egy Alany alanyból és Áll
% állítmányból álló mondattá. Alany lehet első vagy második személyű
% névmás, vagy egyetlen szóból álló (harmadik személyű) alany.
mondat(Alany, Áll) -->
    {én_te(Alany, Ige)}, én_te_perm(Alany, Ige, Áll).
mondat(Alany, Áll) -->
    szó(Alany), szavak(Áll).

% én_te(Alany, Ige):
% Az Alany első/második személyű névmásnak megfelelő létige az Ige.
én_te("én", "vagyok").
én_te("te", "vagy").

% én_te_perm(Ki, Ige, Áll, L0, L): L0-L kielemezhető egy Ki
% névmásból, Ige igealakból és Áll állítmányból álló mondattá.
én_te_perm(Alany, Ige, Áll) -->
    (    szó(Alany), szó(Ige), szavak(Áll)
    ;    szó(Alany), szavak(Áll), szó(Ige)
    ;    szavak(Áll), szó(Ige), szó(Alany)
    ;    szavak(Áll), szó(Ige)
    ).
```

## Példa: „természetes” nyelvű beszélgetés — szavak elemzése

---

```
% szó(Sz, L0, L): L0-L egy Sz betűsorozatból álló (nem üres) szó.
szó(Sz) -->
    betű(B), szómaradék(SzM), {illik([B|SzM], Sz)}, köz.

% szómaradék(Sz, L0, L): L0-L egy Sz kódlistából álló (esetleg üres) szó.
szómaradék([B|Sz]) -->
    betű(B), !, szómaradék(Sz).
szómaradék([]) --> [].

% illik(Szó0, Szó): Szó0 = Szó, vagy a kezdő kis-nagy betűben különböznek.
illik([B0|L], [B|L]) :-
    ( B = B0 -> true
    ;   abs(B-B0) == 32
    ).

% köz(L0, L): L0-L nulla, egy vagy több szóköz.
köz --> ( " " -> köz ; "" ).

% betű(K, L0, L): L0-L egy K kódú "betű" (különbözik a " .?" jelektől)
betű(K) --> [K], {\+ member(K, " .?")}.

% szavak(SzL, L0, L): L0-L egy SzL szó-lista.
szavak([Sz|SzK]) -->
    szó(Sz), ( szavak(SzK)
    ;   {SzK = []}
    ).
```

## Példa: „természetes” nyelvű beszélgetés — párbeszéd-szervezés

---

```
% :- type mondás ---> kérdez(szó) ; kijelent(szó,list(szó)) ; un.
% Megvalósít egy párbeszédet.
párbeszéd :-
    repeat,
        read_line(L),    % beolvas egy sort, L a karakterkódok listája
        (   menet(Mondás, L, [])
        -> feldolgoz(Mondás)
        ;   write('Nem értem\n'), fail
        ),
    Mondás = un, !.

% menet(Mondás, L0, L): Az L0-L kielemezett alakja Mondás.
menet(kérdez(Alany)) -->
    {kérdő(Szó)}, mondat(Alany, [Szó]), "?".
menet(kijelent(Alany,Áll)) -->
    mondat(Alany, Áll), ".".
menet(un) -->
    szó("unlak"), ".".

% kérdő(Szó): Szó egy kérdőszó.
kérdő("mi").
kérdő("ki").
kérdő("kicsoda").
```

## Példa: „természetes” nyelvű beszélgetés — válaszok előállítása

---

```
:- dynamic tudom/2.

% feldolgoz(Mondás): feldolgozza a felhasználótól érkező Mondás üzenetet.
feldolgoz(un) :-
    write('Én is.\n').
feldolgoz(kijelent(Alany, Áll)) :-
    assertz(tudom(Alany,Áll)),
    write('Felfogtam.\n').
feldolgoz(kérdez(Alany)) :-
    tudom(Alany, _), !,
    válasz(Alany).
feldolgoz(kérdez(_)) :-
    write('Nem tudom.\n').

% Felsorolja az Alany ismert tulajdonságait.
válasz(Alany) :-
    tudom(Alany, Áll),
    (   member(Szó, Áll), format('~s ', [Szó]), fail
    ;   nl
    ),
    fail.
válasz(_).
```



## Beszélgetős DCG példa — egy párbeszéd

---

| ?- párbeszéd.  
|: Magyar legény vagyok én.  
Felfogtam.  
|: Ki vagyok én?  
Magyar legény  
|: Péter kicsoda?  
Nem tudom.  
|: Péter tanuló.  
Felfogtam.  
|: Péter jó tanuló.  
Felfogtam.  
|: Péter kicsoda?  
tanuló  
jó tanuló  
|: Boldog vagyok.  
Felfogtam.

|: Én vagyok Jeromos.  
Felfogtam.  
|: Te egy Prolog program vagy.  
Felfogtam.  
|: Ki vagyok én?  
Magyar legény  
Boldog  
Jeromos  
|: Okos vagy.  
Felfogtam.  
|: Ki vagy te?  
egy Prolog program  
Okos  
|: Valóban?  
Nem értem  
|: Unlak.  
Én is.

## A DCG formalizmus felhasználása elemzésen kívül

---

- A DCG szabályok kényelmesen használhatók általános akkumulálásra

- Listák akkumulálása — az elemi akkumulálási lépést a 'C'/3 adja

*% anbn(+N, ?L): Az L lista N db a-ból és azt követő N db b-ből áll.*

*% Nem csak elemzésre, hanem L felépítésére is használható!*

*anbn(N, L) :- anbn(N, L, []).*

*% anbn(N, L0, L): L0-L N db a-ból és azt követő N db b-ből áll.*

*anbn(0) --> !.*

*anbn(N) --> {N > 0, N1 is N-1}, [a], anbn(N1), [b].*

*% a fenti DCG szabály kifejtve:*

*anbn(N, L0, L) :-*

*N > 0, N1 is N-1, L0=[a|L1], anbn(N1, L1, L2), L2=[b|L].*

- Egyébként az elemi akkumulálási lépést DCG-n kívül kell megírni:

*% sum(L, S0, S): L összege S-S0.*

*sum([]) --> [].*

*sum([X|L]) -->*

*plus(X), sum(L).*

*% L számlista összege S.*

*sum(L, S) :- sum(L, 0, S).*

*plus(X, S0, S) :- S is S0+X.*

# „HAGYOMÁNYOS” BEÉPÍTETT ELJÁRÁSOK

## Aritmetikai beépített eljárások

- $X \text{ is } Kif$ :  $Kif$  aritmetikai kifejezés kell legyen, értékét egyesíti  $X$ -szel.
- $Kif1 \ \rho \ Kif2$ :  $Kif1$  és  $Kif2$  aritmetikai kifejezések kell legyenek, értékeik között elvégzi a  $\rho$  összehasonlítást ( $\rho$  lehet  $=$ ,  $=\backslash$ ,  $<$ ,  $=<$ ,  $>$ ,  $>=$ ).
- Aritmetikai kifejezésekben felhasználható funktorok:

Infix operátorok			
$+$ összeadás	$//$ egész osztás	$/\backslash$ bitenkénti és	
$-$ kivonás	$**$ hatványozás	$\backslash/$ bitenkénti vagy	
$*$ szorzás	$\text{mod}$ modulus képzés	$<<$ bitenkénti balra léptetés	
$/$ osztás	$\text{rem}$ maradék képzés	$>>$ bitenkénti jobbra léptetés	
Prefix operátorok:	$-$ negáció	$\backslash$ bitenkénti negáció	

Függvény jelölésűek			
$\text{abs}/1$	$\text{exp}/1$	$\text{floor}/1$	$\text{sign}/1$
$\text{atan}/1$	$\text{float}/1$	$\text{log}/1$	$\text{sin}/1$
$\text{ceiling}/1$	$\text{float\_fractional\_part}/1$	$\text{max}/2, \text{min}/2$	$\text{sqrt}/1$
$\text{cos}/1$	$\text{float\_integer\_part}/1$	$\text{round}/1$	$\text{truncate}/1$

## Listakezelő beépített eljárások

---

- Lista hossza: `length(?L, ?N)`
  - Jelentése: az L lista hossza N.
  - `length(-L, +N)` módban adott hosszúságú, csupa különböző változóból álló listát hoz létre.
  - `length(-L, -N)` módban rendre felsorolja a 0, 1, ... hosszú listákat.
  - Megvalósítását lásd korábban.
- Lista rendezése: `sort(@L, ?S)`
  - Jelentése: az L lista @< szerinti rendezése S,  
(==/2 szerint azonos elemek ismétlődését kiszűrve).
- Lista kulcs szerinti rendezése: `keysort(@L, ?S)`
  - Az L argumentum Kulcs-Érték alakú kifejezések listája.
  - Az eljárás jelentése: az S lista az L lista Kulcs értékei szerinti szabványos (@< általi) rendezése, ismétlődéseket nem szűr.

## Kifejezések kiírása

---

- `write(@X)`: Kiírja X-et, ha szükséges operátorokat, zárójeleket használva.
- `writeln(@X)`: Mint `write(X)`, csak gondoskodik, hogy szükség esetén az névkonstansok idézőjelek közé legyenek téve.
- `write_canonical(@X)`: Mint `writeln(X)`, csak operátorok nélkül, minden struktúra szabványos alakban jelenik meg.
- `write_term(@X, +Opciók)`: Az Opciók opciólista szerint kiírja X-et.
- `format(@Formátum, @AdatLista)`: A Formátum-nak megfelelő módon kiírja AdatLista-t. A formázójelek alakja: `~<szám esetleg><formázójel>`.

<code>?- write('Helló világ').</code>	$\Rightarrow$ Helló világ
<code>?- writeln('Helló világ').</code>	$\Rightarrow$ 'Helló világ'
<code>?- write_canonical('*' - '%').</code>	$\Rightarrow$ -(*, '%')
<code>?- write_canonical([1,2]).</code>	$\Rightarrow$ '.(1, '.(2, [])
<code>?- write_term([1,2,3], [max_depth(2)]).</code>	$\Rightarrow$ [1,2 ...]
<code>?- format('X=~s --- ~3d s', [[0'j,0'ó],3245]).</code>	$\Rightarrow$ X=jó --- 3.245 s

## Kifejezések kiírása — felhasználó vezérelte formázás

---

- `print(@X)`: Alapértelmezésben azonos `write`-tal. Ha a felhasználó definiál egy `portray/1` eljárást, akkor a rendszer minden a `print`-tel kinyomtatandó részkifejezésre meghívja `portray`-t. Ennek sikere esetén feltételezi, hogy a kiírás megtörtént, megghiúsulás esetén maga írja ki a részkifejezést.

A rendszer a `print` eljárást használja a változó-behelyettesítések és a nyomkövetés kiírására!

- `portray(@Kif)` (felhasználó által definiálandó ún. *kampó eljárás*): Igaz, ha `Kif` kifejezést a Prolog rendszernek *nem* kell kiírnia (és ekkor maga a `portray` kell, hogy elvégezze a kiírást).

- Példa:

```
portray(Matrix) :-
    Matrix = [[_|_] | _],
    (    member(Row, Matrix),
        nl, print(Row), fail
    ;    true
    ).
```

```
| ?- X = [[1,2],[3,4],[5,6]].
```

```
X =
[1,2]
[3,4]
[5,6] ?
```

## Karakterek kiírása és beolvasása

---

- `put_code(+Kód)`: Kiírja az adott kódú karaktert.
- `tab(+N)`: Kiír  $N$  szóközt feltéve, hogy  $N > 0$ .
- `nl`: Kiír egy soremelést.
- `get_code(?Kód)`: Beolvas egy karaktert és (karakterkódját) egyesíti Kód-dal. (File végénél Kód = -1.)
- `peek_code(?Kód)`: A soronkövetkező karakter kódját egyesíti Kód-dal. A karaktert nem távolítja el a bemenetről. (File végénél Kód = -1.)

- Példa:

```
% rd_line(L): L a következő sor karakterkódjainak listája.
% read_line néven beépített eljárás SICStus 3.9.0-tól.
rd_line(L) :-
    peek_code(0'\n), !, get_code(_), L = [].
rd_line([C|L]) :-
    get_code(C), rd_line(L).

| ?- rd_line(L), tab(20), member(X, L), put_code(X), tab(1), fail ; nl.
|: Hello world!

      H e l l o   w o r l d !
```



## Példa: számbeolvasás

---

```
% számbe(Szám): a Szám szám következik az input-folyamban.
számbe(Szám) :-
    számjegy(Érték),
    számbe(Érték, Szám).

% Az eddig beolvasott Szám0-val együtt az input-folyamban következő
% szám értéke Szám.
számbe(Szám0, Szám) :-
    számjegy(E), !,
    Szám1 is Szám0*10+E,
    számbe(Szám1, Szám).
számbe(Szám, Szám).

% Érték értékű számjegy következik.
számjegy(Érték) :-
    peek_code(Kar),
    Kar >= 0'0, Kar =< 0'9,
    get_code(_),
    Érték is Kar - 0'0.

| ?- számbe(X), get_code(_), számbe(Y).
|: 123 456
       $\implies$  X = 123, Y = 456
```

## Kifejezések beolvasása

---

- `read(?Kif)`: Beolvas egy ponttal lezárt kifejezést és egyesíti `Kif`-fel.  
(File végénél `Kif = end_of_file`.)
- `read_term(?Kif, +Opciók)`: Mint `read/1`, de az `Opciók` opciólistát is figyelembe veszi.
- Példa — botcsinálta programbeolvasó:

```
consult_body :-
    repeat,
        read(Term),
        (   Term = end_of_file -> true
          ;   assertz(Term), fail
        ),
    !.
```

```
| ?- consult_body.
|: p(X) :- q(X), r(X).
|: ^D
yes
```

```
| ?- listing([p/1]).
p(A) :-
        q(A),
        r(A).
yes
```

## Be- és kiviteli csatornák

---

- Csatornák megnyitása és kezelése:

- `open(@Filenév, @Mód, -Csatorna)`: Megnyitja a `Filenév` nevű állományt `Mód` módban (`read`, `write` vagy `append`). A `Csatorna` argumentumban visszaadja a megnyitott csatorna „nyelét”.
- `set_input(@Csatorna), set_output(@Csatorna)`: Az ezt követő beviteli/kiviteli eljárások `Csatorna`-t használják majd (jelenlegi csatorna).
- `current_input(?Csatorna), current_output(?Csatorna)`: A jelenlegi beviteli/kiviteli csatornát egyesíti `Csatorna`-val.
- `close(@Csatorna)`: Lezárja a `Csatorna` csatornát.

- Explicit csatornamegadás be- és kiviteli eljárásokban

- Az eddig ismertetett összes be- és kiviteli eljárásnak van egy eggyel több argumentumú változata, amelynek első argumentuma a csatorna. Ezek: `write/2`, `writeq/2`, `write_canonical/2`, `write_term/3`, `print/2`, `read/2`, `read_term/3`, `format/3`, `put_code/2`, `tab/2`, `nl/1`, `get_code/2`, `peek_code/2`.

## Egy egyszerűbb be- és kiviteli szervezés: DEC10 I/O

---

- `see(@Filenév), tell(@Filenév)`: Megnyitja a `Filenév` file-t olvasásra/írásra és a jelenlegi csatornává teszi. Újabb híváskor csak a jelenlegi csatornává teszi.
- `seeing(?Filenév), telling(?Filenév)`: A jelenlegi beviteli/kiviteli csatorna állománynevét egyesíti `Filenév`-vel.
- `seen, told`: Lezárja a jelenlegi beviteli/kiviteli csatornát.
- Példák — nagyon egyszerű consult variánsok:

```
consult_dec10_style(File) :-  
    seeing(Old), see(File),  
    repeat,  
        read(Term),  
        (    Term = end_of_file  
        ->  seen  
        ;    assertz(Term), fail  
        ),  
    !,  
    see(Old).
```

```
consult_with_streams(File) :-  
    open(File, read, S),  
    repeat,  
        read(S, Term),  
        (    Term = end_of_file  
        ->  close(S)  
        ;    assertz(Term), fail  
        ),  
    !.
```

## Hibakezelési beépített eljárások

---

- Hibahelyzetet beépített eljárás rossz argumentumokkal való meghívása, vagy a `throw/1` (`raise_exception/1`) eljárás válthat ki.
- Minden hibahelyzetet egy Prolog kifejezés (ún. hiba-kifejezés) jellemez.
- Hiba „dobása”, azaz a `HibaKif` hibahelyzet kiváltása:  
`throw(@HibaKif),`  
`raise_exception(@HibaKif)`
- Hiba „elkapása”:  
`catch(:+Cél, ?Minta, :+Hibaág),`  
`on_exception(?Minta, :+Cél, :+Hibaág)`
  - Hatása: Futtatja a Cél hívást.
    - Ha Cél végrehajtása során hibahelyzet nem fordul elő, futása azonos Cél-lal.
    - Ha Cél-ban hiba van, a hiba-kifejezést egyesíti Minta-val.
    - Ha ez sikeres, meghívja a Hibaág-at.
    - Ellenkező esetben továbbdobja a hiba-kifejezést, hogy a további körülvéő `catch` eljárások esetleg elkaphassák azt.

## Programfejlesztési beépített eljárások (SICStus specifikusak)

---

- `set_prolog_flag(+Jelző, @Érték)`: Jelző értékét Érték-re állítja.
- `current_prolog_flag(?Jelző, ?Érték)`: Jelző pillanatnyi értéke Érték.
- Néhány fontos Prolog jelző:
  - `language`: végrehajtási mód (`sicstus`, `iso`).
  - `argv`: csak olvasható, a parancssorbeli argumentumok listája.
  - `unknown`: viselkedés definiálatlan eljárás hívásakor (`trace`, `fail`, `error`).
  - `source_info`: forrásszintű nyomkövetés (`on`, `off`, `emacs`).
- `consult(:@Files), [:@File, ...]`: Betölti a File(ok)at, interpretált alakban.
- `compile(:@File)`: Betölti a File(ok)at, lefordított alakot hozva létre.
- `listing`: Kiírja az összes interpretált eljárást az aktuális kimenetre.
- `listing(:@EljárásSpec)`: Kiírja a megnevezett interpretált eljárásokat.
- Itt és később: `EljárásSpec` — név vagy funktor, eseteg modul-kvalifikációval ellátva, ill. ezek listája, pl. `listing(p)`, `listing([m:q,p/1])`.

## Programfejlesztési eljárások (folytatás)

---

- `statistics`: Különféle statisztikákat ír ki az aktuális kimenetre.
- `statistics(?Fajta, ?Érték)`: Érték a Fajta fajtájú mennyiség értéke.
  - Példa: `statistics(runtime, E)  $\implies$  E=[Tdiff, T]`, Tdiff az előző lekérdezés óta, T a rendszerindítás óta eltelt idő, ezredmásodpercben.
- `break`: Egy új interakciós szintet hoz létre.
- `abort`, `halt`: Kilép a legkülső interakciós szintre ill. a Prolog rendszerből.
- `trace`: Elindítja az interaktív nyomkövetést.
- `debug`, `zip`: Elindítja a szelektív nyomkövetést, csak spion-pontoknál áll meg. (A zip mód gyorsabb, de nem gyűjt annyi információt mint a debug mód.)
- `nodebug`, `notrace`, `nozip`: Leállítja a nyomkövetést.
- `spy(:@EljárásSpec)`: Spion-pontot tesz a megadott eljárásokra.
- `nospy(:@EljárásSpec)`: Megszünteti a megadott spion-pontokat.
- `nospyall`: Az összes spion-pontot megszünteti.

# FEJLETTEBB NYELVI ÉS RENDSZERELEMEK



## Külső nyelvi interfész

---

- Hagyományos (pl. C nyelvű) programrészek meghívásának módja:
  - A Prolog rendszer elvégzi az átalakítást a Prolog alak és a külső nyelvi alak között. Kényelmesebb, biztonságosabb mint a másik módszer, de kevésbé hatékony. Többnyire csak egyszerű adatokra (egész, valós, atom). (MProlog)
  - A külső nyelvi rutin pointereket kap Prolog adatstruktúrákra, valamint hozzáférési algoritmusokat ezek kezelésére. Nehézkesebb, veszélyesebb, de jóval hatékonyabb mint az előző megoldás. Összetett adatok adásvételére is jó. (SWI, SICStus)

## Külső nyelvi interfész — példa

---

- A példa a `library(bdb)` megvalósításából származik.
- A C nyelven megírandó eljárás Prolog hívási alakja:  
`index_keys(+Spec, +Kif, -Kulcs, -Szám)`
- A megírandó eljárás jelentése:
  - Ha *Spec* és *Kif* különböző funktorú kifejezések, akkor *Szám* = -1 és *Kulcs* = [].
  - Egyébként, ha *Spec* valamelyik argumentuma + és *Kif* megfelelő argumentuma változó, akkor *Szám* = -2 és *Kulcs* = [].
  - Egyébként *Szám* a *Spec* argumentumaként előforduló + névkonstansok száma, *Kulcs* pedig *Kif* megfelelő argumentumainak *kivonatából* képzett lista. A kivonat lényegében az argumentum funktora, azzal az eltéréssel, hogy a konstansok kivonata maga a konstans, struktúrák esetén pedig a struktúra neve és az aritása külön elemként kerül a kivonat-listába.

## Külső nyelvi interfész — példa

---

- A példaeljárás használata

```
| ?- [ixtest].  
| ?- index_keys(f(+, -, +, +),  
               f(12.3, _, s(1, _, z(2)), t),  
               Kulcs, Szam).  
Kulcs = [12.3,s,3,t], Szam = 3 ?
```

- Az ixtest.pl Prolog file tartalmazza az interfész specifikációját:

```
foreign(ixkeys, index_keys(+term, +term, -term, [-integer])).  
        % 1. arg: bemenő, általános kifejezés  
        % 2. arg: bemenő, általános kifejezés  
        % 3. arg: kimenő, általános kifejezés  
        % 4. arg: a C függvény értéke, egész (long)  
foreign_resource(ixkeys, [ixkeys]).  
  
:- load_foreign_resource(ixkeys).
```

- A C programot elő kell készíteni a Prolog számára az splfr (link foreign resource) eszköz segítségével:

```
splfr ixkeys ixtest.pl +c ixkeys.c
```

## Külső nyelvi interfész — a C kód (ixkeys.c állomány)

---

```

#include <sicstus/sicstus.h>

#define NA -1 /* not applicable */
#define NI -2 /* instantiatedness */

long ixkeys(SP_term_ref spec,
            SP_term_ref term, SP_term_ref list)
{
    unsigned long sname, tname, plus;
    int sarity, tarity, i;
    long ret = 0;
    SP_term_ref arg = SP_new_term_ref(),
                tmp = SP_new_term_ref();

    SP_get_functor(spec, &sname, &sarity);
    SP_get_functor(term, &tname, &tarity);
    if (sname != tname || sarity != tarity)
        return NA;

    plus = SP_atom_from_string("+");

    for (i = sarity; i > 0; --i) {
        unsigned long t;
        SP_get_arg(i, spec, arg);
        SP_get_atom(arg, &t); /* no check */
        if (t != plus) continue;

        SP_get_arg(i, term, arg);
        switch (SP_term_type(arg)) {
            case SP_TYPE_VARIABLE:
                return NI;
            case SP_TYPE_COMPOUND:
                SP_get_functor(arg, &tname, &tarity);
                SP_put_integer(tmp, (long)tarity);
                SP_cons_list(list, tmp, list);
                SP_put_atom(arg, tname);
                break;
        }
        SP_cons_list(list, arg, list); ++ret;
    }
    return ret;
}

```

## Hasznos lehetőségek SICStus Prolog-ban

---

- Tetszőleges nagyságú egész számok

pl.:

```
| ?- fakt(40,F).
```

```
F = 815915283247897734345611269596115894272000000000 ?
```

- Globális változók (Blackboard)

```
bb_put(Kulcs, Érték)
```

A `Kulcs` kulcs alatt eltárolja `Érték`-et, az előző értéket, ha van, törölve. (`Kulcs` egy (kis) egész szám vagy névkonstans lehet.)

```
bb_get(Kulcs, Érték)
```

Előhívja `Érték`-be a `Kulcs` értékét.

```
bb_delete(Kulcs, Érték)
```

Előhívja `Érték`-be a `Kulcs` értékét, majd kitörli.

## Hasznos lehetőségek SICStus Prolog-ban *(folytatás)*

---

- Visszaléptethető módon változtatható kifejezések

`create_mutable(Adat, ValtKif)`

Adat kezdőértékkel létrehoz egy új változtatható kifejezést, ez lesz ValtKif. Adat nem lehet üres változó.

`get_mutable(Adat, ValtKif)`

Adat-ba előveszi ValtKif pillanatnyi értékét.

`update_mutable(Adat, ValtKif)`

A ValtKif változtatható kifejezés új értéke Adat lesz. Ez a változtatás visszalépéskor visszacsinálódik. Adat nem lehet üres változó.

- Takarító eljárás

`call_cleanup(Hivas, Tiszito)`

Meghívja `call(Hivas)`-t és ha az véglegesen befejezte futását, meghívja `Tiszito`-t. Egy eljárás akkor fejezte be véglegesen a futását, ha további alternatívák nélkül sikerült, megghiúsult vagy kivételt dobott.

## Fejlett vezérlési lehetőségek SICStusban: Blokk-deklarációk

---

- Példa:

```
:- block p(-, ?, -, ?, ?).
```

Jelentése: ha az első és a harmadik argumentum is behelyettesítetlen változó (blokkolási feltétel), akkor a `p` hívás felfüggesztődik.

Ugyanarra az eljárásra több vagylagos feltétel is szerepelhet, pl.

```
:- block p(-, ?), p(?, -).
```

- Végtelen választási pontok kiküszöbölése blokk-deklarációval

```
:- block append(-, ?, -).
```

```
append([], L, L).
```

```
append([X|L1], L2, [X|L3]) :-  
    append(L1, L2, L3).
```


## Blokk-deklarációk (folytatás)

---

- Generál-és-ellenőriz típusú programok gyorsítása
  - általában nem hatékonyak (pl megrajzolja\_1), mert túl sok visszalépést használnak
  - korutinszervezéssel a generáló és ellenőrző rész „automatikusan” összefésülhető
  - ehhez az ellenőrző részt kell előre tenni és megfelelően blokkolni
- Korutinszervezésre épülő programok
  - Példa: egyszerűsített Hamming feladat
    - keressük a  $2^i * 3^j (i \geq 1, j \geq 1)$  alakú számok közül az első N darabot nagyság szerint rendezve.
    - „stream-and-parallelism” közelítésmódot használva korutinszervezéssel egyszerűen lehet megoldani



## Hamming probléma



% A H lista az első N, csak a 2 és 3 tényezőkből álló szám.

hamming(N, H) :-

U = [1|H], times(U, 2, X), times(U, 3, Y),

merge(X, Y, Z), prefix(N, Z, H).

% times(X, M, Z): A Z lista az X elemeinek M-szerese

:- block times(-, ?, ?).

times([A|X], M, Z) :- B is M\*A, Z = [B|U], times(X, M, U).

times([], \_, []).

## Hamming probléma (folyt.)

---

```
% merge(X, Y, Z): Z az X és Y összefésülése.
:- block merge(-, ?, ?), merge(?, -, ?).
% Csak akkor fusson, ha az első két argumentum ismert
merge([A|X], [B|Y], V) :-
    A < B, !, V = [A|Z], merge(X, [B|Y], Z).
merge([A|X], [B|Y], V) :-
    B < A, !, V = [B|Z], merge([A|X], Y, Z).
merge([A|X], [A|Y], [A|Z]) :-
    merge(X, Y, Z).
merge([], X, X) :- !.
merge(_, [], []).

% prefix(N, X, Y): Az X lista első N eleme Y.
prefix(0, _, []) :- !.
prefix(N, [A|X], [A|Y]) :-
    N > 0, N1 is N-1, prefix(N1, X, Y).
```

## Korutinszervező eljárások

---

- `freeze(X, Hivas)`

Hivast felfüggeszti mindaddig, amig X behelyettesítetlen változó.

- `frozen(X, Hivas)`

Az X változó miatt felfüggesztett hívás(oka)t egyesíti Hivas-sal.

- `dif(X, Y)`

X és Y nem egyesíthető. Mindaddig felfüggesztődik, amig ez el nem dönthető.

- `call_residue(Hivas, Maradék)`

Hivas-t végrehajtja, és ha a sikeres lefutás után maradnak felfüggesztett hívások, akkor azokat visszaadja Maradékban. Pl.

```
| ?- call_residue(dif(X, f(Y)), Maradek).
```

```
    => Maradek = [[X]-(prolog:dif(X,f(Y)))]
```

```
| ?- call_residue((dif(X, f(Y)), X=f(Z)), Maradek).
```

```
    => X = f(Z), Maradek = [[Y,Z]-(prolog:dif(f(Z),f(Y)))]
```

## SICStus könyvtárak

---

### ● Könyvtár betöltése

```
:- use_module(library(könyvtárnév)).
```

### ● A legfontosabb könyvtárak

- `arrays` Logaritmikus elérési idejű kiterjeszthető tömbök megvalósítását tartalmazza.
- `assoc` AVL fák segítségével valósítja meg az „asszociációs listák”, azaz véges Prolog kifejezéshalmazokon definiált kiterjeszthető leképezések fogalmát.
- `atts` tetszőleges attributumokat enged a Prolog változókhoz rendelni, ezeket tárolórekeszként és a Prolog egyesítési mechanizmusának módosítására is engedi használni.
- `heaps` A bináris kazal (heap) fogalmát valósítja meg, amely főként prioritásos sorok (priority queue) megvalósítására használható.
- `lists` Biztosítja a listakezelő alapl műveleteket.
- `terms` Különböző kifejezéskezelő eljárásokat tartalmaz.
- `ordsets` Halmazműveleteket definiál (halmaz  $\equiv$  @< szerint rendezett lista).
- `queues` Sorokra (queue, FIFO store) vonatkozó műveleteket definiál.
- `random` Egy véletelenszám-generátort tartalmaz.

- `system` Különféle operációsrendszer-szolgáltatások elérését biztosítja.
- `trees` Az `arrays` könyvtárhoz hasonló, de nem-kiterjeszthető logaritmikus elérési idejű tömbfogalmat valósít meg, bináris fákkal (kicsit hatékonyabb mint az `arrays` könyvtár).
- `ugraphs` Irányított és irányítatlan gráffogalmat valósít meg, élcimkék nélkül.
- `wgraphs` Olyan irányított és irányítatlan gráffogalmat valósít meg, ahol minden él egy egészértékű súllyal rendelkezik.
- `sockets` A socket-ek kezelésére szolgáló eljárásokat biztosít.
- `linda/client` és `linda/server` Linda-szerű processzkommunikációs eszközöket ad.
- `bdb` Felhasználó által definiált többszörös indexelést lehetővé tevő, Prolog kifejezések állományokban való tárolására szolgáló adatbázis-rendszer.
- `clpb` Boole-értékekre vonatkozó feltétel-megoldó (constraint solver).
- `clpq` és `clpr` Feltétel-megoldó a  $Q$  (racionális számok) ill.  $R$  (valós számok) tartományán.
- `clpfd` Véges tartományokra vonatkozó feltétel-megoldó (constraint solver).
- `tcltk` A *Tcl/Tk* nyelv és eszközkészlet elérését biztosítja.
- `gauge` Prolog programok a profilozására szolgáló, a `tcltk` -n alapuló grafikus eszköz.
- `charsio` Karaktersorozatból olvasó ill. abba író be- és kiviteli eljárások gyűjteménye.
- `timeout` Lehetőséget ad arra, hogy célok futási idejét korlátozzuk.

# ÚJ IRÁNYZATOK A LOGIKAI PROGRAMOZÁSBAN

## Új irányzatok a logikai programozásban — kitekintés

---

- Bevezetés a Logikai Programozásba c. jegyzet 6. fejezete:
  - Párhuzamos megvalósítások
  - Az Andorra-I rendszer rövid bemutatása
  - A Mercury nagyhatékonyságú LP megvalósítás
  - CLP (Constraint Logic Programming)
- Az utolsó két témával foglalkozik a „**Nagyhatékonyságú logikai programozás**” c. választható tárgy (általában őszi félévben)
- Rövid izelítőként áttekintjük a korlát-logikai programozás (CLP) témakörét.
- Constraint = megszorítás, kényszer, korlátozás, korlát, ...
- A továbbiakban a „constraint” angol kifejezésre a „korlát” fordítást használjuk

# KORLÁT-LOGIKAI PROGRAMOZÁS – RÖVID ÁTTEKINTÉS



## A korlát-logikai programozás (CLP, Constraint Logic Programming) alapgondolata

---

- A CLP( $\mathcal{X}$ ) séma

Prolog + egy valamilyen  $\mathcal{X}$  adattartományra és azon értelmezett korlátokra (relációkra) vonatkozó „erős” következtetési mechanizmus.

- Példák az  $\mathcal{X}$  tartomány megválasztására

- $\mathcal{X} = \mathbb{Q}$  vagy  $\mathbb{R}$  (a racionális vagy valós számok)

korlátok = lineáris egyenlőségek és egyenlőtlenségek

következtetési mechanizmus = Gauß elimináció és szimplex módszer

- $\mathcal{X} = \text{FD}$  (egész számok Véges Tartománya, angolul FD — Finite Domain)

korlátok = különféle aritmetikai és kombinatorikus relációk

következtetési mechanizmus = MI CSP-módszerek (CSP = Korlát-Kielégítési Probléma)

- $\mathcal{X} = \mathbb{B}$  (0 és 1 Boole értékek)

korlátok = ítétekalkulusbeli relációk

következtetési mechanizmus = MI SAT-módszerek (SAT — Boole kielégíthetőség)

## A CLP következtetés alapelvei

---

### ● A CLP következtetés

- közege az ún. korlát-tár, amelyben a korlátok gyűlnek, egyre pontosabban közelítve a megoldást;
- elemei az ún. primitív korlátok (a megengedett korlátok egy részhalmaza)
- a korlát-tár mindig konzisztens, ellentmondás esetén visszalépés;
- visszalépés esetén a korlát-tár is visszaáll a korábbi állapotba
- a következtetés fajtái:
  - **teljes**, pl. CLP(R) lineáris esetben, CLP(B) — minden korlát bekerül a tárba;
  - **részleges**, pl. CLP(FD) — csak bizonyos egyszerű korlátok mennek a tárba, a többi, nem-primitív korlátok ágensként (démonként) várakoznak arra, hogy:
    - a. primitív korláttá váljanak
    - b. a tárat egy primitív korláttal bővíthessék (az ún. erősítés)

## A SICStus clp(Q,R) könyvtárak

---

### Alapelemek

#### Tartomány:

clpr: lebegőpontos számok, clpq: racionális számok

#### Függvények:

+ - \* / min max pow exp (kétargumentumúak, pow  $\equiv$  exp),

+ - abs sin cos tan (egyargumentumúak).

#### Korlát-relációk: = ::= < > =< >= =\ (= $\equiv$ ::=)

#### Primitív korlátok (a korlát-tár elemei): lineáris kifejezéseket tartalmazó relációk

#### Megoldó algoritmus: lineáris programozási módszerek (Gauss elimináció, szimplex módszer)

### A könyvtár betöltése:

use\_module(library(clpq)), vagy use\_module(library(clpr))

### A fő beépített eljárás

{ *Constraint* }, ahol *Constraint* változókból és (egész vagy lebegőpontos) számokból a fenti műveletekkel felépített reláció, vagy ilyen relációknak a ( , operátorral képzett) konjunkciója.

## Példák a SICStus clpq könyvtárának használatára

---

```
| ?- use_module(library(clpq)).  
  
| ?- {X=Y+4, Y=Z-1, Z=2*X-9}.           % lineáris egyenlet  
    X = 6, Y = 2, Z = 3 ?  
  
| ?- {X+Y+9<4*Z, 2*X=Y+2, 2*X+4*Z=36}.   % egyenlőtlenség is lehet  
    {X<29/5}, {Y= -2+2*X}, {Z=9-1/2*X} ?  
  
| ?- {(Y+X)*(X+Y)/X = Y*Y/X+100}.         % lineárisra egyszerűsíthető  
    {X=100-2*Y} ?  
  
| ?- {(Y+X)*(X+Y) = Y*Y+100*X}.           % így már nem...  
    clpq:{2*(X*Y)-100*X+X^2=0} ?  
  
| ?- {exp(X+Y+1,2) = 3*X*X+Y*Y}.          % nem lineáris...  
    clpq:{1+2*X+2*(Y*X)-2*X^2+2*Y=0} ?  
  
| ?- {exp(X+Y+1,2) = 3*X*X+Y*Y}, X=Y.     % így már igen...  
    X = -1/4, Y = -1/4 ?  
  
| ?- {2 = exp(8, X)}.                     % nem-lineárisak is megoldhatók  
    X = 1/3 ?
```

## A SICStus clpb könyvtár

---

- Alapelemek:

- **Tartomány:** logikai értékek (1 és 0, igaz és hamis)

- **Függvények** (egyben korlát-relációk):

- $\sim P$        $P$  hamis (*negáció*).

- $P * Q$        $P$  és  $Q$  mindegyike igaz (*konjunkció*).

- $P + Q$        $P$  és  $Q$  legalább egyike igaz (*diszjunkció*).

- $P \# Q$        $P$  és  $Q$  pontosan egyike igaz (*kizáró vagy*).

- $P ::= Q$       Ugyanaz mint  $\sim(P \# Q)$ .

- **Constraint-megoldó algoritmus:** Boole-egyesítés.

- A `library(clpb)` könyvtár eljárásai

- `sat(Kifejezés)`, ahol *Kifejezés* változókból, a 0 1 számkonstansokból és névkonstansokból (ún. szimbolikus konstansok) a fenti műveletekkel felépített logikai kifejezés. Hozzáveszi *Kifejezést* a korlát-tárhoz.

- `labeling(Változók)`. Behelyettesíti a *Változókat* 0 1 értékekre, úgy, hogy a tár teljesüljön. Visszalépéskor felsorolja az összes lehetséges értéket.

## Példa a clpb könyvtár használatára: tranzisztoros áramkör verifikálása

```

n(D, G, S) :-                % Gate => Drain = Source
    sat( G*D == G*S).

p(D, G, S) :-                % ~ Gate => Drain = Source
    sat( ~G*D == ~G*S).

xor(A, B, Out) :-
    p(1, A, X),              n(B, X, Out),
    n(0, A, X),              p(A, B, Out),
    p(B, A, Out),            n(X, B, Out).

| ?- n(D, 1, S).             S = D ?


| ?- n(D, 0, S).             true ?

| ?- p(D, 0, S).             S = D ?

| ?- p(D, 1, S).            true ?

| ?- xor(a, b, X).           sat(X==a#b) ?

```



## A SICStus clpfd könyvtár

---

- A clpfd könyvtár alapelemei

- Tartomány: egészek (negatívak is!)

- Függvények (aritmetika):  $+$   $-$   $*$   $/$   $\dots$

- Constraint-relációk

- aritmetikaiak:**  $\#<$ ,  $\#>$ ,  $\#=<$ ,  $\#>=$ ,  $\# = \# \backslash =$

- halmazműveletek:**  $X \text{ in } \textit{Halmaz}$ , pl.  $X \text{ in } 1..5$

- logikai műveletek:**  $\#/\backslash$ ,  $\#\backslash/$ ,  $\#\backslash$  (negáció),  $\#<=>$  (ekvivalencia),  $\dots$

- egyszerű korlátok (korlát tár elemei):  $X \text{ in } \textit{Halmaz}$

- Constraint-megoldó algoritmus:

- aritmetikaiak:** ún. intervallum-konzisztencia (csak a határokat szűkítik)

- halmazműveletek:** teljes konzisztencia (ún. tartomány-konzisztencia)

- A tipikus CLP(FD) megoldási folyamat (forrás: CSP = Constraint Satisfaction Problems)

- a változók tartományának megadása

- korlátok felvétele

- címkézés (visszalépéses keresés) — pl. a `labeling(Options, Változók)` könyvtári eljárás segítségével.

## Példa a clpfd könyvtár használatára: N királynő a sakktáblán

---

```
% A Qs lista N királynő biztonságos elhelyezését mutatja egy N*N-es sakktáblán:
% a lista i. eleme j ==> az i. királynőt az i. sor j. oszlopába kell helyezni.
queens(N, Qs):-                length(Qs, N), domain(Qs, 1, N), safe(Qs).

% safe(Qs): A Qs királynő-lista biztonságos.
safe([]).
safe([Q|Qs]):-                no_attack(Qs, Q, 1), safe(Qs).

% no_attack(Qs, Q, I): A Qs lista által leírt királynők egyike sem támadja a
% Q oszlopban levő királynőt, feltéve hogy Q és Qs távolsága I.
no_attack([],_,_).
no_attack([X|Xs], Y, I):- no_threat(X, Y, I), J is I+1, no_attack(Xs, Y, J).

% Az X és Y oszlopokban I sortávolságra levő királynők nem támadják egymást.
no_threat(X, Y, I) :-        Y #\= X, Y #\= X-I, Y #\= X+I.

| ?- queens(4, Qs).
      Qs = [_A,_B,_C,_D], _A in 1..4, _B in 1..4, _C in 1..4, _D in 1..4 ?
| ?- queens(4, Qs), Qs = [1|_].
      Qs = [1,_A,_B,_C], _A in 3..4, _B in {2}\{4}, _C in 2..3 ?
| ?- queens(4, Qs), Qs = [1|_], labeling([], Qs).
      no
| ?- queens(4, Qs), Qs = [2|_], labeling([], Qs).
      Qs = [2,4,1,3] ?
```



## Egy példasor: Lovagok és lókötők

---

### ● A feladat

- Egy szigeten minden bennszülött lovag vagy lókötő.
- A lovagok mindig igazat mondanak.
- A lókötők mindig hazudnak.
- Egy vagy több bennszülöttnak saját magukra vonatkozó kijelentése alapján meg kell határozni a bennszülött típusát.
- Példa: Találkozunk két bennszülöttel Alfréd-dal és Bélával. Alfréd azt mondja: van köztünk lókötő. Milyen típusú Alfréd és Béla.
- Irodalom: Raymond Smullyan: Mi a címe ennek a könyvnek?, A hölgy és a tigris, Typotex kiadó.
- Továbbfejlesztés: a szigeten lehetnek normális emberek is, akik néha hazudnak, néha igazat mondanak.

## Lovagok és lóköttők – A megoldás elvei

---

- Készítünk egy egyszerű formális nyelvet a bennszülöttek kijelentéseire, pl. Alfréd mondja Alfréd = lóköttő vagy Béla = lóköttő
- A bennszülöttek nevei (pl. Alfréd) Prolog változók, amelyek a lovag vagy lóköttő értéket veszik fel.
- A nyelv egyetlen alap-relációja az =.
- Az összekötő jeleket (mondja, és, vagy, nem) Prolog operátornak deklaráljuk.
- Egy egyszerű Prolog programmal definiáljuk a „bennszülött logikát”, azaz a nyelv állításainak igazságértékét.
- A feladat: egy adott mondat esetén megkeresni azokat a változó-behelyettesítéseket, amelyekre a mondat a „bennszülött logika” szerint igaz lesz.

## Lovagok és lóköttők: 1. változat (Prolog)

---

```

:- op(700, fy, nem).      :- op(900, yfx, vagy).
:- op(800, yfx, és).      :- op(950, xfy, mondja).

% Az A bennszülött mondhatja az Áll állítást.
A mondja Áll :- értéke(A mondja Áll, 1).

% értéke(Állítás, Érték): Állítás igazságértéke Érték (1 = igaz, 0 = hamis).
értéke(X = X, 1).
értéke(X = Y, 0) :-      különbözõ(X, Y).
értéke(lovag mondja M, E) :- értéke(M, E).
értéke(lóköttõ mondja M, E) :- értéke(nem M, E).
értéke(M1 és M2, E) :-      értéke(M1, E1), értéke(M2, E2),  E is E1 /\ E2.
értéke(M1 vagy M2, E) :-      értéke(M1, E1), értéke(M2, E2),  E is E1 \/ E2.
értéke(nem M, E) :-          értéke(M, E1),                      E is 1-E1.

% különbözõ(A, B): A és B különbözõ típusú bennszülöttek.
különbözõ(lovag, lóköttõ).      különbözõ(lóköttõ, lovag).

| ?- Alfréd mondja Alfréd = lóköttõ vagy Béla = lóköttõ.
    Béla = lóköttõ, Alfréd = lovag ? ; no

| ?- A mondja B = C.
    A = lovag, C = B ? ;
    A = lóköttõ, B = lovag, C = lóköttõ ? ;
    A = lóköttõ, B = lóköttõ, C = lovag ? ; no

```

## Lovagok és lóköttők: 2., CLP(B) változat

---

(A bennszülöttek típusát numerikusan jelöljük: lovag  $\rightarrow 1$ , lóköttő  $\rightarrow 0$ .)

```
:- use_module(library(clpb)).

:- op(700, fy, nem).      :- op(900, yfx, vagy).
:- op(800, yfx, és).      :- op(950, xfy, mondja).

A mondja Áll :- értéke(A mondja Áll, 1).

% értéke(Állítás, Érték): Az Állítás igazságértéke Érték.
értéke(X = Y, E) :-      sat((X == Y) == E).
értéke(X mondja M, E) :- értéke(M, E0), sat((E0 == X) == E).
értéke(M1 és M2, E) :-   értéke(M1, E1), értéke(M2, E2), sat(E == E1*E2).
értéke(M1 vagy M2, E) :- értéke(M1, E1), értéke(M2, E2), sat(E == E1+E2).
értéke(nem M, E) :-      értéke(M, E0), sat(E == ~E0).

| ?- Alfréd mondja Alfréd = 0 vagy Béla = 0.
      Béla = 0, Alfréd = 1 ? ; no
| ?- A mondja B mondja C mondja A = C.
      B = 1 ? ; no
| ?- A mondja B = C.
      sat(B=\=C#A) ? ; no
| ?- A mondja B = C, labeling([A,B,C]).
      A = 0, B = 1, C = 0 ? ; A = 0, B = 0, C = 1 ? ;
      A = 1, B = 0, C = 0 ? ; A = 1, B = 1, C = 1 ? ; no
```

## Lovagok és lóköttők: 3., CLP(FD) változat

---

```
:- use_module(library(clpfd)).
```

```
:- op(700, fy, nem).      :- op(900, yfx, vagy).
:- op(800, yfx, és).      :- op(950, xfy, mondja).
```

```
A mondja Áll :- értéke(A mondja Áll, 1).
```

```
% értéke(Állítás, Érték): Az Állítás igazságértéke Érték.
```

```
értéke(X = Y, E) :-      X in 0..1, Y in 0..1, E #<=> (X #= Y).
```

```
értéke(X mondja M, E) :- X in 0..1, értéke(M, E0), E #<=> (E0 #= X).
```

```
értéke(M1 és M2, E) :-   értéke(M1, E1), értéke(M2, E2), E #<=> E1 #/\ E2.
```

```
értéke(M1 vagy M2, E) :- értéke(M1, E1), értéke(M2, E2), E #<=> E1 #\/ E2.
```

```
értéke(nem M, E) :-      értéke(M, E0), E #<=> #\ E0.
```

```
| ?- Alfréd mondja Alfréd = 0 vagy Béla = 0.
```

```
    Alfréd in 0..1, Béla in 0..1 ? ; no
```

```
| ?- Alfréd mondja Alfréd = 0 vagy Béla = 0, labeling([], [Alfréd,Béla]).
```

```
    Béla = 0, Alfréd = 1 ? ; no
```

```
| ?- A mondja B = C, labeling([], [A,B,C]).
```

```
    A = 0, B = 0, C = 1 ? ; A = 0, B = 1, C = 0 ? ;
```

```
    A = 1, B = 0, C = 0 ? ; A = 1, B = 1, C = 1 ? ; no
```

## Lovagok, lóköttők (és normálisak): 4., CLP(FD) változat

---

(A bennszülöttek típusa: normális  $\rightarrow 2$ , lovag  $\rightarrow 1$ , lóköttő  $\rightarrow 0$ .)

```
:- use_module(library(clpfd)).

:- op(700, fy, nem).      :- op(900, yfx, vagy).
:- op(800, yfx, és).      :- op(950, xfy, mondja).

A mondja Áll :- értéke(A mondja Áll, 1).

% értéke(Állítás, Érték): Az Állítás igazságértéke Érték.
értéke(X = Y, E) :-      X in 0..2, Y in 0..2, E #<=> (X #= Y).
értéke(X mondja M, E) :- X in 0..2, értéke(M, E0), E #<=> (X #= 2 #\ E0 #= X).
értéke(M1 és M2, E) :-   értéke(M1, E1), értéke(M2, E2), E #<=> E1 #/\ E2.
értéke(M1 vagy M2, E) :- értéke(M1, E1), értéke(M2, E2), E #<=> E1 #/\ E2.
értéke(nem M, E) :-      értéke(M, E0), E #<=> #\ E0.

% http://www.math.wayne.edu/~boehm/Probweek2w99sol.htm: We are given three
% people, A, B, C, one of whom is a knight, one a knave, and one a normal
% (but not necessarily in that order). They make the following statements.
%   A: I am normal, B: A is telling the truth, C: I am not normal
% What are A, B, and C?

| ?- A mondja A = 2, B mondja A = 2, C mondja nem C =2, all_different([A,B,C]),
    labeling([], [A,B,C]).
    A = 0, B = 2, C = 1 ? ; no
```

## CLP rendszerek a nagyvilágban

---

- Néhány implementáció

- clp(R) — az első CLP(X) rendszer (Monash Univ, Australia, IBM és CMU)
- CHIP — FD, Q és B (ECRC, Németo., Cosytec, Franciaó.); CHARME (Bull); Decision Power (ICL)
- Prolog III, Prolog IV (PrologIA, Marseille), Q (nem-lineáris is), B, FD, listák, intervallumok
- ILOG solver (ILOG, Franciaó.) — C++ könyvtár: R (nem-lineáris is), FD, halmazok
- SICStus Prolog (SICS, Svédo.) — R/Q, FD, B
- GNU Prolog (INRIA, Franciaó.) — FD (C-re fordít)
- Oz (DFKI, Németo.) — korlát alapú elosztott funkcionális nyelv.

- Kommerciális rendszerek (a fentiek között)

- ILOG, CHIP, Prolog III–IV, SICStus
- a szakma óriása: ILOG
  - szakterület: CLP + vizualizációs eszközök + szabályalapú eszközök
  - felvásárolta az egyik vezető operációkutatási céget, a CPLEX-et
  - 400 munkatárs 7 országban, 55M USD éves bevétel, NASDAQ-on jegyzett

## Mire használják a CLP rendszereket — néhány példa

---

- Ipari erőforrás optimalizálás
  - termék- és gépkonfiguráció
  - gyártásütemezés
  - emberi erőforrások ütemezése
  - logisztikai tervezés
- Közlekedés, szállítás
  - repülőtéri allokációs feladatok (beszállókapu, poggyász-szalag stb.)
  - repülő-személyzet járatokhoz rendelése
  - menetrendkészítés
  - forgalomtervezés
- Távközlés, elektronika
  - GSM átjátszók frekvencia-kiosztása
  - lokális mobiltelefon-hálózat tervezése
  - áramkörtervezés és verifikálás